Semiconductor Parameter Analyzer

Programmer's Guide

Printing History
HP Part Number
04155-90110
First edition, February 1994
Second edition, December 1994

Printed in Japan

Product Warranty

This Hewlett-Packard product is warranted against defects in material and workmanship for a period of one year from date of shipment. During the warranty period, Hewlett-Packard will, at its option, either repair or replace products which prove to be defective.

For warranty service or repair, this product must be returned to a service facility designated by Hewlett-Packard. Buyer shall prepay shipping charges to Hewlett-Packard and Hewlett-Packard shall pay shipping charges to return the product to Buyer. However, Buyer shall pay all shipping charges, duties, and taxes for products returned to Hewlett-Packard from another country.

Hewlett-Packard warrants that its software and firmware designated by Hewlett-Packard for use with an instrument will execute its programming instructions when properly installed on that instrument. Hewlett-Packard does not warrant that the operation of the instrument, or software, or firmware will be uninterrupted or error free.

Limitation of Warranty

The foregoing warranty shall not apply to defects resulting from improper or inadequate maintenance by Buyer, Buyer-supplied software or interfacing, unauthorized modifications or misuse, operation outside of the environment specifications for the products, or improper site preparation or maintenance.

No other warranty is expressed or implied. Hewlett-Packard specifically disclaims the implied warranties of merchantability and fitness for a particular purpose.

Exclusive Remedies

The remedies provided herein are the Buyer's sole and exclusive remedies. Hewlett-Packard shall not be liable for any direct, indirect, special, incidental, or consequential damages, whether based on contract, tort, or any other legal theory.

Assistance

Product maintenance agreements and other customer assistance agreements are available for Hewlett-Packard products.

For any assistance, contact your nearest Hewlett-Packard Sales Office.

Certification

Hewlett-Packard Company certifies that this product met its published specifications at the time of shipment [from the factory]. Hewlett-Packard further certifies that its calibration measurements are traceable to the **National Institute of Standards and Technology** [NIST], to the extent allowed by the Institute's calibration facility, and to the calibration facilities of other International Standards Organization members.

Yokogawa-Hewlett-Packard, Ltd.
9-1, Takakura-Cho, Hachioji-Shi,
Tokyo, 192 Japan

Programming Overview

The HP 4155A/56A can be *fully* controlled from an external computer or by using built-in HP Instrument BASIC (IBASIC) controller. IBASIC is a programming environment that allows full control of HP 4155A/56A without using an external computer.

To fully control the HP 4155A/56A, you execute programs that contain SCPI commands. SCPI means Standard Commands for Programmable Instruments. SCPI is the default program language of the HP 4155A/56A, and can control all functions of the HP 4155A/56A. These programs can be run from IBASIC or an external computer.

The HP 4155A/56A has two command modes:

- *HP 4155/56 command mode* is the default mode of HP 4155A/56A. You use this mode in *all* cases, *except* when directly running an HP 4145A/B program. In this mode, you can use a program that contains SCPI commands, which allow you to control *all* functions of the HP 4155A/56A.
- *HP 4145 syntax command mode*, which allows you to execute HP 4145A/B programs on the HP 4155A/56A directly with little or no modification. In this command mode, you *cannot* control all functions of HP 4155A/56A.

How to Migrate HP 4145A/B Programs

HP 4145A/B Auto Sequence Program (ASP) programs run on the HP 4145A/B built-in programming environment and allow *basic* control of HP 4145A/B without using an external computer. To run the ASP programs on the HP 4155A/56A, you do one of the following and execute the program in the *HP 4155/56 command mode*:

- Create a program that performs the same operations as the desired ASP program by using the IBASIC editor typing aid softkeys to enter commands that correspond to each ASP command. This program can run on IBASIC only, *not* on an external computer. Refer to "Creating ASP-like IBASIC Programs" in Chapter 1 for details.
- Create a program using SCPI commands that performs same operations as the desired ASP program. This program can run on IBASIC or on an external computer. Refer to "Programming Example for HP 4145 Users" in Chapter 4 for details.

HP 4145A/B HP-IB programs run on an external computer and allow *full* control of the HP 4145A/B. To run these programs on the HP 4155A/56A, do one of the following:

- Directly run the HP 4145A/B program on the HP 4155A/56A with little or no modification. You must run this program in the *HP 4145 syntax command mode* from IBASIC or an external computer. Refer to Chapter 5 for details.
- Create a program using SCPI commands that performs same operations as the HP 4145A/B program. You must run this program in *HP 4155/56 command mode* from IBASIC or an external computer.

In This Manual

This manual describes how to control the HP 4155A/4156A by using HP-IB commands from an external computer or built-in HP Instrument BASIC.

This manual consists of the following chapters:

- Using HP Instrument BASIC
- Reference: HP Instrument BASIC
- Getting Started on Programming the HP 4155A/4156A
- HP 4155A/4156A SCPI Programming
- Running HP 4145A/B Program Directly on HP 4155A/4156A
- Sample Application Programs
- Manual Changes Depending on ROM Version

Refer to the HP 4155A/4156A *HP-IB Command Reference* for detailed syntax of each command.

See the HP 4155A/4156A *User's Task Guide* and *User's Dictionary Reference* for information about HP 4155A/4156A itself.

Text Conventions.
The following text conventions are used in this manual:

Front-panel key	Represents a key physically located on HP 4155A/4156A.
`Softkey`	Represents a softkey that appears on screen of HP 4155A/4156A.
`Screen Text`	Represents text displayed on HP 4155A/4156A.
Italic	Refers to a related document, or is used for emphasis.

Contents

Figures

Tables

1

Using HP Instrument BASIC

Using HP Instrument BASIC

HP Instrument BASIC (IBASIC) is a system controller built into your HP 4155A/4156A. By using IBASIC, you can run a program to control the HP 4155A/4156A and other instruments (connected via interfaces of HP 4155A/4156A) without using an external computer.

IBASIC is a subset of HP BASIC. Programs created by IBASIC can run on an HP BASIC controller with little or no modification.

This chapter consists of the following ten sections.

The following six sections provide step-by-step instructions to operate IBASIC by using examples. You can learn the basics of IBASIC programming and operations. If you are not familiar with IBASIC, we recommend to read through these sections first.

- Before Operating IBASIC
- Creating and Executing a Simple IBASIC Program
- Modifying Program by using Editor Functions
- Saving and Getting a Program
- Summary of Softkeys and Keyboard Operations for Editor
- Other Basic Features of IBASIC

The following three sections are a task oriented reference for IBASIC. You can quickly find the desired IBASIC task.

- IBASIC Basic Operation Tasks
- IBASIC Editor Tasks
- Control from External Computer

The following section introduces how to easily create a program by using the typing aid softkeys in the IBASIC editor. This method of creating a program is similar to using the ASP environment on the HP 4145A/B semiconductor parameter analyzer.

- Creating ASP-like IBASIC Programs

Before Operating IBASIC

HP 4155A/4156A provides the following three screen modes for operating IBASIC.

"All IBASIC" screen	Entire screen including softkeys is used for IBASIC, so no instrument page is displayed. You can execute programs, but no instrument page appears in this mode.
"IBASIC Status" screen	Softkeys and bottom two lines are used for IBASIC. Rest of screen is for instrument page. In this mode, you can start the IBASIC editor. The displayed softkeys are for IBASIC operation. You can execute IBASIC commands interactively. Characters you type are displayed at the bottom of the screen.
"All Instrument" screen	.This is regular instrument screen and the default display mode at power on. Entire screen is for instrument page, and all softkeys are for interactive use of instrument. In this mode, you *cannot* use the IBASIC editor. Only the front-panel keys of IBASIC key group and Ctrl+U (Run) and Ctrl+P (Pause) on external keyboard are available to execute or pause program for HP Instrument BASIC from this screen mode.

For details about HP Instrument BASIC screens, refer to "IBASIC Screen" in Chapter 2.

To Switch Screen Mode

To switch the screen mode, repeat the following operation until the desired screen is displayed.

Front Panel

1. Press (Display) of IBASIC key group.

External Keyboard

1. Press (Ctrl)+(G).

This operation toggles screen display as follows:



PG02003 120x20

To Use the Help Function

By using the built-in help function of the HP 4155A/4156A, you can easily get information (name, syntax, and description) about programming commands, and can enter the desired command into the program without typing.

To start the help function for the programming commands, press (Help) key while you are in the IBASIC editor.

In the help function, the programming commands are divided into the following three categories, which you can access by secondary softkeys.

Softkey	Category
`IBASIC`	IBASIC commands.
`PAGE IMAGE COMMAND`	SCPI commands specific for HP 4155A/4156A. These are the help commands associated with the instrument setup pages that begin with `:PAGE`.
`SCPI COMMAND`	Standard SCPI commands.

The upper part of the help screen displays a list of the command names. The lower part displays a description of the selected (by field pointer) command.

To Move the Field Pointer.
To move the field pointer, refer to the following table:

Rotary knob or Arrow keys	Basically, you move the field pointer by using the rotary knob or arrow keys.
PAGE CONTROL keys	Field pointer moves to first PAGE IMAGE command that is associated with the pressed key. PAGE CONTROL keys are (Chan), (Meas), (Display), (Graph/List), (Stress), and (System).
MEASUREMENT keys	Field pointer moves to the PAGE IMAGE command that is associated with the pressed key. MEASUREMENT keys are (Single), (Repeat), and (Append).
(Get), (Save)	Field pointer moves to the associated SCPI command.
Alphabetical keys	Field pointer moves to next command that has a keyword that begins with same letter as the pressed key. If you are in the PAGE IMAGE command category, search is only within the instrument *page group* of the currently selected command.

To Search for a Command.
To search for a command:

1. Press `SEARCH` secondary softkey.
2. Type in command string that you want to search for, then press (Enter).

To Enter a Command into the Editor.
The command specified by the field pointer is displayed on the entry line. If you press (Enter), the command is entered into the editor.

If command specified by the field pointer is a PAGE IMAGE or SCPI command, first select the `OUTPUT @Hp415x` secondary softkey. The entry line becomes `OUTPUT @Hp415x;"`*command*`"`, where *command* is command specified by field pointer. Then, press (Enter).

`OUTPUT @Hp415x;"`*command*`"` is entered into the editor.

Creating and Executing a Simple IBASIC Program

In this section, let's try to create and execute a simple program.

Before creating a program in the IBASIC editor, first change the screen display mode to IBASIC Status screen mode or All IBASIC screen mode by pressing IBASIC (Display) key as described in "To Switch Screen Mode". In following sections, the All IBASIC screen display mode is used.

Step 1. Editing

Select `EDIT` secondary softkey or type `EDIT`, then press (Enter).

```
10
```

To start the editor at a specific program line or label

Type `EDIT` *linenum* or `EDIT` *label*.

For example, if you type `EDIT 30`, the cursor appears at line 30. If you do not specify a line number or label, the cursor will appear at line 10.

The following program prints the numbers from 1 to 10. Type as follows:

```
10   FOR I=1 TO 10
20   PRINT I
30   NEXT I
40   END
```

Always Insert Mode

Editor is always in insert mode, and *cannot* be changed to overwrite mode. If you mistyped, use Backspace to move back a character, or move cursor using arrow key ◀, then use Delete to delete a character. Then type correct characters.

Program End

In IBASIC, `END` must be at end of main program. In above example, line 40 is the last line of the program.

Step 2. Exiting from Editor

Select `End edit` primary softkey to exit from the editor.

Step 3. Executing Program

To execute the program, press (Run) of the IBASIC key group, select `RUN` primary softkey, or type `RUN` and press (Enter). The following should be displayed on the screen:

```
1
2
:
9
10
```

If an error message appears

If an error message appears, you probably mistyped. The error message indicates the line number where the error occurs. You need to modify the line.

Modifying Program by using Editor Functions

In this section, you can learn the following editor functions:

1. Inserting lines
2. Deleting a line
3. Renumbering
4. Inserting characters
5. Recalling deleted line
6. Indenting
7. Changing line numbers

Step 1. Inserting Lines

Type `EDIT 20`, then press (Enter). Cursor appears at line 20.

```
10   FOR I=1 TO 10
20   PRINT I
30   NEXT I
40   END
```

Select `Insert line` or press (Insert) to insert a line above line 20.

```
10   FOR I=1 TO 10
11
20   PRINT I
30   NEXT I
40   END
```

Type as follows:

```
10   FOR I=1 TO 10
11   PRINT I^2
12
20   PRINT I
30   NEXT I
40   END
```

`I^2` means the second power of I. The above program increments `I` from 1 to 10, and displays second power of I and I for each step.

Select `End edit` to exit editor, then press (Run) to execute the program. The following is displayed:

```
1
1
4
2
9
3
⋮
81
9
100
10
```

Step 2. Deleting a Line

Type `EDIT 20` to start editor at line 20.

```
10   FOR I=1 TO 10
11   PRINT I^2
20   PRINT I
30   NEXT I
40   END
```

Then, select `Delete line` or press (Shift)+(Delete) to delete line 20. The result is as follows:

```
10   FOR I=1 TO 10
11   PRINT I^2
30   NEXT I
40   END
```

The above program increments `I` from 1 to 10, and displays the second power of I at each step.

If you exit editor and execute the program, the following is displayed:

```
1
4
9
⋮
81
100
```

Step 3. Renumbering

In above example, line numbers are not in equal increments. To change the line number increment to 10, select `Re-number` softkey. Line numbers will be changed as follows:

```
10   FOR I=1 TO 10
20   PRINT I^2
30   NEXT I
40   END
```

If you use the `Re-number` softkey, the renumbering is always as follows: first line is 10 and the increment is 10.

If you desire other numbering, you need to exit the editor, and use the **REN** command. For example, if you want first line number to be 100 and increment to be 20, type as follows:

```
REN 100, 20 Enter
```

Step 4. Inserting Characters

Type `EDIT 20`, then press Enter.

```
10   FOR I=1 TO 10
20   PRINT I^2
30   NEXT I
40   END
```

Move the cursor by using → key.

```
10   FOR I=1 TO 10
20   PRINT I^2
30   NEXT I
40   END
```

Then type `I,` as follows:

```
10   FOR I=1 TO 10
20   PRINT I, I^2
30   NEXT I
40   END
```

Above program increments `I` from 1 to 10, and displays I and the second power of I on one line at each step. Exit editor, then execute the program. The following is displayed:

```
1            1
2            4
3            9
:            :
9            81
10           100
```

Step 5. Recalling Deleted Line

To restore the most recently deleted line, press Recall front-panel key.

Step 6. Indenting

Move to desired line, then select `Indent` to indent the line. Indenting makes the program flow easier to understand.

```
10   FOR I=1 TO 10
20      PRINT I,I^2
30   NEXT I
40   END
```

Saving and Getting a Program

The created program can be saved to a diskette. So, you can get the saved program from the diskette, then execute it.

In this section, you can learn the following file operation tasks:

1. Saving a Program
2. Listing Contents of Diskette
3. Clearing a Program
4. Getting a Program

Step 1. Saving a Program

Insert a diskette into the built-in drive. Then, type `SAVE "`*filename*`"`, then press Enter. For this example, we will type `SAVE "PROG1"`.

Step 2. Listing Contents of Diskette

Type `CAT` to list contents of the diskette.

If you are using an MS-DOS format diskette, the display is similar to the following example:

```
DIRECTORY : \:INTERNAL,4
LABEL: HP4156
FORMAT: DOS
AVAILABLE SPACE :      5692
                 FILE     NUM   REC     MODIFIED
FILE NAME        TYPE    RECS   LEN    DATE    TIME PERMISSION
============ ====== ======= ===== ========= ===== ==========
PROG1        DOS          65     1 27-Jun-94 14:30  RW-RW-RW-
```

Filer

You can also check the contents of the diskette by using the filer (SYSTEM: FILER page) of the HP 4155A/4156A. But you *cannot* save and get the IBASIC programs by using the filer.

Step 3. Clearing a Program

To clear the program, enter the editor, then select `Scratch` softkey. Then, select `Yes` secondary softkey.

Existing program will be cleared, and following is displayed:

```
    10 COM @Hp415x
    20 ASSIGN @Hp415x TO 800
    30 !
  9990 END
```

`COM @Hp415x` and `ASSIGN @Hp415x TO 800` are used to control the HP 4155A/4156A. For details, please refer to "Step 1. Getting the Setup File and Making a Measurement" in Chapter 3, and refer to "Subprograms and COM Blocks".

Step 4. Getting a Program

Type `GET "PROG1"`, then press Enter. When the LED turns off, enter the editor if you want to display the program.

Summary of Softkeys and Keyboard Operations for Editor

Front-panel keys

Arrow keys (◀), (▶), (▲), (▼)	Move the cursor
Delete	Delete character
Recall	Recall most recently deleted line

External Keyboard

Esc	Exit editor
F1 to F8	Primary softkeys
Shift + F1 to F7	Secondary softkeys
F9	Toggle screen mode
F11	Clear to end
Insert	Insert line
Delete	Delete character at cursor
Shift + Delete	Delete line
Home	Beginning of line
End	End of line
Page Up or Page Down	Scroll pages

Primary Softkeys

`Back space`	Delete character before cursor
`Insert line`	Insert line
`Delete line`	Delete line
`Re-number`	Renumber the lines
`Indent`	Indent the line
`Scratch`	Clear program
`End edit`	Exit editor

Other Basic Features of IBASIC

This section describes the following:

- Branching/Repeating
- Subprograms and COM Blocks

Branching/Repeating

Table 1-1. Branch and Repeat Keywords of IBASIC

IBASIC Keyword	Function
`FOR, NEXT`	Repeat specified number of times.
`IF THEN, ELSE, END IF`	Branch.
`WHILE, END WHILE`	Repeat until specified condition is *false*.
`REPEAT, UNTIL`	Repeat until specified condition is *true*.

Following program tests 1000 devices, and judges them pass or fail.

```
1   COM Data(1:1000)
2   DIM Id(1:1000)
10  EXECUTE ("GETSETUP 'SAMPL.MES'")
20  FOR I=1 TO 1000
30     EXECUTE ("SINGLE")
40     EXECUTE ("READDATAVAR 'Id'")
50     IF Id(I)<1E-6 THEN
60        PRINT "FAIL LOWER"
70        Data(I)=1
80     ELSE
90        IF Id(I)>1E-4 THEN
100          PRINT "FAIL HIGHER"
110          Data(I)=2
120       ELSE
130          PRINT "PASS"
140          Data(I)=0
150       END IF
160    END IF
170 NEXT I
180 CALL Save_data
190 END
200 !
210 SUB Save_data
220 COM Data(*)
230 CREATE "data_file",1
240 ASSIGN @File TO "data_file";FORMAT ON
250 OUTPUT @File;Data(*)
260 ASSIGN @File TO *
270 SUBEND
```

Subprograms and COM Blocks

One of the most powerful constructs available is the subprogram. A subprogram has its own "context" or state that is distinct from the main program and all other subprograms. There are several benefits of subprograms.

- The subprogram allows you to take advantage of the "top-down design" method of programming.
- You can remove all subtasks from the overall logic flow of the main program.
- You can debug the program by testing each subprogram independently.
- The subprograms can be used to reduce the overall size of the program.
- Libraries of commonly used subprograms can be assembled for widespread use.

Refer to the example program in the previous section. Line 180 calls a subprogram to store data into a DOS file.

```
    :
160    END IF
170 NEXT I
180 CALL Save_data
190 END
200 !
210 SUB Save_data
220 COM Data(*)
230 CREATE "data_file",1
240 ASSIGN @File TO "data_file";FORMAT ON
250 OUTPUT @File;Data(*)
260 ASSIGN @File TO *
270 SUBEND
```

COM Blocks.
COM blocks are used by the subprogram to communicate with the main program or with other subprograms.

If you create subprograms and want to use common variables between main program and subprograms, you should use COM blocks.

Refer to the above example.

In the main program, line 1 declares that the `Data` array will be a COM array. Then, the main program assigns values to this array. Line 220 specifies that the subprogram `Save_data` will also use the `Data` array. So, `Data` array of main program can be operated on in the `Save_data` subprogram.

IBASIC Basic Operation Tasks

This section describes the following basic operations to use HP Instrument BASIC.

- Executing HP Instrument BASIC commands
- Executing program
- Listing files
- Retrieving program
- Saving program

To Execute HP Instrument BASIC Commands

1. Confirm your HP 4155A/4156A is in the following status:
   - a program is not executing.
   - another command is not executing.
   - Editor is not running.
   - the screen is "All IBASIC" screen or "IBASIC Status" screen. For "All Instrument" screen, Run and Pause front-panel keys and Ctrl+U (Run) and Ctrl+P (Pause) on external keyboard are available.
2. Type in commands by using front-panel keys in the ENTRY key group or external keyboard.
3. Press Enter front-panel key or Enter key on external keyboard.

To Execute Program

Front Panel

1. Press Run front-panel key in the IBASIC key group.

External Keyboard

1. Press Ctrl+U on external keyboard.

To List Files

1. Confirm your HP 4155A/4156A is in the following status:
   - the screen is "All IBASIC" screen.
   - a program is not executing.
   - another command is not executing.
   - Editor is not running.
2. Insert a 3.5 inch diskette (that contains the files you want to list) into the built-in diskette drive.
3. Select `CAT` secondary softkey, then press Enter front-panel key.

The file names on diskette are listed on the screen.

To Retrieve Program

1. Confirm your HP 4155A/4156A is in the following status:
   - the screen is "All IBASIC" screen or "IBASIC Status" screen.
   - a program is not executing.
   - another command is not executing.
   - Editor is not running.
2. Insert the 3.5 inch diskette (that contains the program you want to retrieve) into the built-in diskette drive.
3. Select `GET ""` secondary softkey.
4. Type in file name to be retrieved. Typed name is inserted after first ".
5. Press Enter front-panel key, or Enter key on external keyboard.

External disk drive

An external disk drive cannot be connected to HP 4155A/4156A. For using a disk drive connected to external controller, see "Control from External Computer".

To Save Program

1. Confirm your HP 4155A/4156A is in the following status:
   - the screen is "All IBASIC" screen or "IBASIC Status" screen.
   - a program is not executing.
   - another command is not executing.
   - Editor is not running.
2. Insert a 3.5 inch diskette into the built-in diskette drive.
3. Select `SAVE ""` secondary softkey.
4. Type in name of file to which you want to save program.

   If the file already exists on the diskette, `SAVE` cannot be used. If you want to overwrite an existing file, select `RE-SAVE` secondary softkey instead of `SAVE` secondary softkey in the previous step.
5. Press Enter front-panel key or Enter key on the external keyboard.

External disk drive

An external disk drive cannot be connected to HP 4155A/4156A. For using a disk drive connected to external controller, see "Control from External Computer".

IBASIC Editor Tasks

This section describes the following tasks to use built-in editor of HP Instrument BASIC.

- Starting the editor
- Quitting the editor
- Moving the cursor
- Inserting characters
- Deleting character
- Inserting line
- Deleting line
- Scrolling pages
- Recalling most recently deleted line

To Start the Editor

1. Confirm the screen is "All IBASIC" screen or "IBASIC Status" screen.
2. Select `EDIT` secondary softkey.
3. Press Enter front-panel key or Enter key on the external keyboard.

   If you want to start the editor to edit a specific program line, type in the line number or label of the program line, then press Enter front-panel key. The editor starts, and cursor is displayed on specified line.
4. If a program is loaded into the HP 4155A/4156A, the program is displayed.

   If no program is loaded, `10` is automatically displayed and rest of screen is empty.

If you start the editor from the "IBASIC Status" screen, the screen switches to "All IBASIC" screen, and the editor starts.

To Quit the Editor

1. Select `End edit` primary softkey.

If you started the editor from the "IBASIC Status" screen, the screen returns from "All IBASIC" screen to the "IBASIC Status" screen after you quit the editor.

To Move the Cursor

To move the cursor, use the following keys.

Direction	Front-panel	Keyboard
Up	⇑ of MARKER/CURSOR group Rotate rotary knob counter-clockwise	▲
Down	⇓ of MARKER/CURSOR group Rotate rotary knob clockwise	▼
Right	⇒ of ENTRY group ⇒ of MARKER/CURSOR group	▶ Ctrl + F
Left	⇐ of ENTRY group ⇐ of MARKER/CURSOR group	◀ Ctrl + B
Beginning of Line	FAST + ⇐ of MARKER/CURSOR group	Home Shift + ◀
End of Line	FAST + ⇒ of MARKER/CURSOR group	End Shift + ▶

To Insert Characters

1. Move the cursor to character you want to insert before.
2. Characters you type will be automatically inserted.
3. After you insert characters, you must select the Enter key to enter the line with inserted characters into the program.

Editor is always in insert mode, and cannot be changed to overwrite mode.

To Delete Character

1. Move the cursor to character you want to delete.
2. Press key according to the following table:

Front-panel	Keyboard
Delete of ENTRY group	Delete

3. After you delete characters, you must select the Enter key to enter the line with deleted characters into the program.

To Insert Line

1. Move the cursor to the line that you want to insert a new line before.
2. Press key or softkey according to following table:

Front-panel	Keyboard
`Insert line` primary softkey	Shift + Insert
	Alt + I

3. After you type in a new line, you must select the Enter key to enter the new line into the program.

To Delete Line

1. Move the cursor to line you want to delete.
2. Press key or softkey according to the following table:

Front-panel	Keyboard
`Delete line` primary softkey	Shift + Delete
	Alt + D

To Scroll Pages

To scroll the editor by one-half screen, use the following keys:

Direction	Front-panel	Keyboard
Up	n.a.	Page Up
Down	n.a.	Page Down

To Recall Most Recently Deleted Line

To display the line most recently deleted line, use the following keys.

Front-panel	Keyboard
Recall↓	Shift + Page Up

If you want to enter the line into the program, you must select the Enter key.

Control from External Computer

This section describes how to control the HP 4155A/4156A from a program that is running on an external computer:

- Downloading a program to HP 4155A/4156A.
- Uploading a program from HP 4155A/4156A.
- Controlling execution of a HP 4155A/4156A program.

Before executing a program on external computer to control HP 4155A/4156A, do as follows:

1. Set the "`HP 4155A/4156A is`" field on the SYSTEM: MISCELLANEOUS page to NOT SYSTEM CONTROLLER.
2. Connect an HP-IB cable from the external computer to the HP-IB connector on rear panel of HP 4155A/4156A.

To Download a Program to HP 4155A/4156A

To download a program from the external computer to HP 4155A/4156A, you need to use the :PROGram[:SELected]:DEFine command.

Example

The following is an example of an HP BASIC program (running on external computer) that reads an HP Instrument BASIC program file (from a disk drive connected to external computer) and downloads it to HP 4156A.

```
10    OPTION BASE 1
20    !
30    DIM Line$[1024]
40    !
50    ASSIGN @Hp4156 TO 717
60    !
70    OUTPUT @Hp4156;"PROG:DEL:ALL"  ! Clears program in HP 4156A
80    File_name$="prog"
90    ASSIGN @File TO File_name$     ! Opens file and assigns data path
100   OUTPUT @Hp4156;"PROG:DEF #0"   ! Sends header to 4156A
110   ON ERROR GOTO Done
120   LOOP
130     Line$=""
140     ENTER @File;Line$            ! Reads one program line
150     OUTPUT @Hp4156;Line$         ! Downloads line to HP 4156A
160   END LOOP
170 Done:   !
180   OFF ERROR
190   OUTPUT @Hp4156;Line$
200   OUTPUT @Hp4156;" " END
210   ASSIGN @File TO *
220   END
```

Line Number	Description
50	assigns I/O path to control HP 4155A/4156A.
70	deletes existing HP Instrument BASIC program in HP 4155A/4156A.
80	name of file (in disk drive of external computer) that contains desired HP Instrument BASIC program
90	opens file and assigns data path
100	#0 indicates that an indefinite length of parameters (program lines) will be downloaded
110 to 160	reads program lines from the file and downloads them until EOF.
210	closes file.

To Upload a Program from HP 4155A/4156A

To upload a program from the HP 4155A/4156A to external computer, you need to use the :PROGram[:SELected]:DEFine? command.

Example

The following is an example of an HP BASIC program (running on external computer) that uploads an HP Instrument BASIC program from HP 4156A and stores the program on a disk drive that is connected to external computer.

```
10    OPTION BASE 1
20    !
30    DIM Num_dig$[2]
40    INTEGER Byte
50    !
60    ASSIGN @Hp4156 TO 717
70    !
80    OUTPUT @Hp4156;":PROG:DEF?"
90    ENTER @Hp4156 USING "%,2A";Num_dig$
100   PRINT Num_dig$
110   !
120   Byte=VAL(Num_dig$[2])
130   !
140   ALLOCATE Data_byt$[Byte]
150   !
160   FOR I=1 TO Byte
170     ENTER @Hp4156 USING "#,A";Data_byt$[I;1] ! Enter length of program
180   NEXT I
190   !
200   D=VAL(Data_byt$)
210   PRINT D
220   ALLOCATE Prog$[D]
230   PRINTER IS CRT;WIDTH D
240   ENTER @Hp4156 USING "-K";Prog$          ! Enter the program into Prog$
250   PRINT Prog$
260   ENTER @Hp4156;B$
270   PRINT B$
280   !
290   CREATE "prog",1
300   ASSIGN @File TO "prog";FORMAT ON
310   OUTPUT @File;Prog$
320   ASSIGN @File TO *
330   !
340   END
```

Line Number	Description
60	Assigns I/O path to control HP 4155A/4156A.
80	Sends :PROGram[:SELected]:DEFine? query command.
90	Reads first two characters of response. These two bytes indicate how many bytes are used to specify length of program.
160 to 180	Reads the bytes that specify length of program.
200	Calculates length of program.
220	Allocates string variables for program.
240	Reads program.
260	Reads terminator.
290	Creates file "prog"
300	Assigns I/O path to "prog"
310	Stores program into "prog" file.
320	Closes file.

To Control State of HP 4155A/4156A Program

:PROGram[:SELected]:STATe command from external computer can control HP Instrument BASIC program in the HP 4155A/4156A as follows:

Example

- To run the program:

```
OUTPUT @Hp4155;":PROGram[:SELected]:STATe RUN"
```

- To continue the program:

```
OUTPUT @Hp4155;":PROGram[:SELected]:STATe CONT"
```

- To stop the program:

```
OUTPUT @Hp4155;":PROGram[:SELected]:STATe STOP"
```

- To pause the program:

```
OUTPUT @Hp4155;":PROGram[:SELected]:STATe PAUSE"
```

Creating ASP-like IBASIC Programs

In the IBASIC editor of the HP 4155A/4156A, there are several typing aid softkeys that allow you to easily create a program.

When you press the softkey, the corresponding IBASIC command is entered into the program, so you do not have to type it, but you may need to type in some parameters.

For HP 4145A/B users, this environment is very familiar because it is similar to the Auto Sequence Program (ASP) programming environment of the HP 4145A/B. For most of the HP 4145 ASP commands, the IBASIC editor has a softkey to enter a corresponding IBASIC command.

These programs can run in IBASIC *only*, not on an external computer.

Step 1. Creating Programs by using the Typing Aid Softkeys

In the IBASIC editor, you can easily create programs that perform the same operations as a desired HP 4145 ASP program by using the typing aid softkeys. These are secondary softkeys. To display more softkeys, select `More` softkey.

For the ASP program shown below, let's create the corresponding IBASIC program:

```
!  ASP Program:        Corresponding IBASIC Program:
!
1  GET P ICBVBE      10  EXECUTE ("GETSETUP 'ICBVBE.PRO'")
2  SINGLE            20  EXECUTE ("SINGLE")
3  SAVE D BV1        30  EXECUTE ("SAVEDATA 'BV1.DAT'")
                     40  END
```

1. Select the `GET SETUP` secondary softkey. The following appears:

```
10    EXECUTE ("GETSETUP "_)
```

You need to specify a filename in this command.
At bottom of screen, `enter fileName[,msus]` is displayed, where `msus` means the mass storage unit specifier. You can specify `,DISK` or `,MEMORY`. Default is `,DISK`.

2. Type a setup file name.

```
10    EXECUTE ("GETSETUP 'ICBVBE.PRO'")
```

File name must be in single quotations ('). Then press Enter.

3. Select `SINGLE` secondary softkey.

```
10    EXECUTE ("GETSETUP 'ICBVBE.PRO'")
20    EXECUTE ("SINGLE")
30
```

4. Select `SAVEDATA` secondary softkey.

```
10    EXECUTE ("GETSETUP 'ICBVBE.PRO'")
20    EXECUTE ("SINGLE")
30    EXECUTE ("SAVEDATA ")
```

5. Specify file name to which you want to save the measurement setup and result data.

```
10    EXECUTE ("GETSETUP 'ICBVBE.PRO'")
20    EXECUTE ("SINGLE")
30    EXECUTE ("SAVEDATA 'BV1.DAT'")
40    END
```

Finally, type `END` as above.

Setup File

In `EXECUTE ("GETSETUP ")`, you can specify a `.PRO` or `.MES` file:

- `.PRO` files are setup files created by the HP 4145B. The HP 4155A/4156A can read `.PRO` files.
- `.MES` files are setup files created by the HP 4155A/4156A.

In `EXECUTE ("SAVEDATA ")`, you specify a `.DAT` file, which is a file for storing the setup and measurement result data.

Step 2. Executing the Program

To execute the program, exit editor, then press Run.

The HP 4155A/4156A gets the setup file from the diskette, performs measurement, then saves setup and results to specified file on the diskette. However, in All IBASIC mode, no graphics results are displayed. To display results graphically, the display mode must be All Instrument mode or IBASIC Status mode.

To execute the program and display the results graphically, change the display mode to All Instrument or IBASIC Status mode, then press Run.

Step 3. Creating a Longer Program

In the program below, the left side is an ASP program example from the HP 4145B manual.

The right side shows a program that was created by using the typing aid softkeys to enter the ASP-like commands (of the HP 4155A/4156A) that correspond to the original ASP commands. These softkeys allow you to easily create a program that runs on the HP 4155A/4156A and performs the same operations as the original ASP program.

```
 1   GET P  ICBVBE                 10   EXECUTE ("GETSETUP 'ICBVBE.PRO'")
 2   SINGLE                        20   EXECUTE ("SINGLE")
 3   WAIT   3                      30   WAIT 3
 4   GET P  HFE1                   40   EXECUTE ("GETSETUP 'HFE1.PRO'")
 5   SINGLE                        50   EXECUTE ("SINGLE")
 6   WAIT   3                      60   WAIT 3
 7   GET P  VCESAT                 70   EXECUTE ("GETSETUP 'VCESAT.PRO'")
 8   SINGLE                        80   EXECUTE ("SINGLE")
 9   WAIT   3                      90   WAIT 3
10   GET P  COLR                  100   EXECUTE ("GETSETUP 'COLR.PRO'")
11   SINGLE                       110   EXECUTE ("SINGLE")
12   WAIT   3                     120   WAIT 3
13   PAUSE                        130   PAUSE
14   GET P  NPN1                  140   EXECUTE ("GETSETUP 'NPN1.PRO'")
```

```
15   SINGLE                        150   EXECUTE ("SINGLE")
16   PAUSE                         160   PAUSE
17   PLOT   100,100,7000,7000      170   EXECUTE ("PRINTPLOT")
18   GET P   BV                    180   EXECUTE ("GETSETUP 'BV.PRO'")
19   SINGLE                        190   EXECUTE ("SINGLE")
20   PLOT   100,100,7000,7000      200   EXECUTE ("PRINTPLOT")
21   PAUSE                         210   PAUSE
22   SINGLE                        220   EXECUTE ("SINGLE")
23   CPLOT  100,100,7000,7000      230   EXECUTE ("CURVEPLOT")
                                   240   END
```

Print/Plot

`EXECUTE ("PRINTPLOT")` prints/plots the information of the present instrument page, not the IBASIC screen. If present page is GRAPH/LIST: GRAPHICS page, the graph is printed/plotted.

`EXECUTE ("CURVEPLOT")` changes to the GRAPH/LIST: GRAPHICS page, then prints/plots the graph.

You need to set the desired settings on the SYSTEM: PRINT/PLOT SETUP page before `EXECUTE ("PRINTPLOT")` and `EXECUTE ("CURVEPLOT")` are performed.

Programming Tips

This section describes features and tips of IBASIC programs in relation to ASP programs. Some examples use an example measurement setup file named "VTH.MES". Before executing these examples, you need to save setup data to a file named "VTH.MES" on the diskette. For an example setup, see "Example Application Setup for Vth Measurement" in Chapter 3.

File Name Variables.
You can specify a string variable for the file name in `SAVEDATA` as follows:

```
Filename$="DATA1.DAT"
EXECUTE ("SAVEDATA Filename$")
```

This feature allows you to create a more simple program as follows.

Example ASP Program

Following ASP program gets a setup file, makes measurements, and saves results to following files: `VTH1`, `VTH2`, . . . `VTH10`. Program is 21 lines.

```
1   GET P  VTH
2   SINGLE
3   SAVE D VTH1
4   SINGLE
5   SAVE D VTH2
6   SINGLE
7   SAVE D VTH3
  :
21  SAVE D VTH10
```

Corresponding IBASIC Program.

The following HP Instrument BASIC (IBASIC) program does the same operation as the above ASP program. The program is simplified by using a filename variable `Filename$` and the `FOR NEXT` keyword.

```
10    EXECUTE ("GETSETUP 'VTH.PRO'")
20    FOR I=1 TO 10
30       EXECUTE ("SINGLE")
40       Filename$="VTH"&VAL$(I)&".DAT"
50       EXECUTE ("SAVEDATA Filename$")
60    NEXT I
```

In line 40, the `Filename$` is defined. For example, `Filename$="VTH1.DAT"` when `I=1`. So, the 21-line ASP program can be converted to a 6-line IBASIC program.

Reading HP 4155/56 Data to IBASIC Variables

You can transfer read-out function values or data variable values (output data, measurement data, and user function values) from the HP 4155/56 to HP Instrument BASIC (IBASIC) variables.

Transferring Multiple Data.
You can transfer multiple data (such as sweep measurement data) to an array variable of IBASIC by using `EXECUTE ("READDATAVAR ")` as follows:

```
EXECUTE ("READDATAVAR 'ID',Id_data")
```

The above example transfers the drain current data `ID` of a sweep measurement to the array variable previously defined as `Id_data`.

Following example program gets `VTH.MES` setup file, performs measurement, then transfers `ID` data to an array. In this example, the array `Id_data` is defined in line 10, and it has elements 1 to 51.

```
10   DIM Id_data(1:51)
20   EXECUTE ("GETSETUP 'VTH.MES'")
30   EXECUTE ("SINGLE")
40   EXECUTE ("READDATAVAR 'ID',Id_data")
50   FOR I=1 TO 51
60     PRINT "Id(";I;")=";Id_data(I);"A"
70   NEXT I
80   END
```

Result with example measurement data is as follows:

```
Id( 1)= 0.00031 A
Id( 2)= 0.00282 A
Id( 3)= 0.00514 A
Id( 4)= 0.01017 A
      :
Id( 51)= 0.08274 A
```

Transferring a Single Data.
In the following example, a single data is to transferred to a variable. For example, `VTH` is a single data point calculated by a user function that was defined by the user.

```
EXECUTE ("READDATAVAR 'VTH',Vthdata")
```

In following example, `EXECUTE ("READDATAVAR")` is used to transfer the `VTH` value to the IBASIC variable `Vthdata`. And for example, `VTH.MES` is a setup file that includes auto analysis setup to extract a threshold voltage `VTH`.

```
10   EXECUTE ("GETSETUP 'VTH.MES'")
20   EXECUTE ("SINGLE")
30   EXECUTE ("READDATAVAR 'VTH',Vthdata")
40   PRINT "Vthdata =";Vthdata;"V"
50   END
```

Result will be for example:

```
Vthdata = 1.2345 V
```

You can also specify a read out function as the item to be transferred:

`EXECUTE ("READDATAVAR '@MX',Vthdata")`

`@MX` is the read out function that reads X-axis value of point where marker is located.

Auto Scaling

Auto scaling can be done by using the following:

`EXECUTE ("AUTOSCALE")`.

In the following example, the image dumps will be scaled for best fit to the printer or plotter even if the measurement results vary greatly.

```
10   EXECUTE ("GETSETUP 'VTH.MES'")
11   FOR I=1 TO 100
20      EXECUTE ("SINGLE")
30      EXECUTE ("AUTOSCALE")
40      EXECUTE ("PRINTPLOT")
41   NEXT I
50   END
```

HP 4145 ASP and HP 4155A/4156A Corresponding Keywords

Following shows HP 4145A/B's ASP keywords and corresponding HP 4155A/4156A keywords. In IBASIC editor, there are typing aid softkeys to help you quickly enter the related HP 4155A/4156A keyword, which must be used in the `EXECUTE( )` directive:

Table 1-2. Corresponding HP 4145 ASP and HP 4155A/4156A Keywords

4145A/B	4155A/4156A	Function	Remark
GET P	GETSETUP	Gets setup `.MES` or `.PRO` file	
SINGLE	SINGLE	Initiates single measurement	
SAVE D	SAVEDATA	Saves data to `.DAT` file	
PLOT	PRINTPLOT	Prints/plots present instrument page.	
CPLOT	CURVEPLOT	Prints/plots measurement graph.	
PRINT	PRINTPLOT	Prints/plots present instrument page.	
PAUSE			Use BASIC keyword PAUSE
WAIT			Use BASIC keyword WAIT
PAGE			Set in the Print/Plot setup
	STANDBY	Sets Standby status on or off	
	STRESS	Initiates stress force	
	AUTOSCALE	Scales dump for best fit.	
	READDATAVAR	Gets data variable from 4155A/56A	
	DEFUSERVAR	Defines user variable	

For WAIT and PAUSE of HP 4145's ASP, there are no related typing aid softkeys. You type in the IBASIC keywords (WAIT and PAUSE).

For more information about IBASIC Keywords, use help functions described in the next chapter or refer to the *HP Instrument BASIC Users Handbook*.

2

Reference:
HP Instrument BASIC

Reference: HP Instrument BASIC

This chapter provides following reference information for HP Instrument BASIC:

- IBASIC Screen
- Front-panel keys
- External Keyboard
- Softkeys for Operating IBASIC
- Softkeys for IBASIC Editor
- HP 4155A/4156A Specific IBASIC Commands
- Available I/O Resources
- ASP-like Commands

IBASIC Screen

HP 4155A/4156A provides the following three screen modes for operating IBASIC.

"All IBASIC" screen	Entire screen including softkeys is used for IBASIC, so no instrument page is displayed. You can execute programs, but no instrument page appears in this mode.
"IBASIC Status" screen	Softkeys and bottom two lines are used for IBASIC. Rest of screen is for instrument page. In this mode, you can start the IBASIC editor. The displayed softkeys are for IBASIC operation. You can execute IBASIC commands interactively. Characters you type are displayed at the bottom of the screen.
"All Instrument" screen	.This is regular instrument screen and the default display mode at power on. Entire screen is for instrument page, and all softkeys are for interactive use of instrument. In this mode, you *cannot* use the IBASIC editor. Only the front-panel keys of IBASIC key group and Ctrl+U (Run) and Ctrl+P (Pause) on external keyboard are available to execute or pause program for IBASIC from this screen mode.

Display front-panel key or Ctrl+G (or F9) on external keyboard are used to toggle the screen display mode as shown in the following figure:



All IBASIC Screen



PG02001 120x80

Figure 2-1. All IBASIC Screen

For the "All IBASIC" screen, the entire screen including softkeys is used for IBASIC. The following describes each part of this screen:

IBASIC Output Area

CRT output commands of IBASIC (such as `PRINT` and `OUTPUT 1;`) display characters in this area. This area has 24 lines and 60 columns.

Command Entry Line

IBASIC command you type is displayed on this line. The length of this line is 60 characters.

System Message and Display Line

For displaying IBASIC error messages and other system messages, and `DISP` and `INPUT` commands of IBASIC.

Blue Key & Green Key Shift Status

This field displays the shift status of ENTRY front-panel keys. The shift status is controlled by using the blue and green front-panel keys:

The following statuses are displayed:

- Non-shift status: `B`, `b`, or `G` is not displayed. You can enter numeric values.
- Uppercase shift status: `B` is displayed, `G` is not displayed. You can enter uppercase alphabet characters.
- Lowercase shift status: `b` is displayed. `G` is not displayed. You can enter lowercase alphabet characters.
- Non-alphanumeric status: `G` is displayed. You can enter one non-alphanumeric character. So, you must press green key before entering each alphanumeric character.

Basically, you can change between these states as follows:

- To toggle between non-shift and shift status: press blue key,
- To toggle between uppercase and lowercase shift status: press green key, then blue key.
- To enter one non-alphanumeric character: press green key, then character.

Execution Status

This field displays the execution status of IBASIC:

`Idle`	IBASIC program is stopped. IBASIC commands can be executed.
`Run`	IBASIC program or command is being executed.
`Pause`	IBASIC program is paused.
`Input?`	IBASIC program is waiting for input from front-panel keys or external keyboard.
`Edit`	IBASIC editor is running.

IBASIC Status Screen



PG02002 120x80

Figure 2-2. IBASIC Status Screen

For the "IBASIC Status" screen, the two bottom lines are used to display the status of IBASIC. These two lines are the same as in "All IBASIC" screen. Refer to "All IBASIC Screen". Also, the softkeys are for IBASIC.

The other part of the screen is the normal 4155A/4156A screen. This is useful if you want to view a graph of the measurement results while executing IBASIC program.

Keys for IBASIC

This section provides information about the following keys for IBASIC:

- Front-panel Keys
- Primary Softkeys
- Secondary Softkeys
- External Keyboard Keys

Front-panel Keys for IBASIC

PAGE CONTROL key group

- In "IBASIC Status" screen:

  Changes to "All Instrument" screen and displays the specified page.

- In "All IBASIC" screen:

  All front-panel keys in this group are *ignored*.

MARKER/CURSOR key group

- In "IBASIC Status" screen:

  When you operate MARKER/CURSOR front-panel keys, the screen changes to "All Instrument" screen and function of operated key is executed.

- In all IBASIC screen:

  Rotary Knob
  - When the editor is running, the rotary knob moves the cursor vertically in the edit area.
  - When the editor is not running, the rotary knob scrolls the IBASIC output area.

  ↑ ↓
  - When the editor is running, these keys move the cursor vertically.
  - When the editor is not running, these keys scroll the IBASIC output area.

  ←, →
  Moves the cursor horizontally on the IBASIC Editor or Command Entry line.

  If you hold down the Fast key, the arrow keys move the cursor faster.

MEASUREMENT key group

- In "IBASIC Status" screen:

Single, Repeat, Append	Changes the displayed page to GRAPH/LIST: GRAPHICS or GRAPH/LIST: LIST page and executes the measurement.
Standby	Toggles the operation state of the standby channels between the standby state and the idle state.
Stop	Stops the measurement or stress forcing.
Stress	Changes the displayed page to the STRESS: STRESS FORCE page and starts to force stress.
Short, Medium, Long	Changes the measurement integration time.

- In "All IBASIC" screen:

Standby	Toggles the operation state of the standby channels between the standby state and the idle state.
Stop	Stops the measurement or stress forcing.

Other front-panel keys in this group are ignored.

IBASIC key group

The following front-panel keys of IBASIC key group are available to control IBASIC in any display mode.

Run	Executes IBASIC program that is loaded into internal memory of HP 4155A/4156A.
Pause	Pauses program execution until CONT command is executed or `Continue` primary softkey is pressed. If the program is modified while paused, RUN command must be used to restart program execution.
Display	Toggles the display mode in the following sequence.



PG02003 120x20

ENTRY key group

(Recall ↓)
- When the editor is running, this key displays the last deleted line. To enter this displayed line as part of the program, press (Enter) front-panel key.
- When the editor is not running, this key cycles through the 10 commands that were most recently entered on the Command Entry line.

(Recall ↑)
- When the editor is running, this key is same as (Recall ↓).
- When the editor is not running, this key is same as (Recall ↓), but cycles through commands in opposite order.

(Save), (Get) These keys are ignored.

Other front-panel keys in ENTRY group are available to enter characters on the Command Entry line or Editor. For the usage of the blue and green front-panel keys to enter characters, see "All IBASIC Screen".

Other Keys

(Help) Displays information about IBASIC. And can be used to select and enter SCPI and IBASIC commands into Editor or Command Entry line.

(Plot/Print) If present screen is "All IBASIC", dumps "All IBASIC" screen image to the printer or plotter.

If present screen is "IBASIC Status", prints/plots instrument part of screen.

Primary Softkeys in Idle, Pause, Run, or Input? execution status

This section describes the primary softkeys that are displayed during the `Idle`, `Pause`, `Run`, or `Input?` execution status.

Refer to "Primary Softkeys in Edit execution status" for primary softkeys that are displayed when the editor is running.

`Step`

1. Executes the paused program line of paused program or the first program line of stopped (idle) program.
2. Displays next program line on system message line of the screen.
3. Pauses program again.

`Conti-nue`

Starts execution of paused program from paused program line.

`RUN`

Starts program execution immediately from first program line.

`Pause`

Pauses program execution immediately. And displays line at which execution was paused.

`Stop`

Stops program execution after current line executes.

`Clear I/O`

Stops I/O operation of program.

`Reset`

Stops program execution immediately.

Secondary Softkeys in Idle or Pause execution status

This section describes the secondary softkeys that are displayed during the `Idle` or `Pause` execution status

For the secondary softkeys that are displayed during the `Run` or `Input?` status, refer to "Secondary Softkeys in Run or Input? execution status".

For the secondary softkeys that are displayed when the editor is running, refer to "Secondary Softkeys in Edit execution status"

`CAT`

Clears the Command Entry line, and types in `CAT` .

To list file names on the disk, press Enter.

`SAVE ""`

1. Clears the Command Entry line.
2. Types in `SAVE "  "`.
3. Positions the cursor after first ".

To save program to diskette, type name of file to which you want to save program, then press Enter.

If file already exists on diskette, program will not be saved.

`RE-SAVE ""`

1. Clears the Command Entry line.
2. Types in `RE-SAVE "  "`.
3. Positions the cursor after first ".

To save program to diskette, type name of file to which you want to save program, then press Enter.

If file already exists on diskette, file will be overwritten, so previous data in file is lost.

`GET ""`

1. Clears the Command Entry line.
2. Types in `GET "  "`.
3. Positions the cursor after first ".

To get a program from diskette, type name of file to get, then press Enter.

`PURGE "
"`

1. Clears the Command Entry line.
2. Types in `PURGE "  "`.
3. Positions the cursor after first ".

To delete a file from diskette, type in the file name to be deleted, then press Enter.

`EDIT`

Clears the Command Entry line and types in `EDIT`. To start the editor, press Enter front-panel key.

`REN
 umber`

Clears the Command Entry line and types in `REN`.

To re-number lines of a program, type in appropriate parameters, then press Enter. For more details about `REN` command, refer to the *HP Instrument BASIC User's Handbook*.

Secondary Softkeys in Run or Input? execution status

When the execution status is `Run` or `Input?`, user-defined softkeys, which are defined by using `ON KEY` command in the program, are displayed in the secondary softkey area.

Primary Softkeys in Edit execution status

This section describes the primary softkeys that are displayed when the IBASIC editor is running (`Edit` execution status is displayed).

`Back space` — Deletes the character before the cursor.

`Insert line` — Inserts a line between the cursor line and the previous line.

`Delete line` — Deletes the cursor line.

`Re-number` — Changes the program line numbers so that first line is 10 and line number increment is 10.

`Indent` — Indents so that all program lines begin at the same position.

`Scratch` — Clears the program and all variables not in COM. Before clearing, `YES` and `NO` secondary softkeys are displayed for confirmation.

`End edit` — Exits the editor.

Secondary Softkeys in Edit execution status

This section describes the secondary softkeys that are displayed when the IBASIC editor is running (`Edit` execution status is displayed).

These softkeys help you enter program commands. For commands that require you to type in some parameters, these softkeys display the command. You must enter the parameters, then you must press Enter key to enter the command into the program. For commands that do not have parameters, the commands are entered directly into the program. Commands are entered at the cursor line.

For the EXECUTE command, refer to "EXECUTE" for details.

For secondary softkeys that are displayed during `Idle` or `Pause` execution status, refer to "Secondary Softkeys in Idle or Pause execution status".

For secondary softkeys that are displayed during `Run` or `Input?` execution status, refer to "Secondary Softkeys in Run or Input? execution status".

In `Edit` execution status, there are three pages of secondary softkeys. To move to next page, press `MORE` secondary softkey.

`GET SETUP`

1. Displays the following program line for loading a setup file:

   ```
   EXECUTE("GETSETUP ")
   ```

2. Positions cursor at second double quotes. You enter the file name to be loaded, then select Enter key.

`SINGLE`

Enters the following program line for triggering a single measurement:

```
EXECUTE("SINGLE")
```

`STANDBY`

1. Displays the following program line for changing the operation state of the standby channels:

   ```
   EXECUTE("STANDBY ")
   ```

2. Positions the cursor at the second double quote. You enter `ON` or `OFF`, then select Enter key.

`STRESS`

Enters the following program line for triggering stress force:

```
EXECUTE("STRESS")
```

`AUTO SCALE`

Enters the following program line for autoscaling:

```
EXECUTE("AUTOSCALE")
```

`SAVE DATA`

1. Displays the following program line for saving measurement data to a file:

   ```
   EXECUTE("SAVEDATA ")
   ```

2. Positions the cursor at the second double quote. You enter file name to which you want to save measurement data, then select Enter key.

`READ DATA VARIABL`

1. Displays the following program line for reading the values of an HP 4155A/4156A data variable, then storing the values into an IBASIC program variable:

   ```
   EXECUTE("READDATAVAR ,")
   ```

2. Positions the cursor at the comma. You enter names of HP 4155A/4156A data variable and IBASIC program variable, then select Enter key.

`DEFINE USER VARIABL`

1. Displays the following program line for defining a user variable:

   ```
   EXECUTE("DEFUSERVAR ,,,")
   ```

2. Positions the cursor at the first comma. You enter the user variable name, number of data, name of IBASIC program variable that contains desired data, and user variable unit, then select Enter key.

`PRINT/ PLOT`

Enters following program line for printing/plotting the instrument window:

```
EXECUTE("PRINTPLOT")
```

`CURVE PLOT`

Enters following program line for printing/plotting a graphics plot of measurement results:

```
EXECUTE("CURVEPLOT")
```

`OUTPUT @Hp415x`

1. Displays the following program line for outputting a command to the HP 4155A/4156A:

```
OUTPUT @Hp415x;""
```

2. Positions the cursor at the second double quotes. You enter desired command, then select Enter key.

`ENTER @Hp415x`

1. Displays the following program line for entering data from the HP 4155A/4156A:

```
ENTER @Hp415x;
```

2. Positions the cursor after the semicolon. You enter desired variable, then select Enter key.

`PAUSE`

Enters the following program line for pausing a program:

```
PAUSE
```

`DISP`

1. Displays the following program line for displaying a message:

```
DISP ""
```

2. Positions the cursor at the second double quotes. You enter the message that you want to display, then select Enter key.

`INPUT`

1. Displays the following program line for assigning keyboard input to program variable:

```
INPUT "",
```

2. Positions the cursor at the second double quote. Enter string that you want to be displayed on CRT, and name of variable in which you want to store keyboard input, then select Enter key.

`IF THEN`
`ELSE`
`END IF`

1. Displays the following program lines for conditional branching:

```
IF  THEN
ELSE
END IF
```

2. Positions the cursor before `THEN`. Fill in as desired, then select Enter key.

`WHILE`
`END WHI`
`LE`

1. Displays the following program lines for defining a loop:

```
WHILE
END WHILE
```

2. Positions the cursor after `WHILE`. Fill in as desired, then select Enter key.

`FOR`
`NEXT`

1. Displays the following program lines for defining a loop:

```
FOR = TO  STEP
NEXT
```

2. Positions the cursor at `=`. Fill in as desired, then select Enter key.

External Keyboard

You can connect an external keyboard to the HP 4155A/4156A and use to enter text.

Also, you can use the keyboard for other tasks as described in this chapter.

In this section, the notation "KeyA+KeyB" means to hold down KeyA and press KeyB.

Esc	Exits the editor	
F1 through F8	Primary softkeys	Corresponds to the primary softkeys.
Shift+ F1 through F7	Secondary softkeys	Corresponds to the secondary softkeys.
F9	Screen mode	Toggles the screen mode as follows:



PG02003 120x20

Same as Ctrl+G.

F11	Clear to end	Deletes characters from cursor to end of line.

Same as Ctrl+Delete

Shift+F11	Clear line	When editor is running, same as F11. When editor is not running, deletes characters on the Command Entry line.

F12	Clear display	Clears the display for IBASIC. When the editor is running, exits from the editor, and clears the display for IBASIC.
Print Screen	Clear I/O	Stops I/O operation of program.
Scroll Lock	Stop	Stops program execution after executing the current line.
	Same as Shift+Pause.	
Shift+Scroll Lock	Reset	Stops program execution immediately.
Pause	Pause	Pauses program execution until `CONT` is executed or `Continue` primary softkey is pressed. If the program is modified while paused, `RUN` must be used to restart program execution.
	Same as Ctrl+P.	
Shift+Pause	Stop	Stops program execution after executing the current line.
	Same as Scroll Lock.	
Insert	Insert	When the editor is running, opens a new line before the current line. When the editor is not running, inserts text at the cursor (press Insert again to end insert mode).
	Same as Alt+I.	
Shift+Insert	Insert	Same as Insert.
Delete	Delete	Deletes character at the cursor.
Shift+Delete	Delete line	When the editor is running, deletes the current line. When the editor is not running, deletes character at cursor.
	Same as Alt+D.	

Ctrl + Delete	Clear to end Same as F11.	Deletes characters from cursor to end of line.
Home	Beginning of line Same as Shift + ◀.	Moves the cursor to beginning of the line.
Shift + Home	Page move	When the editor is running, same as Page Up. Also, same as Shift + ▲. When the editor is not running, jumps to the top of the IBASIC output area. Also, same as Shift + ▼.
End	End of line Same as Shift + ▶.	Moves cursor to end of line.
Shift + End	Page move	When the editor is running, same as Page Down. Also, same as Shift + ▼. When the editor is not running, jumps to the bottom of the IBASIC output area. Same as Shift + ▲.
Page Up	Page move	When the editor is running, moves the cursor one-half display page toward the beginning of the program. Same as Shift + Home. Same as Shift + ▲. When the editor is not running, moves display down one page.
Shift + Page Up	Recall	When the editor is running, displays last deleted line. To enter line into program, press Enter. When the editor is not running, cycles through the 10 commands that were most recently entered on the Command Entry line.

Page Down	Page move	When the editor is running, moves the cursor one-half display page toward the end of the program. Same as Shift + End. Same as Shift + ▼. When the editor is not running, moves display up one page.
Shift + Page Down	Recall backward	When the editor is running, same as Shift + Page Up. When the editor is not running, cycles through the 10 commands that were most recently entered on the Command Entry line in reverse order of Shift + Page Up.
▲	Previous line	When the editor is running, moves cursor up one line. When the editor is not running, display on the IBASIC output area moves one line toward the end.
Shift + ▲	Page move	When the editor is running, same as Page Up. Also, same as Shift + Home. When the editor is not running, jumps to the bottom of the IBASIC output area. Same as Shift + End.
Alt + ▲	Recall backward	When the editor is running, same as Shift + Page Up. When the editor is not running, same as Shift + Page Down.
▼	Next line	When the editor is running, cursor moves down one line. When editor is not running, display on IBASIC output area moves one line toward beginning.
Shift + ▼	Page move	When the editor is running, same as Page Down. Also, same as Shift + End. When the editor is not running, jumps to the top of the IBASIC output area. Same as Shift + Home.
Alt + ▼	Recall	Same as Shift + Page Up.

◀, ▶	Move cursor	Moves the cursor one character in indicated direction.
Shift+◀	Beginning of line Same as Home.	Moves the cursor to beginning of line.
Shift+▶	End of line Same as End.	Moves the cursor to end of line.
Backspace	Backspace	When the editor is running, deletes the character before cursor. When the editor is not running, deletes the character before cursor (if mode is insert mode). If mode is not insert move, moves cursor to left by one cursor.
Alt+D	Delete line Same as Shift+Delete.	When the editor is running, deletes the current line. When the editor is not running, deletes the character at the cursor.
Alt+I	Insert line Same as Insert.	When the editor is running, opens a new line before the current line. When the editor is not running, inserts text at the cursor (press Insert again to end insert mode).
Ctrl+U	Run	Executes the program.
Ctrl+P	Pause Same as Pause.	Pauses program execution until `CONT` is executed or `Continue` primary softkey is pressed. If the program is modified while paused, `RUN` must be used to restart program execution.

Ctrl + G

Screen mode

Toggles the screen mode as follows:



PG02003 120x20

Same as F9

HP 4155A/4156A Specific IBASIC Keywords

The following keywords are not standard IBASIC keywords, or are standard keywords, but with a difference. These keywords are specific to HP 4155A/4156A.

EXECUTE — Not standard IBASIC keyword. Refer to "ASP-like Commands" for details.

ON KEY — Standard IBASIC keyword, except the range of *key selector* is 1 to 7.
1 to 7 of *key selector* corresponds to secondary softkeys 1 to 7, respectively.

PEN — Standard IBASIC keyword, except the range of *pen selector* is 7. The following table shows the corresponding color for each *pen selector*.

pen selector	**Color**
1	color defined for Foreground on SYSTEM: COLOR SETUP page.
2	color defined for Y1 Axis on SYSTEM: COLOR SETUP page.
3	color defined for Y2 Axis on SYSTEM: COLOR SETUP page.
4	color defined for Marker/Cursor/Line on SYSTEM: COLOR SETUP page.
5	color defined for Active Mkr/Csr/Lne on SYSTEM: COLOR SETUP page.
6	color defined for Advisory on SYSTEM: COLOR SETUP page.
7	color defined for Title on SYSTEM: COLOR SETUP page.

The following IBASIC keywords are not implemented in HP 4155A/4156A's Instrument BASIC.

- ALPHA ON/OFF
- AREA
- CLIP
- CONTROL
- DUMP
- EDGE
- FILL
- FRAME
- GESCAPE
- GLOAD
- GRAPHICS
- GRID
- GSTORE
- LINE TYPE
- PLOTTER IS
- POLYGON
- POLYLINE
- RATIO
- RECTANGLE
- SET PEN
- SHOW
- STATUS
- VIEWPORT
- WINDOW

Available I/O Resources for IBASIC

This section provides information about available I/O resources for IBASIC of HP 4155A/4156A.

The following I/O resources are available for IBASIC.

- CRT Display
- External keyboard and front-panel keyboard
- HP-IB Interface on rear panel
- Internal pseudo HP-IB Interface (to control HP 4155A/4156A itself)
- Serial Interface
- Built-in Disk Drive (no select code)

Table 2-1 shows available I/O interfaces and their select codes.

Table 2-1. Available I/O Interfaces and Select Code

Select Code	Interface
1	CRT
2	External and front-panel keyboard
7	HP-IB Interface on rear panel
8	Internal pseudo HP-IB Interface
9	Serial Interface

CRT Display

IBASIC can display text or graphics on the CRT display of the HP 4155A/4156A.

Text display.
Text can be displayed in the IBASIC output area of "All IBASIC" screen.

Graphics display.
In "All IBASIC" screen, you can display a graphical plot.

The x and y coordinate values are as follows:

- lower left corner of screen: (0,0)
- upper right corner of screen: (545,400).

HP-IB Interfaces

- Internal pseudo HP-IB

  By using select code 8, you can control HP 4155A/4156A via internal pseudo HP-IB interface. The HP-IB address of HP 4155A/4156A has no meaning, so you can use any address (0 through 30).

- HP-IB on rear panel

  You can access HP-IB interface on rear panel by using select code 7.

Serial Interface

You can access serial interface on rear panel by using select code 9.

Receiving data

To receive data from serial interface successfully, be sure to do the following:

- Use `ASSIGN` IBASIC command to open I/O path. Refer to the following example:

```
ASSIGN @Serial TO 9
ENTER @Serial;A$
```

  Do not use `ENTER 9;A$` because every time it is executed, the I/O path is opened, which may cause errors.

- If you receive multiple lines of data (such as by a loop of ENTER statements), make sure no other operation or statement occurs between receptions. It may cause an overrun error. If overrun error is caused, use a slower baud rate for receiving.

Built-in Diskette Drive

If you specify optional volume specifier when accessing the built-in diskette drive, the volume specifier must be ":INTERNAL,4".

Available diskettes.
You can use the following types of 3.5 inch diskettes:

- 2HD 1.44 MB
- 2DD 720 KB

Diskette must be formatted as LIF or the following DOS format:

- 80 tracks/side
- 18 sectors/track (2HD)
  9 sectors/track (2DD)
- 512 bytes/sector

ASP-like Commands

EXECUTE is an IBASIC keyword for executing function directives, which allow you to easily create simple programs in a way similar to creating Auto Sequence Programs (ASP) on the HP 4145A/B Semiconductor Parameter Analyzer.

Compatibility Consideration

EXECUTE is *not* a standard IBASIC or HP BASIC keyword. So, if you use this keyword in your program, it will *not* execute on another IBASIC or HP BASIC system.

EXECUTE

Keyboard Executable	Yes
Programmable	Yes
In an IF . . . THEN . . .	Yes

This keyword can execute the function directives that are described on the following pages.

Syntax

EXECUTE ("*directive_keyword* [,*parameter*]")

directive

Some directives require parameters. There must be one or more spaces between *directive_keyword* and *parameter*.

The following pages describe the *directives* that can be used in the EXECUTE command.

GET SETUP Directive

This directive loads the specified HP 4155A/4156A setup file.

Directive syntax

```
GETSETUP file_name [,DISK|MEMORY]
```

Directive parameter

Parameter	Type	Explanation
file_name	string	name of setup file with extension |.MES or .STR| to be loaded. You must enclose the name with single quotes or double-double quotes.
DISK	character	|default| loads setup data from internal diskette.
MEMORY	character	loads setup data from internal memory.

Example

```
EXECUTE("GETSETUP 'SWEEP.MES'")
EXECUTE("GETSETUP ""SWEEP.MES""")
EXECUTE("GETSETUP File$,DISK")
EXECUTE("GETSETUP 'MEM1.MES',MEMORY")
```

SINGLE Directive

This directive executes measurement.

Directive syntax

```
SINGLE
```

Example

```
EXECUTE("SINGLE")
```

STANDBY directive

This directive changes `STBY ON` channels to standby state or idle state.

Directive syntax

```
STANDBY ON|OFF
```

Directive parameter

Parameter	Type	Explanation
ON	character	changes `STBY ON` channels from idle state to standby state.
OFF	character	changes `STBY ON` channels from standby state to idle state.

Example

```
EXECUTE("STANDBY ON")

EXECUTE("STANDBY OFF")
```

STRESS Directive

This directive forces stress.

Directive syntax

```
STRESS
```

Example

```
EXECUTE("STRESS")
```

AUTO-SCALE Directive

This directive changes page to GRAPH/LIST: GRAPHICS and executes auto-scaling function.

Directive syntax

```
AUTOSCALE
```

Example

```
EXECUTE("AUTOSCALE")
```

SAVE DATA Directive

This directive stores measurement data file to internal diskette or internal memory.

Directive syntax

SAVEDATA *file_name* [,DISK|MEMORY]

Directive parameter

Parameter	Type	Explanation
file_name	string	name of measurement data file with extension |.DAT| to be stored. You must enclose the name with single quotes or double-double quotes.
DISK	character	|default| stores measurement data to internal diskette.
MEMORY	character	stores measurement data to internal memory.

Example

```
EXECUTE("SAVEDATA 'SWEEP.DAT'")
EXECUTE("SAVEDATA ""SWEEP.DAT""")
EXECUTE("SAVEDATA File$,DISK")
EXECUTE("SAVEDATA 'MEM1.DAT',MEMORY")
```

READ DATA VARIABLE Directive

This directive gets values of specified HP 4155A/4156A data variable, and stores the values in an IBASIC variable.

Directive syntax

`READDATAVAR` *data_variable_name*`,`*ibasic_variable_name*`)`

Directive parameter

Parameter	Type	Explanation
data_variable_name	string	name of the data variable of HP 4155A/4156A. You must enclose the name with single quotes or double-double quotes. Name is case sensitive.
ibasic_variable_name	string	name of numeric variable or numeric array of IBASIC program. *ibasic_variable_name* is not case sensitive.

Example

```
EXECUTE("READDATAVAR 'V1',V")

EXECUTE("READDATAVAR ""V1"",v")
```

DEFINE USER VARIABLE Directive

This directive defines an HP 4155A/4156A user variable, and transfers values from an IBASIC variable to the user variable.

Directive syntax

`DEFUSERVAR` *user_variable_name*`,`*no_of_points*`,`*ibasic_variable_name* [`,`*unit*]

Directive parameter

Parameter	Type	Explanation
user_variable_name	string	user variable name that you want to define. You must enclose the name with single quotes or double-double quotes.
no_of_points	numeric	number of data for the user variable
ibasic_variable_name	string	name of numeric variable or numeric array of IBASIC program. The data in this variable will be transferred to the user variable.
unit	string	unit of user variable. You must enclose the unit with single quotes or double-double quotes.

Example

```
EXECUTE("DEFUSERVAR 'U_var',101,Vth,'V'")
```

PRINT/PLOT Directive

This directive prints/plots the information of the present instrument page, not the IBASIC screen. If present page is GRAPH/LIST: GRAPHICS page, the graph is printed/plotted.

Directive syntax

```
PRINTPLOT
```

Example

```
EXECUTE("PRINTPLOT")
```

CURVE PLOT Directive

This directive changes to GRAPH/LIST: GRAPHICS page, then prints/plots the graph.

Directive syntax

```
CURVEPLOT
```

Example

```
EXECUTE("CURVEPLOT")
```

3

Getting Started on Programming the HP 4155A/4156A

Getting Started on Programming the HP 4155A/4156A

This chapter provides step-by-step tutorials for programming to control the HP 4155A/4156A along with programming examples.

This chapter consists of the following sections:

- Creating a Simple Measurement Program

  This section introduces how to create a measurement program.

- Programming for Data Extraction

  This section provides the programming tutorials for data extraction.

- Complete Example Program for Vth Measurement

  This section shows complete example program based on the parts described in the other sections.

- Example Application Setup for Vth Measurement

  This section describes an example application setup that you should save to the file named `VTH.MES` on diskette before executing program examples (that use `VTH.MES`) described in this chapter.

In addition to this chapter, Chapter 6 provides some application examples which are helpful to increase your understanding.

Creating a Simple Measurement Program

This section introduces how to create a measurement program.

A simple measurement program created by using built-in IBASIC controller is provided as an example and you learn step-by-step how to create a measurement program.

This section consists of the following:

1. Getting a setup file and making a measurement
2. Changing the sweep setup parameters
3. Changing the display setup parameters
4. Saving the measurement results to a diskette
5. Printing the measurement results

Before Creating a Program

This section assumes that you have already saved a measurement setup file for Vth measurement to diskette.

Before starting this section, prepare the diskette and save the measurement setup (described in "Example Application Setup for Vth Measurement") to the file named "VTH.MES" on the diskette.

Step 1. Getting the Setup File and Making a Measurement

In this step, you can create a program to get a setup file from the diskette and execute a measurement.

1. Press IBASIC (Display) key until screen display mode is All IBASIC mode.
2. Select `EDIT` softkey, then press (Enter) key to start the IBASIC editor
3. If there is an existing program, save it if necessary.
4. Delete existing program and assign I/O path to control HP 4155A/56A.

   Type `SCRATCH`, then (Enter). Or select `Scratch` primary softkey to delete the program.

   Existing program is deleted and the following program lines are entered automatically. These lines are for assigning HP 4155A/56A control I/O path.

```
   10 COM @Hp415x
   20 ASSIGN @Hp415x TO 800
   30 !
 9990 END
```

Line number	Description
10	Declares COM so that subprograms can access the I/O path that is assigned in line 20 for controlling the HP 4155A/56A. Refer to the *HP Instrument BASIC Users Handbook* for details.
20	Assigns the I/O path for controlling the HP 4155A/56A . `800` means built-in IBASIC controller.

To ASSIGN I/O path

- Built-in IBASIC controller

  Specify select code 8. For the HP-IB address, you can use *any* number between 0 to 31. Refer to the following example:

  ```
  10 ASSIGN @Hp4155 TO 800
  ```

- HP BASIC on an external computer

  Specify the select code of the external computer. And specify the HP-IB address that you entered into the HP-IB ADDRESS field on the SYSTEM: MISCELLANEOUS page. In the following example, the select code of the external computer is 7 and HP-IB address of HP 4155A/56A is 17:

  ```
  10 ASSIGN @Hp4155 TO 717
  ```

5. Select `OUTPUT @Hp415x` secondary softkey.

   The following characters are automatically entered:

   ```
   30 OUTPUT @Hp415x;""
   ```

   Do not press Enter yet.

6. Use the help function to find the command for getting a setup file:

   a. Press Help.

   b. Press Get.

The cursor in help window automatically jumps to the command (:MMEM:LOAD:STAT) for getting a setup file.

7. Press (Enter) to insert the command into the program line.

   Now line 30 is as follows:

   ```
   30 OUTPUT @Hp415x;":MMEM:LOAD:STAT"
   ```

8. Type in the command parameters as in following example:

   ```
   30 OUTPUT @Hp415x;":MMEM:LOAD:STAT 0,'VTH.MES','DISK'"
   ```

Parameter	Description
0	No meaning (dummy parameter).
'VTH.MES'	File name to be loaded.
'DISK'	Source mass storage is diskette.

   Then press (Enter). Then select the `Insert line` softkey.

9. Select `OUTPUT @Hp415x` secondary softkey.

   ```
   40 OUTPUT @Hp415x;""
   ```

10. Press (Help), then press (Single), (Append), or (Repeat) to find the command for executing a measurement.

Execution Key	Command
(Single)	:PAGE:SCON:MEAS:SING
(Append)	:PAGE:SCON:MEAS:APP
(Repeat)	:PAGE:SCON:MEAS:REP

11. Press (Enter) to insert the found command into the program line.

    ```
    10 COM @Hp415x
    20 ASSIGN @Hp415x TO 800
    30 OUTPUT @Hp415x;":MMEM:LOAD:STAT 0,'VTH.MES','DISK'"
    40 OUTPUT @Hp415x;":PAGE:SCON:MEAS:SING"
  9990 END
    ```

    Then press (Enter)

12. Press `End edit` to exit from the editor.

Now you have created a measurement program.

To execute the program, do as follows:

1. Press IBASIC (Display) key until screen display mode is All Instrument or IBASIC Status mode. This allows you to monitor the measurement on GRAPH/LIST: GRAPHICS page.
2. Press (Run) front-panel key. The measurement program is executed.

Controlling from External Computer

You must do as follows before controlling HP 4155A/4156A from an external computer:

1. Connect the HP-IB interface of external computer to HP-IB interface on rear panel of HP 4155A/4156A.
2. Set the `HP 4155A/56A is` field on the SYSTEM: MISCELLANEOUS page to NOT SYSTEM CONTROLLER.
3. Enter the HP-IB address of your HP 4155A/4156A into the `HP-IB ADDRESS` field.

Step 2. Changing the Sweep Setup Parameters

Modify measurement program created in previous step so that you can enter new sweep start and stop values while program is running:

1. Press IBASIC (Display) key until the screen display mode is All IBASIC mode.
2. Select `EDIT` softkey, then press (Enter) key to start the IBASIC editor
3. Insert program lines that allow you to enter the sweep start and stop values from the keyboard during program running.
   a. Move the cursor to program line 30.
   b. Select `Insert line` primary softkey.
   c. Type the following program lines:

```
21 !
22 INPUT "Sweep Start (V)?",Start_v
23 INPUT "Sweep Stop (V)?",Stop_v
24 !
```

4. If you do not know the SCPI commands for changing the sweep start and stop parameters, do as follows:
   a. Press IBASIC (Display) key until screen display mode is All Instrument mode.
   b. Press (Meas) to change to MEASURE: SWEEP SETUP page.
   c. Move the field pointer to the parameter that you want to change.
   d. Press (Help) key. The corresponding command is displayed at the bottom of the help window:

Command	Description
:PAGE:MEAS:SWE:VAR1:STAR	VAR1 sweep start
:PAGE:MEAS:SWE:VAR1:STOP	VAR1 sweep stop

   You need to remember the commands, so that you can enter them in the next step.
   e. Select the `EXIT HELP` softkey.

5. Press IBASIC (Display) key until screen display mode is All IBASIC mode. Then, do the following to insert the program lines for changing the sweep start and stop values.

   a. Move the cursor to the program line 40.
   b. Select `Insert line` primary softkey.
   c. Select the `OUTPUT @Hp415x` softkey.
   d. Type in the SCPI command. Or you can use the help function to enter the command. For the help function, see "To Use the Help Function" in Chapter 1.

   After you finish, the program lines should look as follows:

```
31 OUTPUT @Hp415x;":PAGE:MEAS:SWE:VAR1:STAR",Start_v
32 OUTPUT @Hp415x;":PAGE:MEAS:SWE:VAR1:STOP",Stop_v
```

   Now the program is as follows:

```
  10 COM @Hp415x
  20 ASSIGN @Hp415x TO 800
  21 !
  22 INPUT "Sweep Start (V)?",Start_v
  23 INPUT "Sweep Stop (V)?",Stop_v
  24 !
  30 OUTPUT @Hp415x;":MMEM:LOAD:STAT 0,'VTH.MES','DISK'"
  31 OUTPUT @Hp415x;":PAGE:MEAS:SWE:VAR1:STAR",Start_v
  32 OUTPUT @Hp415x;":PAGE:MEAS:SWE:VAR1:STOP",Stop_v
  40 OUTPUT @Hp415x;":PAGE:SCON:MEAS:SING"
9990 END
```

6. Select `End edit` to exit from the editor.
7. Press IBASIC (Display) key until screen display mode is IBASIC Status mode.
8. Press (Run) to execute the program.
9. `Sweep Start (V)?` is displayed on the display line. Enter the desired sweep start voltage.
10. `Sweep Stop (V)?` is displayed on the display line. Enter the desired sweep stop voltage.

Step 3. Changing the Display Setup Parameters

In this step, change X-axis range of display setup parameters to match the sweep start and stop values.

1. Press IBASIC (Display) key until screen display mode is All IBASIC mode.
2. Select `EDIT` softkey, then press (Enter) key to start the IBASIC editor
3. If you do not know the SCPI commands for changing the X-axis parameters, do as follows:
   a. Press IBASIC (Display) key until screen display mode is All Instrument mode.
   b. Press PAGE CONTROL (Display) key to change to DISPLAY: DISPLAY SETUP page.
   c. Move the field pointer to the parameter that you want to change.
   d. Press (Help) key. The corresponding commands are displayed at the bottom of the help window. You need to remember the commands, so that you can enter them in the next step.

Command	Description
:PAGE:DISP:SET:GRAP:X:MIN	minimum value of X-axis
:PAGE:DISP:SET:GRAP:X:MAX	maximum value of X-axis

   e. Select the `EXIT HELP` softkey.

4. Press IBASIC (Display) key until screen display mode is All IBASIC mode. Then, do the following to insert the program lines for changing the X-axis display parameters:

   a. Move the cursor to the program line 40.
   b. Select `Insert line` primary softkey.
   c. Select the `OUTPUT @Hp415x` softkey.
   d. Type in the SCPI command. Or you can use the help function to enter the command. For the help function, see "To Use the Help Function" in Chapter 1. After you finish, the program lines should look as follows:

```
33 OUTPUT @Hp415x;":PAGE:DISP:SET:GRAP:X:MIN",Start_v
34 OUTPUT @Hp415x;":PAGE:DISP:SET:GRAP:X:MAX",Stop_v
```

   Now the program is as follows:

```
 10 COM @Hp415x
 20 ASSIGN @Hp415x TO 800
 21 !
 22 INPUT "Sweep Start (V)?",Start_v
 23 INPUT "Sweep Stop (V)?",Stop_v
 24 !
 30 OUTPUT @Hp415x;":MMEM:LOAD:STAT 0,'VTH.MES','DISK'"
 31 OUTPUT @Hp415x;":PAGE:MEAS:SWE:VAR1:STAR",Start_v
 32 OUTPUT @Hp415x;":PAGE:MEAS:SWE:VAR1:STOP",Stop_v
 33 OUTPUT @Hp415x;":PAGE:DISP:SET:GRAP:X:MIN",Start_v
 34 OUTPUT @Hp415x;":PAGE:DISP:SET:GRAP:X:MAX",Stop_v
 40 OUTPUT @Hp415x;":PAGE:SCON:MEAS:SING"
9990 END
```

5. Select `End edit` to exit from the editor.
6. Press IBASIC (Display) key until screen mode is IBASIC Status mode.
7. Press (Run) to execute the program.
8. Enter the desired sweep start and stop values as prompted. The minimum and maximum X-axis values of the graph will be the same as these entered values.

Step 4. Saving All Measurement Results to a Diskette.

In this step, add program lines that save the setup data and measurement results to the diskette.

1. Press IBASIC (Display) key until screen display mode is All IBASIC mode.
2. Select `EDIT` softkey, then press (Enter) key to start the IBASIC editor
3. Move the cursor to program line 9990.
4. Select `Insert line` primary softkey.
5. Insert the following program lines, which wait until the measurement is completed.

```
50 OUTPUT @Hp415x;"*OPC?"
60 ENTER @Hp415x;Complete
```

   When measurement is completed, HP 4155A/56A returns `1` to the `Complete` variable.

6. Insert the following program line, which saves the measurement setup and results to a file named `VTH.DAT`:

```
70 OUTPUT @Hp415x;":MMEM:STOR:TRAC DEF,'VTH.DAT'"
```

   Now the program is as follows:

```
 10 COM @Hp415x
 20 ASSIGN @Hp415x TO 800
 21 !
 22 INPUT "Sweep Start (V)?",Start_v
 23 INPUT "Sweep Stop (V)?",Stop_v
 24 !
 30 OUTPUT @Hp415x;":MMEM:LOAD:STAT 0,'VTH.MES','DISK'"
 31 OUTPUT @Hp415x;":PAGE:MEAS:SWE:VAR1:STAR",Start_v
 32 OUTPUT @Hp415x;":PAGE:MEAS:SWE:VAR1:STOP",Stop_v
 33 OUTPUT @Hp415x;":PAGE:DISP:SET:GRAP:X:MIN",Start_v
 34 OUTPUT @Hp415x;":PAGE:DISP:SET:GRAP:X:MAX",Stop_v
 40 OUTPUT @Hp415x;":PAGE:SCON:MEAS:SING"
 50 OUTPUT @Hp415x;"*OPC?"
 60 ENTER @Hp415x;Complete
 70 OUTPUT @Hp415x;":MMEM:STOR:TRAC DEF,'VTH.DAT'"
9990 END
```

7. Select `End edit` to exit from the editor.

8. Press IBASIC (Display) key until screen display mode changes to IBASIC Status mode.
9. Press (Run) to execute the program.

   The measurement setup and results are automatically saved to the diskette after measurement is performed.

Step 5. Printing the Measurement Results

Printing from IBASIC

If you use built-in IBASIC controller to control HP 4155A/4156A, you do not need to prepare before controlling HP 4155A/4156A because built-in IBASIC controller is always connected to HP 4155A/4156A via internal HP-IB.

However, *to print/plot* from built-in IBASIC controller, set "`HP 4155A/56A is`" field on the SYSTEM: MISCELLANEOUS page to SYSTEM CONTROLLER.

In this step, add program lines that print the measurement results.

1. Press (Display) key until screen display mode is All IBASIC mode.
2. Select `EDIT` softkey, then press (Enter) key to start the IBASIC editor
3. Move the cursor to the program line 70.
4. Select `Insert line` primary softkey.
5. Insert the following program lines, which print a screen dump of the results:

```
61 OUTPUT @Hp415x;":HCOP:SDUM"
62 DISP "Printing"
63 OUTPUT @Hp415x;"*OPC?"
64 ENTER @Hp415x;Complete
65 DISP "Done"
```

`:HCOP` immediately initiates the plot or print according to the current setup. After printing is finished, HP 4155A/56A returns `1` to the `Complete` variable, then "Done" is displayed on the screen.

Now the program is as follows:

```
  10 COM @Hp415x
  20 ASSIGN @Hp415x TO 800
  21 !
  22 INPUT "Sweep Start (V)?",Start_v
  23 INPUT "Sweep Stop (V)?",Stop_v
  24 !
  30 OUTPUT @Hp415x;":MMEM:LOAD:STAT 0,'VTH.MES','DISK'"
  31 OUTPUT @Hp415x;":PAGE:MEAS:SWE:VAR1:STAR",Start_v
  32 OUTPUT @Hp415x;":PAGE:MEAS:SWE:VAR1:STOP",Stop_v
  33 OUTPUT @Hp415x;":PAGE:DISP:SET:GRAP:X:MIN",Start_v
  34 OUTPUT @Hp415x;":PAGE:DISP:SET:GRAP:X:MAX",Stop_v
  40 OUTPUT @Hp415x;":PAGE:SCON:MEAS:SING"
  50 OUTPUT @Hp415x;"*OPC?"
  60 ENTER @Hp415x;Complete
  61 OUTPUT @Hp415x;":HCOP:SDUM"
  62 DISP "Printing"
  63 OUTPUT @Hp415x;"*OPC?"
  64 ENTER @Hp415x;Complete
  65 DISP "Done"
  70 OUTPUT @Hp415x;":MMEM:STOR:TRAC DEF,'VTH.DAT'"
9990 END
```

Programming for Data Extraction

This section provides the following programming tutorials for data extraction:

- Reading HP 4155/56 setup data
- Reading values of data variables (measurement results)
- Transferring data into a file

Reading HP 4155/56 Setup Data

To read setup data from HP 4155/56 into an IBASIC variable, use the query form of the corresponding setting command. To make the query form of a command, simply add a question mark (`?`) to the end of the command.

Refer to the following program lines of example program:

```
60      OUTPUT @Hp415x;":PAGE:MEAS:SWE:VAR1:STAR?"
70      ENTER @Hp415x;Vd_start
80      OUTPUT @Hp415x;":PAGE:MEAS:SWE:VAR1:STOP?"
90      ENTER @Hp415x;Vd_stop
100     OUTPUT @Hp415x;":DISP:ALL BAS"
110     CLEAR SCREEN
120     PRINT TABXY(1,1);"Vd START=";Vd_start;"(V)"
130     PRINT TABXY(1,2);"Vd STOP =";Vd_stop;"(V)"
```

Line 60 — This query command tells the HP 4155A/56A to put the VAR1 start value in its output buffer.

`:PAGE:MEAS:SWE:VAR1:STAR` is the command for setting the VAR1 start value. By adding `?`, the command becomes the query command for reading the VAR1 start value.

Line 70 — This gets the start value from the output buffer, then enters it in the `Vd_start` variable.

Line 80 to 90 — These lines tell the HP 4155A/56 to put VAR1 stop value in its output buffer, then the value is entered into the `Vd_stop` variable.

Reading HP 4155/56 Measurement Data

To read read-out function values or data variable values (output data, measurement data, and user function values) from HP 4155/56 to IBASIC variables, use the `:DATA?` command.

Refer to the following program lines in the example program:

```
410     OUTPUT @Hp415x;":PAGE:SCON:MEAS:SING"
420     OUTPUT @Hp415x;"*OPC?"
430     ENTER @Hp415x;Complete
440     OUTPUT @Hp415x;":DATA? 'VTH'"
450     ENTER @Hp415x;Vth
```

Line 410 Execute single measurement.

Line 420 to 430 Wait for measurement completion.

Line 440 Send `:DATA?` query command to read the result value of user function "VTH".

Line 450 Store the result value into `Vth` variable.

NOTE

Be aware that data variable names, such as user functions and user variables, are *case sensitive*. For example, if you set up user function name **VTH** on the CHANNEL: USER FUNCTION DEFINITION page, then to read it, you must use :DATA? 'VTH', not :DATA? 'Vth'.

Transferring Specific HP 4155/56 Data to a File

To transfer data from the HP 4155A/56A to a file, do as follows:

1. Create a data file.

   You can create three types of data files: DOS, LIF ASCII, or BDAT as follows:

   ```
   CREATE "data_file",1              ! Creates a DOS file.

   CREATE ASCII "ascii_file",100     ! Creates a LIF ASCII file.

   CREATE BDAT "binary_file",100     ! Creates a BDAT file.
   ```

   DOS files are compatible with MS-DOS, which are easy to transfer to PCs and other computers.

   LIF ASCII files are compatible with HP computers that support this file type, so this type is best is you are transferring files among HP computers that support this file type.

   BDAT (binary data) files provide more flexibility (can specify both number of records and record length) and faster transfer rate. But BDAT files cannot be interchanged with as many other systems.

   The first parameter of each statement specifies the file name to create.

   The second parameter specifies number of records to allocate for the file as follows:

   DOS — Second parameter specifies how many records are to be *initially* allocated for the file. A DOS file system automatically allocates additional space for the file as new data is written to it, so you can always specify `1` for this parameter.

   LIF ASCII — Second parameter specifies *total* number of records to allocate for the file, so you must specify a sufficient number of records. The length of one record is 256 bytes.

   For example, the following statement would create a file with 100 records (each record is 256 bytes):

   ```
   CREATE ASCII "File",100
   ```

BDAT — Second parameter specifies *total* number of records to allocate for the file, so you must specify a sufficient number of records. You can specify a record length by using an optional third parameter (default length is 256 bytes).

For example, the following statement creates a file with 7 records (each record is 128 bytes):

```
CREATE BDAT "B_file",7,128
```

The following statement creates a file with 7 records (each record is 256 bytes):

```
CREATE BDAT "B_file",7
```

2. Open an I/O path for transferring data into the file.

   To open an I/O path to the file, assign an I/O path name to the file by using an ASSIGN statement as in the following example:

```
340    INPUT "Enter file name to store data",File$
350    CREATE File$,1
360    ASSIGN @File TO File$;FORMAT ON
```

   Line 350 creates a DOS file, then line 360 opens an I/O path to the file.

   For DOS and BDAT files, ASSIGN statement can also specify the following:

   FORMAT ON — ASCII data representations are used. Specify this if you need to transport data between IBASIC and other machines.

   FORMAT OFF — IBASIC internal data representations are used. Specify this if you need a faster transfer rate and space efficiency.

3. Store data into the file.

   To store data into a file, use `OUTPUT` and `ENTER` statements as in the following examples:

```
340    INPUT "Enter file name to store data",File$
350    CREATE File$,1
360    ASSIGN @File TO File$;FORMAT ON
       :
390    REPEAT
       :
440      OUTPUT @Hp415x;":DATA? 'VTH'"
450      ENTER @Hp415x;Vth
```

```
460      OUTPUT @File;Vth
  :
630    UNTIL Stop$="S" OR Stop$="s"
```

The above program repeats appending `Vth` variable value to a DOS file in ASCII format.

In addition to numeric data, array data and string data can be stored to a file as in following examples:

- Array data:

```
1    DIM Vth(1:100)
  :
340    INPUT "Enter file name to store data",File$
350    CREATE File$,1
360    ASSIGN @File TO File$;FORMAT ON
  :
390  FOR I=1 TO 100
  :
440      OUTPUT @Hp415x;":DATA? 'VTH'"
450      ENTER @Hp415x;Vth(I)
460  NEXT I
470  OUTPUT @File;Vth(*)
  :
```

- String data:

```
10  DIM Data$[10](1:100)
20  CREATE "DATAFILE",1
30  ASSIGN @File TO "DATAFILE";FORMAT ON
40  FOR I=1 TO 100
50     Data$(I)="ABC"
60  NEXT I
70  OUTPUT @File;Data$(*)
  :
```

4. Close the I/O path.

   To close an I/O path to a file, `ASSIGN` the path name to an * (asterisk) as in the following example:

```
340    INPUT "Enter file name to store data",File$
350    CREATE File$,1
360    ASSIGN @File TO File$;FORMAT ON
  :
460    OUTPUT @File;Vth
  :
590    ASSIGN @File TO *
```

   In this program, line 590 closes the I/O path that was opened by line 360.

Complete Example Program for Vth Measurement

The example program shown below uses the measurement setup file described in "Example Application Setup for Vth Measurement". This is a complete example program based on the parts described in the previous sections.

```
10     COM @Hp415x
20     ASSIGN @Hp415x TO 800
30     OUTPUT @Hp415x;":MMEM:LOAD:STAT 0,'VTH.MES','DISK'"
40    !
50    ! Read and Disp. Measurement Conditions
60     OUTPUT @Hp415x;":PAGE:MEAS:SWE:VAR1:STAR?"
70     ENTER @Hp415x;Vd_start
80     OUTPUT @Hp415x;":PAGE:MEAS:SWE:VAR1:STOP?"
90     ENTER @Hp415x;Vd_stop
100    OUTPUT @Hp415x;":DISP:ALL BAS"
110    CLEAR SCREEN
120    PRINT TABXY(1,1);"Vd START=";Vd_start;"(V)"
130    PRINT TABXY(1,2);"Vd STOP =";Vd_stop;"(V)"
140   !
150   ! Parameter Change
160    Change$="n"
170 Change:   !
180    INPUT "Change these parameters? (y/n default=n)",Change$
190    SELECT Change$
200    CASE "Y","y"
210      INPUT "New Vd START (V)?",Vd_start
220      INPUT "New Vd STOP (V)?",Vd_stop
230      OUTPUT @Hp415x;":PAGE:MEAS:SWE:VAR1:STAR";Vd_start
240      OUTPUT @Hp415x;":PAGE:MEAS:SWE:VAR1:STOP";Vd_stop
250      PRINT TABXY(1,1);"Vd START=";Vd_start;"(V)         "
260      PRINT TABXY(1,2);"Vd STOP=";Vd_stop;"(V)           "
270    CASE "N","n"
280      GOTO Store_file
290    CASE ELSE
300      GOTO Change
310    END SELECT
320   !
330 Store_file:   !
340    INPUT "Enter file name to store data",File$
350    CREATE File$,1
360    ASSIGN @File TO File$;FORMAT ON
370   !
380    No_of_data=0
390    REPEAT
```

```
400       OUTPUT @Hp415x;":DISP:ALL BST"
410       OUTPUT @Hp415x;":PAGE:SCON:MEAS:SING"
420       OUTPUT @Hp415x;"*OPC?"
430       ENTER @Hp415x;Complete
440       OUTPUT @Hp415x;":DATA? 'VTH'"
450       ENTER @Hp415x;Vth
460       OUTPUT @File;Vth
470       No_of_data=No_of_data+1
480       OUTPUT @Hp415x;":DISP:ALL BASIC"
490       PRINT TABXY(1,10);"Last measured Vth =";Vth;"(V)                    "
500       PRINT TABXY(1,11);"Total number of die tested=";No_of_data
510 Stop_query:INPUT "Continue to next die or Stop test? (c/s)",Stop$
520       SELECT Stop$
530       CASE "C","c"
540         DISP "Move to the next die, then press [Continue]"
550         PAUSE
560         DISP ""
570       CASE "S","s"
580         PRINT TABXY(1,24);"Test Stopped!!"
590         ASSIGN @File TO *
600       CASE ELSE
610         GOTO Stop_query
620       END SELECT
630     UNTIL Stop$="S" OR Stop$="s"
640     END
```

Example Application Setup for Vth Measurement

This section describes an example application setup that you should save to the file named `VTH.MES` on diskette before executing program examples (that use `VTH.MES`) described previously in this chapter.

A frequently used method of measuring Vth is to synchronously sweep the exact same voltage to gate and drain, and measure the characteristics in the saturation region.

The theoretical value of drain current in the saturation region is calculated as follows:

$$Id = \beta * (Vg - Vth)^2$$

Where β is the gain factor, which is $-1/2 * (\mu\epsilon_{ox} W/L) * t_{ox}$. Therefore, if you take the square root of both sides of the equation:

$$\sqrt{Id} = \sqrt{\beta} * (Vg - Vth)$$

$\sqrt{Id}$ is proportional to Vg, and the slope is $\sqrt{\beta}$. At the point where $\sqrt{Id}$ is equal 0, *Vth* is equal to *Vg*. So, to know *Vth*, we need to find that point.

The measurement conditions are as follows:

- Channel Definition

Measurement Mode	Sweep

UNIT	VNAME	INAME	MODE	FCTN
SMU1	Vd	Id	V	VAR1'
SMU2	Vg	Ig	V	VAR1
SMU3	Vs	Is	COMMON	CONST
SMU4	Vsb	Isb	V	CONST

- Sweep Setup

	VAR1
UNIT	SMU2
NAME	Vg
SWEEP MODE	Single
LIN/LOG	LINEAR
START	0.0000 V
STOP	5.000 V
STEP	100.0 mV
NO OF STEP	51
COMPLIANCE	100.0 mA
POWER COMP	OFF

	VAR1'
UNIT	SMU1
NAME	Vd
OFFSET	0.0000 V
RATIO	1.000
COMPLIANCE	100.00 mA
POWER COMP	OFF

	CONSTANT
UNIT	SMU4
NAME	Vsb
MODE	V
SOURCE	0.0000 V
COMPLIANCE	100.00 mA

By this setup, Id-Vg characteristics can be measured. On the CHANNELS: USER FUNCTION DEFINITION page shown in the following table, we defined the square root of Id (SQRTId), and the differential coefficient (GRAD) of SQRTId versus Vg.

Also, we defined VTH and BETA by using Read Out Functions. VTH is @L1X (X-intercept of line 1) and BETA is @L1G^2 (slope of line 1 to second power). Line 1 is drawn according to DISPLAY: ANALYSIS SETUP page, which is described later.

NAME	UNIT	DEFINITION
SQRTId		SQRT(Id)
GRAD		DIFF(SQRTId,Vg)
VTH	V	@L1X
BETA		@L1G^2

As shown in the following tables, we set the DISPLAY: DISPLAY SETUP page to plot two curves: SQRTId versus Vg, and GRAD versus Vg. And VTH and BETA will be displayed in the data variables display area.

	X-axis	Y1-axis	Y2-axis
NAME	Vg	SQRTId	GRAD
SCALE	LINEAR	LINEAR	LINEAR
MIN	0 V	0	0
MAX	5 V	200 m	80 m

GRID	LINE PARAMETER	DATA VARIABLES
ON	ON	VTH
		BETA

The auto analysis functions are defined on DISPLAY: ANALYSIS SETUP page as shown below. A tangent line (line 1) is drawn to "SQRTId versus Vg" curve (Y1) at point where GRAD is maximum. VTH is the X-intercept of this line. Also, the marker is moved to point where GRAD is maximum.

```
*LINE1: [TANGENT] line on [Y1] at a point where
  [GRAD] = [MAX(GRAD)]
```

If you execute a single measurement, the two curves are drawn. Right after the measurement, a tangent line is drawn as specified in DISPLAY: ANALYSIS SETUP page, and resulting VTH and BETA values are displayed.

4

HP 4155A/4156A SCPI Programming

HP 4155A/4156A SCPI Programming

Standard Commands for Programmable Instruments (SCPI) is a universal programming language for electronic test and measurement instruments, and based on IEEE 488.1 and IEEE488.2.

This chapter describes how to create programs that contain SCPI commands to control the HP 4155A/56A. These programs can be run from an external computer or from the built-in HP Instrument BASIC (IBASIC) controller.

This chapter explains the following programming tasks along with programming examples:

- SCPI Programming Basics
- Measurement Setup
- Measurement Execution
- File Operation
- Data Transfer
- Print/Plot Operation
- Other Programming Tips
- Example for HP 4145 Users

If you are not familiar with HP 4155A/4156A programming, Chapter 3 provides step-by-step tutorials for programming and helps you to understand quickly.

In addition to this chapter, "Example Application Setup for Vth Measurement" in Chapter 3 provides some application examples which are helpful to increase your understanding.

SCPI Programming Basics

This section provides the following basic tasks to control and program the HP 4155A/4156A:

- Preparation before controlling the HP 4155A/56A via HP-IB
- SCPI Command Hierarchy
- To control HP 4155A/4156A by HP BASIC programming

Preparation before Controlling the HP 4155A/56A via HP-IB

SCPI programs to control the HP 4155A/4156A via HP-IB can be run from an external computer or from the built-in HP Instrument BASIC (IBASIC) controller.

To run these programs, the HP 4155A/56A must be set to *HP 4155A/56A command mode*, which is the default mode.

The HP 4155A/56A has two command modes: HP 4155A/56A command mode (default) and HP 4145 syntax command mode.

You use the HP 4145 syntax command mode if you need to execute HP 4145A/B programs on the HP 4155A/56A. For information about this, refer to Chapter 5.

Controlling from External Computer.
You must do as follows before controlling HP 4155A/4156A from an external computer:

1. Connect the HP-IB interface of external computer to HP-IB interface on rear panel of HP 4155A/4156A.
2. Set the `HP 4155A/56A is` field on the SYSTEM: MISCELLANEOUS page to NOT SYSTEM CONTROLLER.
3. Enter the HP-IB address of your HP 4155A/4156A into the `HP-IB ADDRESS` field.

Controlling from built-in IBASIC controller.
If you use built-in IBASIC controller to control HP 4155A/4156A, you do not need to prepare before controlling HP 4155A/4156A because built-in IBASIC controller is always connected to HP 4155A/4156A via internal HP-IB.

However, *to print/plot* from built-in IBASIC controller, set "`HP 4155A/56A is`" field on the SYSTEM: MISCELLANEOUS page to SYSTEM CONTROLLER.

SCPI Command Hierarchy

SCPI commands use a hierarchical structure for subsystem commands similar to the file system.

For example, in `:PAGE:MEASURE:SWEEP` command, the hierarchy is as follows:

`PAGE`	root
`MEASURE`	sub-level 1
`SWEEP`	sub-level 2

The colon at the beginning of the command means root.
The colons between two command keywords means moving down to a lower level.

Using a Semicolon to Reduce Typing

A semicolon enables two commands to be sent on the same line.

For example, `:PAGE:MEAS:VAR1:START 0;STOP 5` is the same as the following two commands:

```
:PAGE:MEAS:VAR1:START 0
:PAGE:MEAS:VAR1:STOP 5
```

So, using a semicolon reduces typing and simplifies the program.

A command terminator (such as a `<newline>`) resets the path to root.

To Control HP 4155A/56A by HP BASIC Programming

1. Assign I/O path for controlling HP 4155A/4156A.

   Use ASSIGN command to assign I/O path:

   - Built-in IBASIC

     Specify select code 8. For the HP-IB address, you can use *any* number between 0 to 31. Refer to the following example:

     ```
     10 ASSIGN @Hp4155 TO 800
     ```

   - HP BASIC on an external computer

     Specify the select code of the external computer. And specify the HP-IB address that you entered into the HP-IB ADDRESS field on the SYSTEM: MISCELLANEOUS page. In the following example, the select code of the external computer is 7 and HP-IB address of HP 4155A/56A is 17:

     ```
     10 ASSIGN @Hp4155 TO 717
     ```

2. Use OUTPUT command to send commands to HP 4155A/56A.
3. Use ENTER command to get query response from HP 4155A/56A.

Example

The following is the example program to control HP 4155A/4156A:

```
10    DIM I3(1:501)
20    !
30    ASSIGN @Hp4155 TO 717
40    !
50    OUTPUT @Hp4155;"*RST"
60    !
70    OUTPUT @Hp4155;":MMEM:LOAD:STAT 0,'SWP.MES','DISK'"
80    !
90    OUTPUT @Hp4155;":PAGE:SCON:SING"
100   OUTPUT @Hp4155;"*OPC?"
110   ENTER @Hp4155;Complete
120   !
130   OUTPUT @Hp4155;"FORM:DATA ASC"
140   OUTPUT @Hp4155;":DATA? 'I3'"
150   ENTER @Hp4155;I3(*)
160   !
170   END
```

Line Number	Description
30	Assigns I/O path to control HP 4155A/4156A.
50	Resets HP 4155A/4156A by sending *RST command.
70	Loads measurement setup data from diskette file `SWP.MES`.
90 to 110	Executes measurement
130 to 150	Gets the measurement data

Programming: Measurement Setup

To set up a measurement, you can use SCPI commands to set the setup pages of HP 4155A/4156A the same way that you can by interactive operation.

Basically, there are the following three ways to set up a measurement via SCPI programming:

- Load the measurement setup data from diskette or internal memory.

  Load the measurement setup data by SCPI programming. The data was previously defined and stored to a diskette file or internal memory interactively or by SCPI programming.

- Load the measurement setup data, then change some of the settings.

  Load the measurement setup data from diskette or internal memory, then change desired settings by SCPI programming.

- Set all settings.

  Set all settings for measurement setup by SCPI programming.

This section describes the following tasks:

- To set or change setup data values.
- To read setup data values

To load previously defined measurement setup data, refer to “Programming: File Operation”.

To Set or Change HP 4155/56 Setup Data Values

1. Send `:PAGE` subsystem commands that correspond to the setup data values that you want to change or set.

There is a command subsystem for each setup page as shown in Table 4-1. Each command subsystem has commands for setting the setup data of the corresponding setup page.

Table 4-1. :PAGE Subsystem Commands for Measurement Setup

Setup Page	Command Subsystem
CHANNELS: CHANNEL DEFINITION	:PAGE:CHANnels[:CDEFinition]
CHANNELS: USER FUNCTION DEFINITION	:PAGE:CHANnels:UFUNction
CHANNELS: USER VARIABLE DEFINITION	:PAGE:CHANnels:UVARiable
MEASURE: SWEEP SETUP	:PAGE:MEASure[:SWEep]
MEASURE: SAMPLING SETUP	:PAGE:MEASure:SAMPling
MEASURE: PGU SETUP	:PAGE:MEASure:PGUSetup
MEASURE: MEASURE SETUP	:PAGE:MEASure:MSETup
MEASURE: OUTPUT SEQUENCE	:PAGE:MEASure:OSEQuence
DISPLAY: DISPLAY SETUP	:PAGE:DISPlay[:SETup]
DISPLAY: ANALYSIS SETUP	:PAGE:DISPlay:ANALysis
STRESS: CHANNEL DEFINITION	:PAGE:STRess:[CDEFinition]
STRESS: STRESS SETUP	:PAGE:STRess:SETup

Example

To load measurement setup data, then change the sweep start and stop values:

```
10    ASSIGN @Hp4155 TO 800
20    !
30    OUTPUT @Hp4155;":MMEM:LOAD:STAT 0,'SWP.MES','DISK'"
40    !
50    Swp_start=1
60    Swp_stop=10
70    !
80    OUTPUT @Hp4155;":PAGE:MEAS:VAR1:STAR";Swp_start
90    OUTPUT @Hp4155;":PAGE:MEAS:VAR1:STOP";Swp_stop
100   !
110   END
```

Line Number	Description
10	Assigns I/O path to control HP 4155A/4156A.
30	Loads measurement setup data from diskette file `SWP.MES`.
80	Changes start value of VAR1.
90	Changes stop value of VAR1.

To Read HP 4155/56 Setup Data Values

To read setup data from HP 4155/56 into an IBASIC variable, do as follows:

1. Send `:PAGE` subsystem query command that corresponds to setup data that you want to read.

Example

To load measurement setup data, then read the sweep start and stop values:

```
10    ASSIGN @Hp4155 TO 717
20    !
30    OUTPUT @Hp4155;":MMEM:LOAD:STAT 0,'SWP.MES','DISK'"
40    !
50    OUTPUT @Hp4155;":PAGE:MEAS:VAR1:STAR?"
60    ENTER @Hp4155;Swp_start
70    OUTPUT @Hp4155;":PAGE:MEAS:VAR1:STOP?"
80    ENTER @Hp4155;Swp_stop
90    !
100   PRINT "Sweep-start=";Swp_start,"Sweep-stop=";Swp_stop
110   !
120   END
```

Line Number	Description
10	Assigns I/O path to control HP 4155A/4156A.
30	Loads measurement setup data from diskette file `SWP.MES`.
50 to 60	Reads start value of VAR1.
70 to 80	Reads stop value of VAR1.

Programming: Measurement Execution

To execute a measurement, you can use :PAGE:SCONtrol subsystem commands.

This section describes the following tasks:

- To execute a sweep or sampling measurement
- To force stress
- To start the knob sweep function
- To control standby units

To Execute a Sweep or Sampling Measurement

1. Send `:PAGE:SCONtrol[:MEASurement]:SINGle` command to HP 4155A/4156A.

- If you would like to repeat measurements, send `:PAGE:SCONtrol[:MEASurement]:REPeat` command instead of :PAGE:SCONtrol[:MEASurement]:SINGle command.
- If you would like to append measurement, send `:PAGE:SCONtrol[:MEASurement]:APPend` command instead of :PAGE:SCONtrol[:MEASurement]:SINGle command.

Example

Example 1.
To execute a sweep or sampling measurement after loading the measurement setup data:

```
10    ASSIGN @Hp4155 TO 717
20    !
30    OUTPUT @Hp4155;":MMEM:LOAD:STAT 0,'SWP.MES'"
40    OUTPUT @Hp4155;":PAGE:SCON:SING"
50    !
60    END
```

Line Number	Description
10	Assigns I/O path to control HP 4155A/4156A.
30	Loads measurement setup data from diskette file `SWP.MES`.
40	Executes measurement.

Example 2.
To load two measurement setups from diskette and store them into internal memory, then execute the measurements sequentially:

```
10    ASSIGN @Hp4155 TO 717
20    !
30    OUTPUT @Hp4155;":MMEM:LOAD:STAT 0,'MEAS1.MES','DISK'"
40    OUTPUT @Hp4155;":MMEM:STOR:STAT 0,'MEM1.MES','MEMORY'"
50    OUTPUT @Hp4155;":MMEM:LOAD:STAT 0,'MEAS2.MES','DISK'"
60    OUTPUT @Hp4155;":MMEM:STOR:STAT 0,'MEM2.MES','MEMORY'"
70    !
80    FOR I=1 TO 5
90      OUTPUT @Hp4155;":MMEM:LOAD:STAT 0,'MEM1.MES','MEMORY'"
100     OUTPUT @Hp4155;":PAGE:SCON:SING"
110     OUTPUT @Hp4155;"*OPC?"
120     ENTER @Hp4155;Complete
130     DISP "Analyze manually then press [Continue]"
140     PAUSE
150     !
160     OUTPUT @Hp4155;":MMEM:LOAD:STAT 0,'MEM2.MES','MEMORY'"
170     OUTPUT @Hp4155;":PAGE:SCON:SING"
180     OUTPUT @Hp4155;"*OPC?"
190     ENTER @Hp4155;Complete
200     DISP "Analyze manually and then press [Continue]"
210     PAUSE
220     !
230     IF I<5 THEN
240       DISP "Move to the next TEG and then press [Continue]"
250       PAUSE
260     END IF
270     !
280   NEXT I
290   !
300   END
```

Line Number	Description
10	Assigns I/O path to control HP 4155A/4156A.
30 to 60	Loads two measurement setups from diskette, then stores them into internal memory.
90 to 120	Executes first measurement, then waits for measurement completion.
160 to 190	Executes second measurement, then waits for measurement completion.

To Force Stress

1. Send `:PAGE:SCONtrol:STRess[:STARt]` command to HP 4155A/4156A.

Example

Example 1.
To force stress after loading the stress setup data:

```
10    ASSIGN @Hp4155 TO 717
20    !
30    OUTPUT @Hp4155;":MMEM:LOAD:STAT 0,'STRS.STR'"
40    OUTPUT @Hp4155;":PAGE:SCON:STR"
50    !
60    END
```

Line Number	Description
10	Assigns I/O path to control HP 4155A/4156A.
30	Loads stress setup data from diskette file `STRS.STR`.
40	Executes stress forcing.

Example 2.
To force stress, then execute sweep measurement:

```
10    ASSIGN @Hp4155 TO 717
20    !
30    OUTPUT @Hp4155;":MMEM:LOAD:STAT 0,'STRS.STR'"
40    OUTPUT @Hp4155;":PAGE:SCON:STR"
50    OUTPUT @Hp4155;"*OPC?"
60    ENTER @Hp4155;Complete
70    !
80    OUTPUT @Hp4155;":MMEM:LOAD:STAT 0,'SWP.MES'"
90    OUTPUT @Hp4155;":PAGE:SCON:SING"
100   !
110   END
```

Line Number	Description
10	Assigns I/O path to control HP 4155A/4156A.
30	Loads stress setup data from diskette file `STRS.STR`.
40	Executes stress forcing.
50 to 60	Waits until stress forcing is completed.
80	Loads measurement setup data from diskette file `SWP.MES`.
90	Executes sweep measurement.

To Start the Knob Sweep Function

1. Send `:PAGE:SCONtrol:KSWeep[:STARt]` command

Example

To start the knob sweep function:

```
10    ASSIGN @Hp4155 TO 717
20    !
30    OUTPUT @Hp4155;":MMEM:LOAD:STAT 0, 'MEM1.MES', 'MEMORY'"
40    OUTPUT @Hp4155;":PAGE:SCON:KSW"
50    !
60    END
```

Line Number	Description
10	Assigns I/O path to control HP 4155A/4156A.
30	Loads sweep setup data from internal memory file **MEM1**.
40	Starts knob sweep function.

To Control Standby Units

To change the standby units from the idle state to the standby state:

1. Send `:PAGE:SCONtrol:STANdby ON`.

   You cannot change which units are standby units after you execute this command. Standby units are units for which STBY is set to ON in the CHANNELS: CHANNELS DEFINITION page.

To change the standby units from the standby state to the idle state.

1. Send `:PAGE:SCONtrol:STANdby OFF` to stop standby units.

Example

To set standby units to standby state (so standby value will be output before and after measurements), then after final measurement, change standby units to idle state:

```
10    ASSIGN @Hp4155 TO 717
20    !
30    OUTPUT @Hp4155;":MMEM:LOAD:STAT 0,'SWP1.MES','DISK'"
40    OUTPUT @Hp4155;":PAGE:SCON:STAN ON"
50    OUTPUT @Hp4155;":PAGE:SCON:SING"
60    OUTPUT @Hp4155;"*OPC?"
70    ENTER @Hp4155;Complete
80    OUTPUT @Hp4155;":MMEM:STOR:TRAC DEF,'MEAS1.DAT','DISK'"
90    !
100   OUTPUT @Hp4155;":MMEM:LOAD:STAT 0,'SWP2.MES','DISK'"
110   OUTPUT @Hp4155;":PAGE:SCON:SING"
120   OUTPUT @Hp4155;"*OPC?"
130   ENTER @Hp4155;Complete
140   OUTPUT @Hp4155;":PAGE:SCON:STAN OFF"
150   OUTPUT @Hp4155;":MMEM:STOR:TRAC DEF,'MEAS2.DAT','DISK'"
160   END
```

Line Number	Description
10	Assigns I/O path to control HP 4155A/4156A.
30	Loads measurement setup data from diskette file `SWP1.MES`.
40	The standby units specified in setup data start to output the standby value.
50	Executes measurement.
60 to 70	Waits for completion of measurement. After measurement, standby units output the standby value.
80	Stores measurement data onto a diskette.
100	Loads another measurement setup data from diskette file `SWP2.MES`. This setup data cannot change which units are the standby units.
110	Executes measurement.
120 to 130	Waits for completion of measurement. After measurement, standby units output the standby value.
140	Standby units stop standby output and change to idle state.
150	Stores measurement data onto a diskette.

Programming: File Operation

This section describes how to use SCPI commands to move data to and from the internal memory or diskette memory of HP 4155A/4156A.

This section covers the following basic file operations:

- To store setup data to diskette or internal memory
- To store measurement data to diskette or internal memory
- To load setup data from diskette or internal memory
- To load measurement data from diskette or internal memory

To Store Setup Data to Diskette or Internal Memory

1. Send `:MMEMory:STORe:STATe` command to HP 4155A/4156A.
   a. Specify the first parameter to be 0. This parameter has no meaning for HP 4155A/4156A, but is necessary for SCPI compatibility.
   b. Specify the second parameter:
      - For diskette:

        File name with extension: `.MES` for measurement setup data or `.STR` for stress setup data.
      - For internal memory:

        Internal memory name (`MEM1`, `MEM2`, `MEM3`, or `MEM4`) with extension: `.MES` for measurement setup data or `.STR` for stress setup data.
   c. Specify the third parameter:
      - For diskette (default): `DISK`
      - For internal memory: `MEMORY`

Example

To store measurement setup data to a diskette file:

```
10    !
20    ASSIGN @Hp4155 TO 717
30    !
40    OUTPUT @Hp4155;":MMEM:STOR:STAT 0,'SWP.MES','DISK'"
50    !
60    END
```

Line Number	Description
20	Assigns I/O path to control HP 4155A/4156A.
40	Stores measurement setup data to diskette file `SWP.MES`.

To Store Measurement Data to Diskette or Internal Memory

1. Send `:MMEMory:STORe:TRACe` command to HP 4155A/4156A.
   a. Specify the first parameter to be `DEFault`. This parameter has no meaning for HP 4155A/4156A, but is necessary for SCPI compatibility.
   b. Specify the second parameter:
      - For diskette:

        File name with extension `.DAT`
      - For internal memory:

        Internal memory name (`MEM1`, `MEM2`, `MEM3`, or `MEM4`) with extension `.DAT`.
   c. Specify the third parameter:
      - For diskette (default): `DISK`
      - For internal memory: `MEMORY`

Example

To store measurement data to a diskette file:

```
10    !
20    ASSIGN @Hp4155 TO 717
30    !
40    OUTPUT @Hp4155;":MMEM:STOR:TRAC DEF,'SWP.DAT','DISK'"
50    !
60    END
```

Line Number	Description
20	Assigns I/O path to control HP 4155A/4156A.
40	Stores measurement data to diskette file `SWP.DAT`.

To Load Setup Data from Diskette or Internal Memory

1. Send `:MMEMory:LOAD:STATe` command to HP 4155A/4156A.
   a. Specify the first parameter to be 0. This parameter has no meaning for HP 4155A/4156A, but is necessary for SCPI compatibility.
   b. Specify the second parameter:
      - From diskette:

        File name with extension: `.MES` for measurement setup data or `.STR` for stress setup data.
      - From internal memory:

        Internal memory name (`MEM1`, `MEM2`, `MEM3`, or `MEM4`) with extension: `.MES` for measurement setup data or `.STR` for stress setup data.
   c. Specify the third parameter:
      - From diskette (default): `DISK`
      - From internal memory: `MEMORY`

Example

To load measurement setup data from a diskette file:

```
10    !
20    ASSIGN @Hp4155 TO 717
30    !
40    OUTPUT @Hp4155;":MMEM:LOAD:STAT 0,'SWP.MES','DISK'"
50    !
60    END
```

Line Number	Description
20	Assigns I/O path to control HP 4155A/4156A.
40	Loads measurement setup data from diskette file `SWP.MES`.

To Load Measurement Data from Diskette or Internal Memory

1. Send `:MMEMory:LOAD:TRACe` command to HP 4155A/4156A.
   a. Specify the first parameter to be `DEFault`. This file has no meaning for HP 4155A/4156A, but is necessary for SCPI compatibility.
   b. Specify the second parameter:
      - From diskette:

        File name with extension `.DAT`
      - From internal memory:

        Internal memory name (`MEM1`, `MEM2`, `MEM3`, or `MEM4`) with extension `.DAT`.
   c. Specify the third parameter:
      - From diskette (default): `DISK`
      - From internal memory: `MEMORY`

Example

To load measurement data from a diskette file:

```
10    !
20    ASSIGN @Hp4155 TO 717
30    !
40    OUTPUT @Hp4155;":MMEM:LOAD:TRAC DEF,'SWP.DAT','DISK'"
50    !
60    END
```

Line Number	Description
20	Assigns I/O path to control HP 4155A/4156A.
40	Loads measurement data from diskette file `SWP.DAT`.

Programming: Data Transfer

This section describes the data transfer between a program and HP 4155A/4156A.

The following programming tasks are described in this section:

- To read measurement data from HP 4155A/4156A
- To transfer data to HP 4155A/4156A

To Read HP 4155/56 Measurement Data

1. Send `:DATA?` query command to get data variable values (output data, measurement data, user function values) or read-out function values from HP 4155/56 to IBASIC variables.

Example

Example 1.
To get measurement data, then store it in a data array:

```
10    DIM I3(1:501)
20    !
30    ASSIGN @Hp4155 TO 717
40    !
50    OUTPUT @Hp4155;":FORM:DATA ASC"
60    !
70    OUTPUT @Hp4155;":DATA? 'I3'"
80    ENTER @Hp4155;I3(*)
90    !
100   END
```

Line Number	Description
30	Assigns I/O path to control HP 4155A/4156A.
50	Specifies ASCII data transfer format.
70 to 80	Gets the values of data variable `I3`.

Example 2.
To get slope of LINE1 for Y2 axis curve on GRAPH/LIST: GRAPHICS page:

```
10    ASSIGN @Hp4155 TO 717
20    !
30    OUTPUT @Hp4155;":DATA? '@L1G2'"
40    ENTER @Hp4155;Slope
50    !
60    PRINT Slope
70    END
```

Line Number	Description
10	Assigns I/O path to control HP 4155A/4156A.
30 to 40	Gets slope of LINE1 for Y2 axis curve on GRAPH/LIST: GRAPHICS page.

To Transfer Data to HP 4155A/4156A (Using User Variable)

To transfer a user variable to HP 4155A/4156A, use DATA|TRACe subsystem commands. A user variable consists of a name, unit, and numeric data.

Transferred user variable data can be used like other data variables in HP 4155A/4156A. You can perform calculations between measurement results and transferred data, plot transferred data on GRAPH/LIST: GRAPHICS page, or list transferred data on GRAPH/LIST: LIST page.

To transfer numeric data to the HP 4155A/4156A:

1. Define the data transfer format by using :FORMat[:DATA] command.
   - For ASCII data transfer format, send `:FORM ASC`.
   - For REAL 64-bit length data transfer format, send `:FORM REAL,64`.
   - For REAL 32-bit length data transfer format, send `:FORM REAL,32`.
2. For REAL data transfer format, define byte order by using :FORMat:BORDer command.
   - For normal order, send `:FORM:BORD NORM`.
   - For swapped order, send `:FORM:BORD SWAP`.
3. Define name of the user variable, unit (optional), and number of numeric data by using the :PAGE:CHANnels:UVARiable:DEFine command.

   You can also define these parameters by using the :DATA:DEFine and :DATA:UNIT command.

   If user variable is already defined, you do not have to perform this step.
4. Transfer data by using :DATA|:TRACe:[:DATA] command.

Example

Example 1.
To transfer data array by using ASCII data transfer format:

```
10    DIM Uvar1(1:5)
20    !
30    ASSIGN @Hp4155 TO 717
40    !
50    Uvar1(1)=1.0
60    Uvar1(2)=1.1
70    Uvar1(3)=1.2
80    Uvar1(4)=1.3
90    Uvar1(5)=1.4
100   !
110   OUTPUT @Hp4155;":FORM:DATA ASC"
120   OUTPUT @Hp4155;":PAGE:CHAN:UVAR:DEF 'UVAR1','V',5"
130   OUTPUT @Hp4155;":TRAC 'UVAR1',";Uvar1(*)
140   !
150   END
```

Line Number	Description
30	Assigns I/O path to control HP 4155A/4156A.
110	Specifies ASCII data transfer format.
120	Defines user variable.
130	Transfers user variable.

Example 2.
To transfer data array by using REAL 64-bit data transfer format:

```
10    DIM Uvar1(1:101)
20    INTEGER I
30    !
40    ASSIGN @Hp4155 TO 717
50    ASSIGN @Form_off TO 717;FORMAT OFF
60    !
70    FOR I=1 TO 101
80      Uvar1(I)=SQRT(I)
90    NEXT I
100   !
110   OUTPUT @Hp4155;":FORM REAL,64"
120   OUTPUT @Hp4155;":FORM:BORD NORM"
130   OUTPUT @Hp4155;":PAGE:CHAN:UVAR:DEF 'UVAR1','',101"
140   OUTPUT @Hp4155;":TRAC 'UVAR1',#0";
150   OUTPUT @Form_off;Uvar1(*),END
160   !
170   END
```

Line Number	Description
40	Assigns I/O path to control HP 4155A/4156A.
50	Assigns I/O path to transfer data.
110 to 120	Specifies REAL 64 bit data transfer format.
130	Defines a user variable.
140 to 150	Transfers user variable.

Example 3.
To transfer data, then display plot of transferred data and measurement results:

```
10    DIM Uvar1(1:101)
20    !
30    ASSIGN @Hp4155 TO 717
40    !
50    FOR I=1 TO 101
60      Uvar1(I)=SQRT(I)
70    NEXT I
80    !
90    OUTPUT @Hp4155;":MMEM:LOAD:STAT 0,'SWP.MES'"
100   OUTPUT @Hp4155;":PAGE:SCON:SING"
110   OUTPUT @Hp4155;"*OPC?"
120   ENTER @Hp4155;Complete
130   !
140   OUTPUT @Hp4155;":FORM ASC"
150   OUTPUT @Hp4155;":DATA:DEF 'UVAR1',101"
160   OUTPUT @Hp4155;":DATA:UNIT 'UVAR1','V'"
170   OUTPUT @Hp4155;":DATA 'UVAR1',";Uvar1(*)
180   !
190   OUTPUT @Hp4155;":PAGE:DISP:GRAP:Y2:NAME 'UVAR1'"
200   OUTPUT @Hp4155;":PAGE:GLIS"
210   END
```

Line Number	Description
30	Assigns I/O path to control HP 4155A/4156A.
90	Loads measurement setup data from diskette file `SWP.MES`.
100	Executes measurement.
110 to 120	Waits for measurement completion.
140	Specifies ASCII data transfer format.
150	Defines user variable.
160	Defines unit of user variable.
170	Transfers user variable.
190	Sets user variable to Y2 axis of graph.
200	Displays GRAPH/LIST: GRAPHICS page.

Programming: Print/Plot Operation

For the print/plot operation, you can use :HCOpy subsystem commands.

This section describes the following tasks:

- To output setup data to printer/plotter
- To output graphics result data to printer/plotter
- To output list result data to printer/plotter
- To dump screen image to printer/plotter
- To save hardcopy image to diskette

Before performing above tasks, the following print or plot settings must be set interactively or by remote commands.

We recommend that you save the following settings in a file, then load it before printing or plotting.

- Interface information

Interface	Setting Parameter	Command
HP-IB	printer address	:SYST:COMM:GPIB:RDEV:ADDR
Serial	baud rate [reception]	:SYST:COMM:SER:BAUD
	parity scheme [reception]	:SYST:COMM:SER:PAR
	stop bits [reception]	:SYST:COMM:SER:SBIT
	software pacing scheme [reception]	:SYST:COMM:SER:PACE
	baud rate [transmission]	:SYST:COMM:SER:TRAN:BAUD
	parity scheme [transmission]	:SYST:COMM:SER:TRAN:PAR
	stop bits [transmission]	:SYST:COMM:SER:TRAN:SBIT
	software pacing scheme [transmission]	:SYST:COMM:SER:TRAN:PACE

- Printer information

Setting Parameter	Command
color or not	:HCOP:DEV:COL
control language	:HCOP:DEV:LANG
resolution (PCL)	:HCOP:DEV:RES
destination	:HCOP:DEST

- Output Items

Item	Command
Title of the print or plot out	:HCOP:ITEM:ANN:STAT
User defined comment for page group	:HCOP:ITEM:ANN2:STAT
Present date and time of the built-in clock	:HCOP:ITEM:TDST:STAT
Page number of the print or plot out	:HCOP:ITEM:PNUM:STAT
User defined comment for print or plot out	:HCOP:ITEM:LAB:STAT
Graphics plot curve	:HCOP:ITEM:TRAC:STAT
Frame and grid	:HCOP:ITEM:TRAC:GRAT:STAT
Marker, cursor, and data variable coordinate fields, and line parameters (gradients and intercepts)	:HCOP:ITEM:TEXT:STAT
Names, units, and scale of the graph axis	:HCOP:ITEM:TEXT2:STAT

- For built-in IBASIC only

  To print from built-in IBASIC, set "`HP 4155A/56A is`" field on the SYSTEM: MISCELLANEOUS page to SYSTEM CONTROLLER.

To Output Setup Data to Printer/Plotter

1. If you want to output print/plot comment, enter comment by using :HCOPy:ITEM:LABel:TEXT command.
2. Specify the range of setup data to print/plot by sending `:HCOPy:OPAGe` command.
   - To print/plot present *page* setup data, send `:HCOPy:OPAGe CURRent`
   - To print/plot present *page group* setup data, send `:HCOPy:OPAGe GROup`
   - To print/plot *all* setup data, send `:HCOPy:OPAGe ALL`
3. Display the page that you want to print/plot by using the appropriate command:

Page	Command
CHANNELS: CHANNEL DEFINITION	:PAGE:CHAN
CHANNELS: USER FUNCTION DEFINITION	:PAGE:CHAN:UFUN
CHANNELS: USER VARIABLE DEFINITION	:PAGE:CHAN:UVAR
MEASURE: SWEEP SETUP	:PAGE:MEAS
MEASURE: SAMPLING SETUP	:PAGE:MEAS:SAMP
MEASURE: PGU SETUP	:PAGE:MEAS:PGUS
MEASURE: MEASURE SETUP	:PAGE:MEAS:MSET
MEASURE: OUTPUT SEQUENCE	:PAGE:MEAS:OSEQ
DISPLAY: DISPLAY SETUP	:PAGE:DISP
DISPLAY: ANALYSIS SETUP	:PAGE:DISP:ANAL
STRESS: CHANNEL DEFINITION	:PAGE:STR
STRESS: STRESS SETUP	:PAGE:STR:SET
STRESS: STRESS FORCE	:PAGE:STR:FORC

If you print/plot from built-in IBASIC, change display mode to All Instrument or IBASIC Status by sending :DISPlay[:WINDow]:ALLocation command.

4. Print/plot the setup data by sending `:HCOPy` command.

   If you print/plot from an external computer, pass Active Controller capability to HP 4155A/4156A after sending `:HCOPy` command because HP 4155A/4156A requires Active Controller capability to print.

   Refer to the following example.

Example

The following two examples load a sweep setup file, then print setup data of the MEASURE: SWEEP SETUP page:

Example 1.
From an external computer:

```
10    ASSIGN @Hp4155 TO 717
20    CONTROL 7,3;21
30    !
40    OUTPUT @Hp4155;"*RST"
50    OUTPUT @Hp4155;"*PCB 21"
60    !
70    OUTPUT @Hp4155;":MMEM:LOAD:STAT 0,'SWP.MES'"
80    OUTPUT @Hp4155;":HCOP:ITEM:PNUM:STAT OFF"
90    OUTPUT @Hp4155;":HCOP:ITEM:LAB:TEXT 'This is an example'"
100   OUTPUT @Hp4155;":HCOP:DEST RDEV"
110   OUTPUT @Hp4155;":HCOP:OPAG CURR"
120   !
130   OUTPUT @Hp4155;":PAGE:MEAS"
140   !
150   OUTPUT @Hp4155;":HCOP"
160   REPEAT
170     OUTPUT @Hp4155;"*ESR?"
180     ENTER @Hp4155;Event_status
190   UNTIL BIT(Event_status,1)
200   !
210   PASS CONTROL @Hp4155
220   DISP "Printing"
230   REPEAT
240     STATUS 7,6;Hpib_status
250   UNTIL BIT(Hpib_status,6)
260   DISP "Done"
270   END
```

Line Number	Description
10	Assigns I/O path to control HP 4155A/4156A from external computer.
20	Sets the HP-IB address of external computer. This will be necessary to return Active Controller capability from HP 4155A/4156A back to the external computer.
40	Resets HP 4155A/4156A
50	Specifies to pass Active Controller capability back to external computer after printing is completed.
70	Loads measurement setup data from diskette file `SWP.MES`.
80	Specifies to not print the page number.
90	Defines a print/plot comment.
100	Selects HP-IB interface. If serial interface, change the parameter to "SER".
110 to 130	Specifies to print/plot the setup data of the MEASURE: SWEEP SETUP page.
150 to 190	Sends print command and waits for Active Controller request from HP 4155A/4156A.
210	Passes Active Controller capability to HP 4155A/4156A, then HP 4155A/4156A starts printing.
230 to 250	Waits until printing is complete.

Example 2.
From built-in IBASIC:

```
10    ASSIGN @Hp4155 TO 800
20    !
30    OUTPUT @Hp4155;"*RST"
40    !
50    OUTPUT @Hp4155;":MMEM:LOAD:STAT 0,'SWP.MES','DISK'"
60    OUTPUT @Hp4155;":HCOP:ITEM:PNUM:STAT OFF"
70    OUTPUT @Hp4155;":HCOP:ITEM:LAB:TEXT 'This is an example'"
80    OUTPUT @Hp4155;":HCOP:DEST RDEV"
90    OUTPUT @Hp4155;":HCOP:OPAG CURR"
100   !
110   OUTPUT @hp4155;":DISP:ALL INST"
110   OUTPUT @Hp4155;":PAGE:MEAS"
120   !
130   OUTPUT @Hp4155;":HCOP"
140   DISP "Printing"
150   OUTPUT @Hp4155;"*OPC?"
160   ENTER @Hp4155;Complete
170   DISP "Done"
180   !
190   END
```

Line Number	Description
10	Assigns I/O path to control HP 4155A/4156A from built-in IBASIC.
30	Resets HP 4155A/4156A
50	Loads measurement setup data from diskette file `SWP.MES`.
60	Specifies to not print the page number.
70	Defines a print/plot comment.
80	Selects HP-IB interface. If serial interface, change the parameter to "SER".
90 to 110	Specifies to print/plot the setup data of the MEASURE: SWEEP SETUP page.
130	Starts printing.
150 and 160	Waits until printing is complete.

To Output Graphics Result Data to Printer/Plotter

1. If you want to output print/plot comment, enter comment by using :HCOPy:ITEM:LABel:TEXT command.
2. Display GRAPH/LIST: GRAPHICS page by using :PAGE:GLISt:[:GRAPhics] command.

   If you print/plot from built-in IBASIC, change display mode to All Instrument or IBASIC Status display mode by sending :DISPlay[:WINDow]:ALLocation command.
3. Execute print/plot by using :HCOPy command.

   If you print/plot from an external computer, pass Active Controller capability to HP 4155A/4156A after sending `:HCOPy` command because HP 4155A/4156A requires Active Controller capability to print.

   Refer to the following example.

Example

The following two examples load a sweep setup file, execute measurement, then print measurement results of GRAPH/LIST: GRAPHICS page:

Example 1.
From an external computer:

```
10    ASSIGN @Hp4155 TO 717
20    CONTROL 7,3;21
30    !
40    OUTPUT @Hp4155;"*RST"
50    OUTPUT @Hp4155;"*PCB 21"
60    !
70    OUTPUT @Hp4155;":MMEM:LOAD:STAT 0,'SWP.MES'"
80    !
90    OUTPUT @Hp4155;":PAGE:SCON:SING"
100   OUTPUT @Hp4155;"*OPC?"
110   ENTER @Hp4155;Complete
120   !
130   OUTPUT @Hp4155;":HCOP:DEST RDEV"
140   !
150   OUTPUT @Hp4155;":PAGE:GLIS"
160   !
170   OUTPUT @Hp4155;":HCOP"
180   REPEAT
190     OUTPUT @Hp4155;"*ESR?"
```

```
200     ENTER @Hp4155;Event_status
210   UNTIL BIT(Event_status,1)
220   !
230   PASS CONTROL @Hp4155
240   DISP "Printing"
250   REPEAT
260     STATUS 7,6;Hpib_status
270   UNTIL BIT(Hpib_status,6)
280   DISP "Done"
290   END
```

Line Number	Description
10	Assigns I/O path to control HP 4155A/4156A from external computer.
20	Sets the HP-IB address of external computer. This will be necessary to return Active Controller capability from HP 4155A/4156A back to the external computer.
40	Resets HP 4155A/4156A
50	Specifies to pass Active Controller capability back to external computer after printing is completed.
70	Loads measurement setup data from diskette file `SWP.MES`.
90 to 110	Executes measurement and waits until completed.
130	Selects HP-IB interface. If serial interface, change the parameter to "SER".
150	Changes page to GRAPH/LIST: GRAPHICS page.
170 to 210	Sends print command and waits for Active Controller request from HP 4155A/4156A.
230	Passes Active Controller capability to HP 4155A/4156A, then HP 4155A/4156A starts printing.
250 to 270	Waits until printing is complete.

Example 2.
From built-in IBASIC:

```
10    ASSIGN @Hp4155 TO 800
20    !
30    OUTPUT @Hp4155;"*RST"
40    !
50    OUTPUT @Hp4155;":MMEM:LOAD:STAT 0,'SWP.MES'"
60    !
70    OUTPUT @Hp4155;":PAGE:SCON:SING"
80    OUTPUT @Hp4155;"*OPC?"
90    ENTER @Hp4155;Complete
100   !
110   OUTPUT @Hp4155;":HCOP:DEST RDEV"
120   !
130   OUTPUT @hp4155;":DISP:ALL INST"
140   OUTPUT @Hp4155;":PAGE:GLIS"
150   !
160   OUTPUT @Hp4155;":HCOP"
170   DISP "Printing"
180   OUTPUT @Hp4155;"*OPC?"
190   ENTER @Hp4155;Complete
200   DISP "Done"
210   END
```

Line Number	Description
10	Assigns I/O path to control HP 4155A/4156A from built-in IBASIC.
30	Resets HP 4155A/4156A
50	Loads measurement setup data from diskette file `SWP.MES`.
70 to 90	Executes measurement and waits until complete.
110	Selects HP-IB interface. If serial interface, change the parameter to "SER".
130 to 140	Changes page to GRAPH/LIST: GRAPHICS page.
160 to 190	Starts printing and waits until completion.

To Output List Results Data to Printer/Plotter

1. Specify the range of measurement results to output by using :HCOPy:LINDex command.
2. If you want to output print/plot comment, enter comment by using :HCOPy:ITEM:LABel:TEXT command.
3. Display GRAPH/LIST: GRAPHICS page by using :PAGE:GLISt:LIST command.

   If you print/plot from built-in IBASIC, change display mode to All Instrument or IBASIC Status display mode by sending :DISPlay[:WINDow]:ALLocation command.
4. Execute print/plot by using :HCOPy command.

   If you print/plot from an external computer, pass Active Controller capability to HP 4155A/4156A after sending `:HCOPy` command because HP 4155A/4156A requires Active Controller capability to print.

   Refer to the following example.

Example

The following two examples load a sweep setup file, execute measurement, then print measurement results of GRAPH/LIST: LIST page:

Example 1.
From an external computer:

```
10    ASSIGN @Hp4155 TO 717
20    CONTROL 7,3;21
30    !
40    OUTPUT @Hp4155;"*RST"
50    OUTPUT @Hp4155;"*PCB 21"
60    !
70    OUTPUT @Hp4155;":MMEM:LOAD:STAT 0,'SWP.MES'"
80    !
90    OUTPUT @Hp4155;":PAGE:SCON:SING"
100   OUTPUT @Hp4155;"*OPC?"
110   ENTER @Hp4155;Complete
120   !
130   OUTPUT @Hp4155;":HCOP:LIND MAX"
140   !
150   OUTPUT @Hp4155;":PAGE:GLIS:LIST"
160   !
```

```
170   OUTPUT @Hp4155;":HCOP"
180   REPEAT
190     OUTPUT @Hp4155;"*ESR?"
200     ENTER @Hp4155;Event_status
210   UNTIL BIT(Event_status,1)
220   !
230   PASS CONTROL @Hp4155
240   DISP "Printing"
250   REPEAT
260     STATUS 7,6;Hpib_status
270   UNTIL BIT(Hpib_status,6)
280   DISP "Done"
290   END
```

Line Number	Description
10	Assigns I/O path to control HP 4155A/4156A from external computer.
20	Sets the HP-IB address of external computer. This will be necessary to return Active Controller capability from HP 4155A/4156A back to the external computer.
40	Resets HP 4155A/4156A
50	Specifies to pass Active Controller capability back to external computer after printing is completed.
70	Loads measurement setup data from diskette file `SWP.MES`.
90 to 110	Executes measurement and waits until completed.
130	Sets the range of list results to be output.
150	Changes page to GRAPH/LIST: LIST page.
170 to 210	Sends print command and waits for Active Controller request from HP 4155A/4156A.
230	Passes Active Controller capability to HP 4155A/4156A, then HP 4155A/4156A starts printing.
250 and 270	Waits until completion of printing.

Example 2.
From built-in IBASIC:

```
10    ASSIGN @Hp4155 TO 800
20    !
30    OUTPUT @Hp4155;"*RST"
40    !
50    OUTPUT @Hp4155;":MMEM:LOAD:STAT 0,'SWP.MES'"
60    !
70    OUTPUT @Hp4155;":PAGE:SCON:SING"
80    OUTPUT @Hp4155;"*OPC?"
90    ENTER @Hp4155;Complete
100   !
110   OUTPUT @Hp4155;":HCOP:LIND MAX"
120   !
130   OUTPUT @Hp4155;":DISP:ALL INST"
140   OUTPUT @Hp4155;":PAGE:GLIS:LIST"
150   !
160   OUTPUT @Hp4155;":HCOP"
170   DISP "Printing"
180   OUTPUT @Hp4155;"*OPC?"
190   ENTER @Hp4155;Complete
200   DISP "Done"
210   END
```

Line Number	Description
10	Assigns I/O path to control HP 4155A/4156A from built-in IBASIC.
30	Resets HP 4155A/4156A
50	Loads measurement setup data from diskette file `SWP.MES`.
70 to 90	Executes measurement and waits until completion.
110	Sets the range of list results to be output.
130 to 140	Changes page to GRAPH/LIST: LIST page.
160 to 190	Starts printing and waits until completion.

To Dump Screen Image to Printer/Plotter

1. Display the page to be dumped.
2. Execute print/plot by using :HCOPy:SDUMp command.

   If you print/plot from an external computer, pass Active Controller capability to HP 4155A/4156A after sending `:HCOPy:SDUMp` command because HP 4155A/4156A requires Active Controller capability to print.

   Refer to the following example.

Example

The following two examples load a sweep setup file, execute measurement, display GRAPH/LIST: GRAPHICS page, then dump screen image of GRAPH/LIST: GRAPHICS page to printer/plotter:

Example 1.
From an external computer:

```
10    ASSIGN @Hp4155 TO 717
20    CONTROL 7,3;21
30    !
40    OUTPUT @Hp4155;"*RST"
50    OUTPUT @Hp4155;"*PCB 21"
60    !
70    OUTPUT @Hp4155;":MMEM:LOAD:STAT 0,'SWP.MES'"
80    !
90    OUTPUT @Hp4155;":PAGE:SCON:SING"
100   OUTPUT @Hp4155;"*OPC?"
110   ENTER @Hp4155;Complete
120   !
130   OUTPUT @Hp4155;":HCOP:DEST RDEV"
140   !
150   OUTPUT @Hp4155;":PAGE:GLIS"
160   !
170   OUTPUT @Hp4155;":HCOP:SDUM"
180   REPEAT
190     OUTPUT @Hp4155;"*ESR?"
200     ENTER @Hp4155;Event_status
210   UNTIL BIT(Event_status,1)
220   !
230   PASS CONTROL @Hp4155
240   DISP "Printing"
250   REPEAT
260     STATUS 7,6;Hpib_status
```

```
270   UNTIL BIT(Hpib_status,6)
280   DISP "Done"
290   END
```

Line Number	Description
10	Assigns I/O path to control HP 4155A/4156A from external computer.
20	Sets the HP-IB address of external computer. This will be necessary to return Active Controller capability from HP 4155A/4156A back to the computer.
40	Resets HP 4155A/4156A
50	Specifies to pass Active Controller capability back to external computer after printing is completed.
70	Loads measurement setup data from diskette file `SWP.MES`.
90 to 110	Executes measurement and waits until completed.
130	Selects HP-IB interface. If serial interface, change the parameter to "SER".
150	Changes page to GRAPH/LIST: GRAPHICS page.
170 to 210	Sends screen dump command and waits for Active Controller request from HP 4155A/4156A.
230	Passes Active Controller capability to HP 4155A/4156A, then HP 4155A/4156A starts printing.
250 and 270	Waits until printing is complete.

Example 2.
From built-in IBASIC:

```
10    ASSIGN @Hp4155 TO 800
20    !
30    OUTPUT @Hp4155;"*RST"
40    !
50    OUTPUT @Hp4155;":MMEM:LOAD:STAT 0,'SWP.MES'"
60    !
70    OUTPUT @Hp4155;":PAGE:SCON:SING"
80    OUTPUT @Hp4155;"*OPC?"
90    ENTER @Hp4155;Complete
100   !
110   OUTPUT @Hp4155;":DISP:ALL INST"
120   OUTPUT @Hp4155;":PAGE:GLIS"
130   !
140   OUTPUT @Hp4155;":HCOP:SDUM"
150   OUTPUT @Hp4155;"*OPC?"
160   ENTER @Hp4155;Complete
170   END
```

Line Number	Description
10	Assigns I/O path to control HP 4155A/4156A from built-in IBASIC.
30	Resets HP 4155A/4156A.
50	Loads measurement setup data from diskette file `SWP.MES`.
70 to 90	Executes measurement and waits until completion.
110 to 120	Displays GRAPH/LIST: GRAPHICS page.
140 to 160	Starts printing and waits until completion.

To Save Hardcopy Image to Diskette

1. To set print/plot destination to diskette file, send :HCOPy:DESTination command with MMEMory parameter.
2. Specify the file name by using :MMEMory:NAME command.
3. Execute the print/plot operation. Refer to print/plot tasks described previously.

Example

To load sweep setup file, execute measurement, and then saves a hardcopy image of the measurement results of GRAPH/LIST: GRAPHICS page to a diskette:

```
10    ASSIGN @Hp4155 TO 717
20    !
30    OUTPUT @Hp4155;"*RST"
40    !
50    OUTPUT @Hp4155;":MMEM:LOAD:STAT 0,'SWP.MES'"
60    !
70    OUTPUT @Hp4155;":PAGE:SCON:SING"
80    OUTPUT @Hp4155;"*OPC?"
90    ENTER @Hp4155;Complete
100   !
110   OUTPUT @Hp4155;":HCOP:DEST MMEM"
120   OUTPUT @Hp4155;":MMEM:NAME 'TEST1'"
130   !
140   OUTPUT @Hp4155;":PAGE:GLIS"
150   !
160   OUTPUT @Hp4155;":HCOP"
170   OUTPUT @Hp4155;"*OPC?"
180   ENTER @Hp4155;Complete
190   END
```

Line Number	Description
10	Assigns I/O path to control HP 4155A/4156A.
30	Resets HP 4155A/4156A
50	Loads measurement setup data from diskette file `SWP.MES`.
70 to 90	Executes measurement and waits until completion.
110	Specifies the destination to be diskette.
120	Specifies the diskette file name.
140	Displays GRAPH/LIST: GRAPHICS page.
160 to 180	Starts printing and waits until completion.

Other Programming Tips

This section provides the advanced programming techniques and useful tips:

- Speed Improvement
- Auto-loading of Files

Disabling Instrument Screen Update to Improve Speed

Most of the commands that control and set the HP 4155A/56A will also update the instrument screen.

For example, :PAGE:CHAN:MODE command changes the measurement mode. This command also changes the instrument screen to the CHANNELS: CHANNEL DEFINITION page and updates the MEASUREMENT MODE field setting.

This instrument screen update is useful for confirming the settings that were changed by the commands, but it takes time.

You can enable or disable this time consuming instrument screen update as follows:

:DISP OFF	Instrument screen is not updated
:DISP ON	Instrument screen is updated

Auto-loading of Files

The HP 4155A/56A can automatically load files when it is turned on.

INIT files for Initial Settings.
If any setup files named `INIT.MES`, `INIT.STR`, `INIT.CST`, or `INIT.DAT` are on the diskette (in the built-in drive) when the HP 4155A/56A is turned on, the HP 4155A/56A automatically loads these setup files to be the initial settings.

This function saves you the trouble of getting application files every time you turn on the HP 4155A/56A.

> **INIT.MES and INIT.DAT files**
>
> `INIT.MES` and `INIT.DAT` both contain measurement setup data. If both these files exist on the diskette, the HP 4155A/56A gets `INIT.DAT`, not `INIT.MES`.

MEM*no* Files.
If any files named `MEM`*no*`.DAT`, `MEM`*no*`.MES`, or `MEM`*no*`.STR` are on the diskette in the drive, the files are automatically loaded from diskette to internal memory when HP 4155A/56A is turned on. Where `MEM`*no* means `MEM1`, `MEM2`, `MEM3`, or `MEM4`, which correspond to the four internal memory areas.

If the same internal memory is specified by multiple files (for example, `MEM1.MES` and `MEM1.DAT`), the priority is as follows:

1. `DAT`
2. `MES`
3. `STR`

IBASIC Program File to Auto-execute.
If an IBASIC program is stored in a file named "AUTOST" on the diskette in the built-in drive, the program is automatically loaded and started when you turn on the HP 4155A/56A.

Programming Example for HP 4145 Users

This section shows a programming example with SCPI commands that performs the same operations as the desired HP 4145 ASP program.

Built-in IBASIC can execute ASP-like commands for controlling the HP 4155A/4156A. Refer to "Creating ASP-like IBASIC Programs" in Chapter 1 on programming this commands.

Following program is the simplest example of creating an HP BASIC program (with SCPI commands) that performs the same operations as the desired HP 4145 ASP program. The ASP program gets a setup file named "VTH" from the diskette, makes a single measurement, then saves measurement to a file named "VTH1".

```
                          1 ASSIGN @Hp415x TO 800
1  GET P    VTH           10 OUTPUT @Hp415x;":MMEM:LOAD:STAT 0,'VTH.PRO'"
2  SINGLE                 20 OUTPUT @Hp415x;":PAGE:SCON:SING"
                          30 OUTPUT @Hp415x;"*OPC?"
                          40 ENTER @Hp415x;Complete
3  SAVE D   VTH1          50 OUTPUT @Hp415x;":MMEM:STOR:TRAC DEF,'VTH1.DAT'"
                          60 END
```

The above HP BASIC program (with SCPI commands) does as follows:

- Line 1 assigns a path named `@Hp415x` to `800`, which is the select code/HP-IB address to use if this is an IBASIC program running in the HP 4155A/56A. If this program will run on an external computer, use the select code of the HP-IB interface and the HP-IB address of the HP 4155A/4156A instead.
- Line 10 gets a measurement setup file named "VTH.MES". So, you need to save setup data to a file named "VTH.MES" on the diskette before executing this program. For an example setup, see "Example Application Setup for Vth Measurement" in Chapter 3.
- Line 20 performs a single measurement.
- Line 50 saves measurement setup and result data to a file named `VTH1.DAT`.

For built-in help function, which makes it easier to enter the desired SCPI command, see the "To Use the Help Function" in Chapter 1.

Following shows HP 4145A/B's ASP keywords and corresponding SCPI commands of HP 4155A/4156A:

Table 4-2. Corresponding HP 4145 ASP and HP 4155A/56A SCPI Commands

4145A/B	SCPI Commands	Function	Remark
GET P	:MMEM:LOAD:STAT	Gets setup `.MES` or `.PRO` file	
SINGLE	:PAGE:SCON:SING	Initiates single measurement	
SAVE D	:MMEM:STOR:TRAC	Saves data to `.DAT` file	
PLOT	:HCOP	Prints/plots present instrument page.	
CPLOT	:HCOP:ITEM:TRAC	Prints/plots measurement graph.	
PRINT	:HCOP	Prints/plots present instrument page.	
PAUSE			Use BASIC keyword PAUSE
WAIT			Use BASIC keyword WAIT
PAGE			Set in the Print/Plot setup

5

Running HP 4145A/B Program Directly on HP 4155A/4156A

Running HP 4145A/B Program Directly on HP 4155A/4156A

This chapter describes how to directly run an HP 4145A/B HP-IB program (non-ASP program) on the HP 4155A/4156A with little or no modification. To run these programs directly, you need to use the *HP 4145 syntax command mode* of the HP 4155A/4156A.

To Enter into HP 4145 Syntax Command Mode

When HP 4155A/4156A is turned on, HP 4155A/4156A is always in HP 4155A/4156A command mode.

To enter into HP 4145 syntax command mode:

- From front-panel

  Set `COMMAND SET` field on the SYSTEM: MISCELLANEOUS page to `HP4145`.

- From remote control

  Send ":SYSTem:LANGuage COMPatibility" command to HP 4155A/4156A.

Usually, you can run these programs with no modification. But sometimes small modifications are required due to the following, which are described in this chapter:

- Non-supported commands
- Consideration about Differences

Non-supported Commands

The following HP 4145A/B commands are not supported in HP 4145B syntax command mode:

`GL0`	Disables HP-GL
`GL1`	Enables HP-GL overlay graphics
`GL2`	Enables HP-GL stand-alone graphics
`MX`	Matrix
`SH`	Schmoo
`SV S`	Save ASP file
`GT S`	Get ASP file
`DM3`	Display mode Matrix
`DM4`	Display mode Schmoo
`AS1`	Auto Sequence Program Start
`AS2`	Auto Sequence Program Continue
`AS3`	Auto Sequence Program Stop

If you have HP 4145A/B programs that include any of the above commands, they will not work with the HP 4155A/4156A. Please refer to Chapter 2 in *HP-IB Command Reference* of HP 4155A/4156A for details.

Considerations about Differences

Spot Measurement

HP 4145A/B can execute a spot measurement by setting both start and stop of the sweep to the same value, but the HP 4155A/4156A executes the measurement twice even if you set both start and stop of the sweep to the same value.

Sweep Steps in Logarithmic Step Mode

Calculation algorithm for primary sweep steps in logarithmic step mode is slightly different between HP 4155A/4156A and HP 4145A/B, so step values and number of steps may be different between HP 4155A/4156A and HP 4145A/B.

Terminator

If you run your program on an external controller, use <CR>+<LF> as the command terminator if you execute serial polling to read a status of HP 4155A/4156A in your program.

If you use only <CR> or <LF> as command terminator, HP 4155A/4156A may respond with incorrect status.

This is due to the differences of reading and parsing commands between HP 4145A/B and HP 4155A/4156A.

The following example and explanation gives a better understanding of this.

```
10 OUTPUT @Hp415x;"ME1"
20 REPEAT
30 Status=SPOLL(@Hp415x)
40 UNTIL BIT(Status,0)
```

line number	Description
10	triggers measurement and clears the data ready bit (bit1) of status register.
20 to 40	waits until the data ready bit of status register is set to 1.

- When the Terminator is only <CR>
  - HP 4145A/B

    At line 10:

    1. HP 4145A/B starts reading data with RFD line set to false (data bus is halted) after each byte.

       In this example:

       `M ⇒ bus halted ⇒ E ⇒ bus halted ⇒ 1 ⇒ bus halted`

    2. After receiving `1`, HP 4145A/B recognizes valid command `ME1`, then executes `ME1`.

       At this time, the program is paused because the controller is trying to send <CR>, which is a terminator, but HP 4145A/B has halted data bus and does not receive <CR>.

3. After HP 4145A/B triggers measurement and clears status bit1, HP 4145A/B reads <CR>, then the program proceeds to next step (line 20).

The program reads the correct status at line 30.

□ HP 4155A/4156A

1. At line 10:

   a. HP 4155A/4156A starts and continues reading data until reading a terminator.

      In this example, HP 4155A/4156A reads `ME1<CR>`, then halts data bus.

   b. HP 4155A/4156A starts executing "ME1". At the same time, the external controller can proceed to the next line, because all data of this line has transferred, then program continues.

2. At line 30, controller can read status of HP 4155A/4156A even if RFD line is false. RFD holdoff is not effective for serial polling.

   However, the clearing of the status register bit by line 10 may not have been completed yet, so line 30 may get the incorrect status.

- When the Terminator is <CR>+<LF>

The example program for HP 4155A/4156A performs as follows:

1. HP 4155A/4156A starts and continues reading data until reading a terminator.

   In this example, HP 4155A/4156A reads `ME1<CR>`, then halts data bus.

2. HP 4155A/4156A executes "ME1".

   At this time, the program is paused because the controller is trying to send <LF>, which is part of the terminator, but HP 4155A/4156A has halted data bus and does not receive <LF>.

3. After HP 4155A/4156A triggers measurement and clears the status bit1, HP 4155A/4156A reads <LF>, then the program proceeds to next step (line 20).

The program reads the correct status at line 30.

6

Sample Application Programs

Sample Application Programs

This chapter describes some sample application programs and setup files, which will be helpful for creating your own applications.

All programs and setup files described in this chapter are stored on a DOS formatted 3.5-inch diskette that is provided with your HP 4155A/4156A. You should copy this diskette to a diskette that you will use as your working diskette. The 3.5-inch diskette includes eight programs. This chapter provides only the following three examples.

- Flash EEPROM Test
- TDDB
- Electromigration

See *Sample Application Programs' Guide Book* if you want to use the other programs on the 3.5-inch diskette. All programs are written in HP Instrument BASIC and ready to run in HP 4155A/4156A's HP Instrument BASIC environment.

CAUTION

These programs are only examples, so you may need to modify these programs and setup files for your own application before executing. If these example programs damage your devices, Hewlett-Packard is *NOT LIABLE* for the damage.

You can modify setup files by remote programming or interactively by front-panel keys.

Flash EEPROM Test

This program forces write and erase pulses, then measures Vth shift.

	name
Program	NOR_TEST
Setup files	ROMVTH.MES, NORWRT.STR, NORERS.STR, NANWRT.STR, NANERS.MES

This program uses NORWRT.STR and NORERS.STR stress setup files for write and erase pulses. These setup files are for NOR type flash EEPROM.

To use this program for NAND type flash EEPROM, please modify as follows to use NANWRT.STR and NANERS.STR stress setup files:

- Modify the following two lines:

```
1990    Wrt_file$="NORWRT.STR"     ! Write Stress Setup File Name
2000    Ers_file$="NORERS.STR"     ! Erase Stress Setup File Name
```

  as follows:

```
1990    Wrt_file$="NANWRT.STR"     ! Write Stress Setup File Name
2000    Ers_file$="NANERS.STR"     ! Erase Stress Setup File Name
```

Program Overview

Device connections for NOR and NAND type flash EEPROM are different.

Device Connection for NOR type flash EEPROM.
As shown in Figure 6-1, one HP 16440A SMU/Pulse Generator Selector is used to switch units for forcing write pulse and erase pulse, and measuring Vth.



Figure 6-1. Device Connection (NOR Type)

The following table shows the selector's state for each phase:

Table 6-1. Selector's State in Each Phase

Selector Channel	Write	Erase	Vth Measure
CH1 |Drain|	PGU	PGU OPEN	SMU
CH2 |Gate|	PGU	PGU	SMU

Device Connection for NAND type flash EEPROM.
As shown in Figure 6-2, two HP 16440A SMU/Pulse Generator Selectors are used to switch units for forcing write pulse and erase pulse, and measuring Vth.



Figure 6-2. Device Connection (NAND Type)

Table 6-2 shows the selector's state for each phase:

Table 6-2. Selector's State in Each Phase

Selector Channel	Write	Erase	Vth Measure
CH1 (Drain)	PGU	PGU	SMU
CH2 (Gate)	PGU	PGU	SMU
CH3 (Source)	PGU	PGU	SMU
CH4 (Substrate)	PGU	PGU	SMU

Main Program.

The following is the main program:

```
1560  !/////////////// Main ////////////////////////////////////////////
1570  CALL Init_hp4155
1580  ON INTR 8 CALL Err_check
1590  ENABLE INTR 8;2
1600  !
1610  CALL Test_setting
1620  CALL Get_file
1630  !
1640  Str_num=1
1650  FOR I=1 TO Meas_points
1660    CALL Stress_loop(I)
1670    IF Meas_str_num(I)>4500 THEN CALL Calibration
1680    !
1690    OUTPUT @Hp4155;":MMEM:LOAD:STAT 0,'MEM2.STR','MEMORY'"
1700    OUTPUT @Hp4155;":MMEM:LOAD:STAT 0,'MEM1.MES','MEMORY'"
1710    CALL Vth_meas("Write",I)
1720    !
1730    OUTPUT @Hp4155;":MMEM:LOAD:STAT 0,'MEM3.STR','MEMORY'"
1740    CALL Vth_meas("Erase",I)
1750    !
1760    CALL Trans_data(I)
1770    CALL Stress_graph(I)
1780    !
1790    IF Vth_w(I)<.1 OR Vth_e(I)<.1 THEN
1800      PRINT "     ### The Device is broken. Test Aborted ###"
1810      PRINT "         Final Stress Times  : ";Str_num
1820      CALL Final_session
1830      STOP
1840    END IF
1850    Str_num=Str_num+1
1860  NEXT I
1870  !
1880  CALL Final_session
1890  !
1900  END
```

Line	Description
1570	initializes HP 4155A/4156A.
	enables the Service Request "Enable" Register for Command, Execution, Device-dependent, and Query errors to generate service requests.
1580 and 1590	enables service request from HP 4155A/56A to interrupt program.
1610	defines names of measurement setup files for Vth measurement and stress setup files for write stress and erase stress, and other stress setup.
1620	loads measurement setup file for Vth measurement and stress setup files for write and erase into internal memories.
1650	*Meas_points* is specified in subprogram "Test_setting".
1660	forces write and erase pulses. Refer to "Stress_loop" for details.
1690 and 1700	loads measurement setup file for Vth measurement and stress setup file for write pulse from internal memories.
1710	forces last write pulse, then measures Vth. Refer to "Vth_meas".
1730	loads stress setup file for erase pulse from an internal memory.
1740	forces last erase pulse, then measures Vth. Refer to "Vth_meas".
1760	transfers measurement results |Vth shifts| to HP 4155A/4156A.
1770	displays measurement results.
1880	stores measurement results onto the diskette.

Stress_loop

Subprogram "Stress_loop" to force write and erase stress is shown below:

```
2610 Stress_loop:SUB Stress_loop(INTEGER I)
2620    COM @Hp4155,@Form_off,Start_time,End_time
2630    COM /Meas_info/ INTEGER Meas_points,REAL Str,Str_num,Meas_str_num(*)
2640    INTEGER K
2650    REAL Str_end
2660    !
2670    OUTPUT @Hp4155;":STAT:MEAS:EVEN?"
2680    ENTER @Hp4155;K
2690    OUTPUT @Hp4155;":STAT:MEAS:ENAB 267"
2700    !
2710    OUTPUT @Hp4155;":PAGE:SCON:STAN ON"
2720    Str_end=Meas_str_num(I)-1
2730    FOR Str=Str_num TO Str_end
2740      DISP VAL$(Str);"/";VAL$(Meas_str_num(I))
2750      OUTPUT @Hp4155;"MMEM:LOAD:STAT 0,'MEM2','MEMORY';:PAGE:SCON:STR;*WAI"
2760      OUTPUT @Hp4155;"MMEM:LOAD:STAT 0,'MEM3','MEMORY';:PAGE:SCON:STR"
2770      OUTPUT @Hp4155;"*OPC?"
2780      ENTER @Hp4155;A
2790    NEXT Str
2800    !
2810    Str_num=Str
2820    OUTPUT @Hp4155;":PAGE:SCON:STAN OFF"
2830    OUTPUT @Hp4155;":STAT:MEAS:ENAB 0"
2840  SUBEND
```

Line	Description
2670 and 2680	clears the Measurement/Stress Status "Event" register.
2690	enables Bit 0 (A/D Overflow), 1 (Oscillation Status), 3 (Compliance Status), and 8 (PGU Status) of enable mask for the Measurement/Stress Status "Event" register.
2710	enables standby state so that state does *not* become idle between write and erase stress. If state becomes idle, the relay will switch after every write and erase stress, which will damage the relay.
2730 to 2790	repeats forcing write/erase pulses until one write/erase pulse before next Vth measurement.
2820	disables standby state.

Vth_meas

Subprogram "Vth_meas" to force last write and erase pulses, then measure Vth:

```
2860 Vth_meas:SUB Vth_meas(Str_type$,INTEGER I)
2870    COM @Hp4155,@Form_off,Start_time,End_time
2880    COM /Meas_info/ INTEGER Meas_points,REAL Str,Str_num,Meas_str_num(*)
2890    COM /Meas_data/ Vth_w(*),Vth_e(*)
2900    INTEGER K
2910    !
2920    OUTPUT @Hp4155;":PAGE:SCON:STR;*OPC?"
2930    ENTER @Hp4155;A
2940    DISP Str_type$;" Times = "&VAL$(Str_num)
2950    !
2960    OUTPUT @Hp4155;":PAGE:CHAN:COMM 'Flash ROM Vth Meas. @"&Str_type$&" Times = "&VAL$(Str_num)&"'"
2970    OUTPUT @Hp4155;":PAGE:GLIS"
2980    OUTPUT @Hp4155;":DISP ON"
2990    OUTPUT @Hp4155;":PAGE:SCON:SING;*OPC?"
3000    ENTER @Hp4155;A
3010    OUTPUT @Hp4155;":DISP OFF"
3020    OUTPUT @Hp4155;":STAT:MEAS:EVEN?"
3030    ENTER @Hp4155;K
3040    OUTPUT @Hp4155;":TRAC? 'VTH'"
3050    SELECT Str_type$
3060    CASE "Write"
3070      ENTER @Hp4155;Vth_w(I)
3080      PRINT USING "#,4X,DESZ,10X,SD.DDD";Str_num,Vth_w(I)
3090    CASE "Erase"
3100      ENTER @Hp4155;Vth_e(I)
3110      PRINT USING "10X,SD.DDD,7X,SD.DDE";Vth_e(I),Vth_w(I)-Vth_e(I)
3120    END SELECT
3130  SUBEND
```

Line	Description
2990 and 3000	executes Vth measurement and waits until completion.
3040	gets measurement result.

Program Customization

This section describes how to customize program for your own application.

Subprogram "Test_setting".
In this subprogram, you may need to customize the following:

- Name of setup files.

  If you want to use your own measurement or stress setup files, store the files on diskette, then modify the file names on the following lines:

  - □ Measurement setup file name for Vth measurement.

```
1980    Vth_file$="ROMVTH.MES"     ! Vth Measurement Setup File Name
```

  - □ Stress setup file name for write pulse.

```
1990    Wrt_file$="NORWRT.STR"     ! Write Stress Setup File Name
```

  - □ Stress setup file name for erase pulse.

```
2000    Ers_file$="NORERS.STR"     ! Erase Stress Setup File Name
```

- File name for saving measurement results.

  Following two lines create following file name for saving measurement results: *time*.DAT. To change this file name, modify these lines:

```
2010    Save_file$=TIME$(TIMEDATE) ! File Name for saving measurement results
2020    Save_file$=Save_file$[1,2]&Save_file$[4,5]&Save_file$[7,7]&".DAT"
```

- Number of times to repeat measurement (FOR loop of Main Program) Following line specifies how many times to measure Vth during stress.

```
2030    Meas_points=16             ! Number of times to repeat Measurement
```

- Stress pulse count data.

  For example, if Meas_points=4, a total of ten write/erase pulses are forced, and Vth is measured after 1st, 2nd, 5th, and 10th pulse.

```
2060 Str_num: !                    ! Stress Pulse Count data
2070    DATA      1,      2,      5
2080    DATA     10,     20,     50
2090    DATA    100,    200,    500
2100    DATA   1000,   2000,   5000
2110    DATA  10000,  20000,  50000
2120    DATA 100000, 200000, 500000
2130    DATA 1000000
```

Measurement setup file for Vth measurement (for NOR type).
Measurement setup for Vth measurement is stored in "ROMVTH.MES" file on provided diskette. As described previously, if you use your own setup file with a different file name, change line 2000.
In the ROMVTH.MES file, the following is set up. You can modify these settings in the ROMVTH.MES file or your own file:

- Gate voltage sweep setup.

Start voltage	Stop voltage	Sweep step	Compliance
0 V	8 V	10 mV	1 nA

  SMU1 is gate voltage source as shown in Figure 6-1 and Figure 6-2.

- Constant source setup.

Units	Output	Compliance
SMU2 (Source)	0 V	100 μA
SMU3 (Drain)	100 mV	2 μA
SMU4 (Substrate)	0 V	100 μA

- Analysis function for Vth extraction.

  In this example, Vth is extracted by moving marker to the point where Id is 1 μA, then reading the voltage at that point. Refer to the following user function and auto-analysis setup:

User Function Definition

Name	Unit	Definition
Vth	V	@MX

Analysis Setup

Setup	Definition
Marker	Id = 1 μA
Interpolate	ON

Stress setup file for write pulse of NOR type.
Stress setup for write pulse of NOR type is stored in "NORWRT.STR" file on provided diskette. As described previously, if you use your own setup file with a different file name, change line 2010.
In the NORWRT.STR file, the following is set up. You can modify these settings in the NORWRT.STR file or your own file:

- PGUs

Unit	Period	Width	Delay Time	Peak Value	Base Value	Leading Time	Trailing Time	Impedance
PGU1 (Gate)	1.03 ms	1.02 ms	0.0 s	14 V	0 V	1 μs	1 μs	50 ohm
PGU2 (Drain)		1.00 ms	10 μs	7 V	0 V	1 μs	1 μs	50 ohm

- Constant source setup

Unit	Source	Compliance
SMU2 (Source)	0 V	100 mA
SMU4 (Substrate)	0 V	100 mA

Stress setup file for erase pulse of NOR type.
Stress setup for erase pulse is stored on "NORERS.STR" file on provided diskette. As described previously, if you use your own setup file with a different file name, change line 2020.

In the NORERS.STR file, the following is set up. You can modify these settings in the NORERS.STR file or your own file:

- Constant source setup

Unit	Source	Compliance
SMU2 (Source)[1]	11 V	100 mA
SMU4 (Substrate)	0 V	100 mA

1 Erase pulse source

- Erase pulse width

  Pulse width of erase pulse is specified as stress DURATION and set to 20ms.

Stress setup file for write pulse of NAND type.
Stress setup for write pulse of NAND type is stored in "NANWRT.STR" file on provided diskette. As described previously, you must change line 2010 to "NANWRT.STR" or your own custom file name.
In the NANWRT.STR file, the following is set up. You can modify these settings in the NANWRT.STR file or your own file:

- PGUs

Unit	Period	Width	Delay Time	Peak Value	Base Value	Leading Time	Trailing Time	Impedance
PGU1 (Gate)	413 μs	400 μs	0.0 s	20 V	0 V	10 μs	10 μs	50 ohm

Unit	Source	Impedance
PGU2[1]	0 V	50 ohm

1 Connected to drain, source, and substrate, and set to constant source.

Stress setup file for erase pulse of NAND type.
Stress setup for erase pulse of NAND type is stored in "NANERS.STR" file on provided diskette. As described previously, you must change line 2020 to "NANERS.STR" or your own custom file name.
In the NANERS.STR file, the following is set up. You can modify these settings in the NANERS.STR file or your own file:

- PGUs

Unit	Period	Width	Delay Time	Peak Value	Base Value	Leading Time	Trailing Time	Impedance
PGU2[1]	5.02 ms	5.00 ms	0.0 s	20 V	0 V	10 μs	10 μs	50 ohm

1 connected to drain, source, and substrate.

Unit	Source	Impedance
PGU1[1]	0 V	50 ohm

1 Connected to gate, and set to constant source

Program Listing

```
1000 !************************************************
1010 !*
1020 !* FILE:         NOR_TEST
1030 !* DESCRIPTION:  Program for NOR-FLASH ROM Stress Test.
1040 !*
1050 !* AUTHOR:       Yukoh Iwasaki , YHP
1060 !* CREATED:      12/21/1993
1070 !* MODIFIED:     01/25/1994
1080 !* PRODUCT:      HP4155A,HP4156A
1090 !* REVISION:     Rev.A.01.02
1100 !*
1110 !*   (c) Copyright 1994, Hewlett-Packard Co,
1120 !*              All rights reserved.
1130 !*
1140 !*
1150 !*  Customer shall have the personal, non-
1160 !* transferable rights to use, copy or modify
1170 !* this SAMPLE PROGRAM for Customer's internal
1180 !* operations.  Customer shall use the SAMPLE
1190 !* PROGRAM solely and exclusively for its own
1200 !* purpose and shall not license, lease, market
1210 !* or distribute the SAMPLE PROGRAM or modification
1220 !* or any part thereof.
1230 !*
1240 !*  HP shall not be liable for the quality,
1250 !* performance or behavior of the SAMPLE PROGRAM.
1260 !* HP especially disclaims that the operation of
1270 !* the SAMPLE PROGRAM shall be uninterrupted or
1280 !* error free.  This SAMPLE PROGRAM is provided
1290 !* AS IS.
1300 !*
1310 !*  HP DISCLAIMS THE IMPLIED WARRANTIES OF
1320 !* MERCHANTABILITY AND FITNESS FOR A PARTICULAR
1330 !* PURPOSE.
1340 !*
1350 !*  HP shall not be liable for any infringement
1360 !* of any patent, trademark, copyright or other
1370 !* proprietary rights by the SAMPLE PROGRAM or
1380 !* its use.  HP does not warrant that the SAMPLE
1390 !* PROGRAM is free from infringements or such
1400 !* rights of third parties.  However, HP will not
1410 !* knowingly infringe or deliver a software that
1420 !* infringes the patent, trademark, copyright or
1430 !* other proprietary right of a third party.
1440 !*
1450 !************************************************
1460  Start_time=TIMEDATE
1470  ASSIGN @Hp4155 TO 800
1480  ASSIGN @Form_off TO 800;FORMAT OFF
1490  COM @Hp4155,@Form_off,Start_time,End_time
1500  COM /File_name/ Vth_file$[12],Wrt_file$[12],Ers_file$[12],Save_file$[12]
1510  COM /Meas_info/ INTEGER Meas_points,REAL Str,Str_num,Meas_str_num(1:55)
1520  COM /Meas_data/ Vth_w(1:55),Vth_e(1:55)
1530  COM /Err/ Err_num(1:6),Err_message$(1:6)[50]
```

```
1540  INTEGER I
1550  !
1560  !/////////////// Main ////////////////////////////////////////////////
1570  CALL Init_hp4155
1580  ON INTR 8 CALL Err_check
1590  ENABLE INTR 8;2
1600  !
1610  CALL Test_setting
1620  CALL Get_file
1630  !
1640  Str_num=1
1650  FOR I=1 TO Meas_points
1660    CALL Stress_loop(I)
1670    IF Meas_str_num(I)>4500 THEN CALL Calibration
1680    !
1690    OUTPUT @Hp4155;":MMEM:LOAD:STAT 0,'MEM2.STR','MEMORY'"
1700    OUTPUT @Hp4155;":MMEM:LOAD:STAT 0,'MEM1.MES','MEMORY'"
1710    CALL Vth_meas("Write",I)
1720    !
1730    OUTPUT @Hp4155;":MMEM:LOAD:STAT 0,'MEM3.STR','MEMORY'"
1740    CALL Vth_meas("Erase",I)
1750    !
1760    CALL Trans_data(I)
1770    CALL Stress_graph(I)
1780    !
1790    IF Vth_w(I)<.1 OR Vth_e(I)<.1 THEN
1800      PRINT "     ### The Device is broken. Test Aborted ###"
1810      PRINT "         Final Stress Times  : ";Str_num
1820      CALL Final_session
1830      STOP
1840    END IF
1850    Str_num=Str_num+1
1860  NEXT I
1870  !
1880  CALL Final_session
1890  !
1900  END
1910  !
1920  !/////////////// Sub /////////////////////////////////////////////////
1930 Test_setting:SUB Test_setting
1940    COM /File_name/ Vth_file$,Wrt_file$,Ers_file$,Save_file$
1950    COM /Meas_info/ INTEGER Meas_points,REAL Str,Str_num,Meas_str_num(*)
1960    COM /Meas_data/ Vth_w(*),Vth_e(*)
1970    !
1980    Vth_file$="ROMVTH.MES"      ! Vth Measurement Setup File Name
1990    Wrt_file$="NORWRT.STR"      ! Write Stress Setup File Name
2000    Ers_file$="NORERS.STR"      ! Erase Stress Setup File Name
2010    Save_file$=TIME$(TIMEDATE) ! File Name for saving measurement results
2020    Save_file$=Save_file$[1,2]&Save_file$[4,5]&Save_file$[7,7]&".DAT"
2030    Meas_points=16              ! Number of times to repeat Measurement
2040    REDIM Meas_str_num(1:Meas_points)
2050    REDIM Vth_w(1:Meas_points),Vth_e(1:Meas_points)
2060 Str_num: !                     ! Stress Pulse Count data
2070    DATA       1,      2,      5
2080    DATA      10,     20,     50
2090    DATA     100,    200,    500
2100    DATA    1000,   2000,   5000
2110    DATA   10000,  20000,  50000
2120    DATA  100000, 200000, 500000
2130    DATA 1000000
2140    RESTORE Str_num
```

```
2150    READ Meas_str_num(*)
2160  SUBEND
2170 !
2180 Init_hp4155:SUB Init_hp4155
2190    COM @Hp4155,@Form_off,Start_time,End_time
2200    !
2210    CLEAR SCREEN
2220    CLEAR @Hp4155
2230    OUTPUT @Hp4155;"*RST"
2240    OUTPUT @Hp4155;"*CLS"
2250    OUTPUT @Hp4155;":STAT:PRES"
2260    OUTPUT @Hp4155;"*ESE 60;*SRE 34;*OPC?"
2270    ENTER @Hp4155;A
2280    OUTPUT @Hp4155;":DISP:WIND:ALL BST"
2290    OUTPUT @Hp4155;":DISP OFF"
2300    PRINT "            <<< Flash ROM Stress Test >>>"
2310    PRINT "Stress Times   Vth Write [V]   Vth Erase [V]   Diff [V]"
2320  SUBEND
2330 !
2340 Get_file:SUB Get_file
2350    COM @Hp4155,@Form_off,Start_time,End_time
2360    COM /File_name/ Vth_file$,Wrt_file$,Ers_file$,Save_file$
2370    !
2380    OUTPUT @Hp4155;":MMEM:COPY '"&Vth_file$&"','DISK','MEM1.MES','MEMORY'"
2390    OUTPUT @Hp4155;":MMEM:COPY '"&Wrt_file$&"','DISK','MEM2.STR','MEMORY'"
2400    OUTPUT @Hp4155;":MMEM:COPY '"&Ers_file$&"','DISK','MEM3.STR','MEMORY'"
2410  SUBEND
2420  !
2430 Calibration:SUB Calibration
2440    COM @Hp4155,@Form_off,Start_time,End_time
2450    !
2460    OUTPUT @Hp4155;":PAGE:SYST:CDI"
2470    OUTPUT @Hp4155;":DISP ON"
2480    OUTPUT @Hp4155;":CAL:ALL?"
2490    ENTER @Hp4155;A
2500    SELECT A
2510    CASE 0
2520      OUTPUT @Hp4155;":PAGE:GLIS"
2530      OUTPUT @Hp4155;":DISP OFF"
2540    CASE ELSE
2550      PRINT "     #### Calibration FAIL ,Test Aborted ####"
2560      CALL Final_session
2570      STOP
2580    END SELECT
2590  SUBEND
2600  !
2610 Stress_loop:SUB Stress_loop(INTEGER I)
2620    COM @Hp4155,@Form_off,Start_time,End_time
2630    COM /Meas_info/ INTEGER Meas_points,REAL Str,Str_num,Meas_str_num(*)
2640    INTEGER K
2650    REAL Str_end
2660    !
2670    OUTPUT @Hp4155;":STAT:MEAS:EVEN?"
2680    ENTER @Hp4155;K
2690    OUTPUT @Hp4155;":STAT:MEAS:ENAB 267"
2700    !
2710    OUTPUT @Hp4155;":PAGE:SCON:STAN ON"
2720    Str_end=Meas_str_num(I)-1
2730    FOR Str=Str_num TO Str_end
2740      DISP VAL$(Str);"/";VAL$(Meas_str_num(I))
2750      OUTPUT @Hp4155;"MMEM:LOAD:STAT 0,'MEM2','MEMORY';:PAGE:SCON:STR;*WAI"
```

```
2760      OUTPUT @Hp4155;"MMEM:LOAD:STAT 0,'MEM3','MEMORY';:PAGE:SCON:STR"
2770      OUTPUT @Hp4155;"*OPC?"
2780      ENTER @Hp4155;A
2790    NEXT Str
2800    !
2810    Str_num=Str
2820    OUTPUT @Hp4155;":PAGE:SCON:STAN OFF"
2830    OUTPUT @Hp4155;":STAT:MEAS:ENAB 0"
2840  SUBEND
2850  !
2860 Vth_meas:SUB Vth_meas(Str_type$,INTEGER I)
2870    COM @Hp4155,@Form_off,Start_time,End_time
2880    COM /Meas_info/ INTEGER Meas_points,REAL Str,Str_num,Meas_str_num(*)
2890    COM /Meas_data/ Vth_w(*),Vth_e(*)
2900    INTEGER K
2910    !
2920    OUTPUT @Hp4155;":PAGE:SCON:STR;*OPC?"
2930    ENTER @Hp4155;A
2940    DISP Str_type$;" Times = "&VAL$(Str_num)
2950    !
2960    OUTPUT @Hp4155;":PAGE:CHAN:COMM 'Flash ROM Vth Meas. @"&Str_type$&" Times = "&VAL$(Str_num)&"'"
2970    OUTPUT @Hp4155;":PAGE:GLIS"
2980    OUTPUT @Hp4155;":DISP ON"
2990    OUTPUT @Hp4155;":PAGE:SCON:SING;*OPC?"
3000    ENTER @Hp4155;A
3010    OUTPUT @Hp4155;":DISP OFF"
3020    OUTPUT @Hp4155;":STAT:MEAS:EVEN?"
3030    ENTER @Hp4155;K
3040    OUTPUT @Hp4155;":TRAC? 'VTH'"
3050    SELECT Str_type$
3060    CASE "Write"
3070      ENTER @Hp4155;Vth_w(I)
3080      PRINT USING "#,4X,DESZ,10X,SD.DDD";Str_num,Vth_w(I)
3090    CASE "Erase"
3100      ENTER @Hp4155;Vth_e(I)
3110      PRINT USING "10X,SD.DDD,7X,SD.DDE";Vth_e(I),Vth_w(I)-Vth_e(I)
3120    END SELECT
3130  SUBEND
3140  !
3150 Trans_data:SUB Trans_data(INTEGER I)
3160    COM @Hp4155,@Form_off,Start_time,End_time
3170    COM /Meas_info/ INTEGER Meas_points,REAL Str,Str_num,Meas_str_num(*)
3180    COM /Meas_data/ Vth_w(*),Vth_e(*)
3190    !
3200    REDIM Meas_str_num(1:I),Vth_w(1:I),Vth_e(1:I)
3210    OUTPUT @Hp4155;":TRAC:DEL:ALL"
3220    OUTPUT @Hp4155;":TRAC:DEF 'Stress',"&VAL$(I)
3230    OUTPUT @Hp4155;":TRAC:DEF 'VthWRT',"&VAL$(I)
3240    OUTPUT @Hp4155;":TRAC:DEF 'VthERS',"&VAL$(I)
3250    OUTPUT @Hp4155;":TRAC:UNIT 'Stress','Times'"
3260    OUTPUT @Hp4155;":TRAC:UNIT 'VthWRT','V'"
3270    OUTPUT @Hp4155;":TRAC:UNIT 'VthERS','V'"
3280    !
3290    OUTPUT @Hp4155;":FORM:DATA REAL,64"
3300    OUTPUT @Hp4155;":FORM:BORD NORM"
3310    OUTPUT @Hp4155;":TRAC 'Stress',#0";
3320    OUTPUT @Form_off;Meas_str_num(*),END
3330    OUTPUT @Hp4155;":TRAC 'VthWRT',#0";
3340    OUTPUT @Form_off;Vth_w(*),END
3350    OUTPUT @Hp4155;":TRAC 'VthERS',#0";
3360    OUTPUT @Form_off;Vth_e(*),END
```

```
3370    OUTPUT @Hp4155;":FORM:DATA ASCII"
3380    REDIM Meas_str_num(1:Meas_points),Vth_w(1:Meas_points),Vth_e(1:Meas_points)
3390  SUBEND
3400  !
3410 Stress_graph:SUB Stress_graph(INTEGER I)
3420    COM @Hp4155,@Form_off,Start_time,End_time
3430    COM /Meas_info/ INTEGER Meas_points,REAL Str,Str_num,Meas_str_num(*)
3440    !
3450    OUTPUT @Hp4155;":PAGE:CHAN:COMM 'Flash ROM Vth Shift(Stress="&VAL$(Meas_str_num(I))&")'"
3460    OUTPUT @Hp4155;":PAGE:CHAN:UFUN:DEF 'Diff','V','VthWRT-VthERS'"
3470    !
3480    OUTPUT @Hp4155;":PAGE:DISP:GRAP:X:NAME 'Stress'"
3490    OUTPUT @Hp4155;":PAGE:DISP:GRAP:Y1:NAME 'VthWRT'"
3500    OUTPUT @Hp4155;":PAGE:DISP:GRAP:Y2:NAME 'VthERS'"
3510    OUTPUT @Hp4155;":PAGE:DISP:GRAP:X:SCAL LOG"
3520    OUTPUT @Hp4155;":PAGE:DISP:GRAP:X:MIN 1"
3530    OUTPUT @Hp4155;":PAGE:DISP:GRAP:X:MAX "&VAL$(MAX(Meas_str_num(Meas_points),2))
3540    OUTPUT @Hp4155;":PAGE:DISP:GRAP:Y1:SCAL LIN"
3550    OUTPUT @Hp4155;":PAGE:DISP:GRAP:Y1:MIN 0"
3560    OUTPUT @Hp4155;":PAGE:DISP:GRAP:Y1:MAX 7"
3570    OUTPUT @Hp4155;":PAGE:DISP:GRAP:Y2:SCAL LIN"
3580    OUTPUT @Hp4155;":PAGE:DISP:GRAP:Y2:MIN 0"
3590    OUTPUT @Hp4155;":PAGE:DISP:GRAP:Y2:MAX 7"
3600    !
3610    OUTPUT @Hp4155;":PAGE:DISP:DVAR:DEL 'VTH'"
3620    OUTPUT @Hp4155;":PAGE:DISP:DVAR 'Diff'"
3630    !
3640    OUTPUT @Hp4155;":PAGE:DISP:ANAL:LINE1:MODE DIS"
3650    OUTPUT @Hp4155;":PAGE:DISP:ANAL:LINE2:MODE DIS"
3660    OUTPUT @Hp4155;":PAGE:DISP:ANAL:MARK:DIS"
3670    !
3680    OUTPUT @Hp4155;":PAGE:GLIS:INT OFF"
3690    OUTPUT @Hp4155;":PAGE:GLIS:LINE OFF"
3700    OUTPUT @Hp4155;":PAGE:GLIS:MARK ON"
3710    OUTPUT @Hp4155;":PAGE:GLIS:MARK:DIR:X MAX"
3720    !
3730    OUTPUT @Hp4155;":DISP ON;:DISP OFF"
3740  SUBEND
3750  !
3760 Final_session:SUB Final_session
3770    COM @Hp4155,@Form_off,Start_time,End_time
3780    COM /File_name/ Vth_file$,Wrt_file$,Ers_file$,Save_file$
3790    COM /Meas_info/ INTEGER Meas_points,REAL Str,Str_num,Meas_str_num(*)
3800    COM /Err/ Err_num(*),Err_message$(*)
3810    !
3820    IF Str_num-1=Meas_str_num(Meas_points) THEN
3830      Save_file$="D"&Save_file$
3840      PRINT "================ Measurement Completed !!! ================="
3850      PRINT "Final Stress Times  : ";Str_num-1
3860    ELSE
3870      Save_file$="F"&Save_file$
3880    END IF
3890    !
3900    PRINT "Save Data File Name : ";Save_file$
3910    PRINT "Test Duration : ";
3920    End_time=TIMEDATE
3930    PRINT DATE$(Start_time);",";TIME$(Start_time);" ~ ";DATE$(End_time);",";TIME$(End_time)
3940    !
3950    DISABLE INTR 8
3960    OUTPUT @Hp4155;":MMEM:STOR:TRAC DEF,'"&Save_file$&"','DISK'"
3970    OUTPUT @Hp4155;"*OPC?"
```

```
3980    ENTER @Hp4155;A
3990    OUTPUT @Hp4155;":SYST:ERR?"
4000    ENTER @Hp4155;Err_num(1),Err_message$(1)
4010    IF Err_num(1)<>0 THEN PRINT "### ";Err_num(1);Err_message$(1);" ###"
4020    !
4030    OUTPUT @Hp4155;":DISP:ALL INST"
4040    OUTPUT @Hp4155;":PAGE:GLIS"
4050    OUTPUT @Hp4155;":DISP ON"
4060  SUBEND
4070  !
4080 Err_check:SUB Err_check
4090    COM @Hp4155,@Form_off,Start_time,End_time
4100    COM /Err/ Err_num(*),Err_message$(*)
4110    INTEGER I,J
4120    !
4130    I=0
4140    REPEAT
4150      I=I+1
4160      OUTPUT @Hp4155;":SYST:ERR?"
4170      ENTER @Hp4155;Err_num(I),Err_message$(I)
4180    UNTIL Err_num(I)=0
4190    !
4200    IF I=1 THEN
4210      CALL Meas_stat_check
4220    ELSE
4230      FOR J=1 TO I-1
4240        PRINT "### ERROR Occurred ###:";Err_num(J);Err_message$(J)
4250        DISP "### ERROR Occurred ###:";Err_num(J);Err_message$(J)
4260      NEXT J
4270      CALL Meas_stat_check
4280      PRINT "              === Test Aborted ==="
4290      CALL Final_session
4300      STOP
4310    END IF
4320  SUBEND
4330  !
4340 Meas_stat_check:SUB Meas_stat_check
4350    COM @Hp4155,@Form_off,Start_time,End_time
4360    COM /Meas_info/ INTEGER Meas_points,REAL Str,Str_num,Meas_str_num(*)
4370    INTEGER K
4380    !
4390    OUTPUT @Hp4155;":STAT:MEAS:EVEN?"
4400    ENTER @Hp4155;K
4410    !
4420    IF K<>0 THEN
4430      PRINT "### Abnormal Stress Status Event Occurred ###:";K
4440      PRINT "    at Stress Number = ";Str;"[Times]"
4450      PRINT "              === Test Aborted ==="
4460      CALL Final_session
4470      STOP
4480    END IF
4490  SUBEND
```

Time Dependent Dielectric Breakdown (TDDB)

This setup forces a constant voltage to the gate until the gate oxide breakdowns or a maximum time limit is reached, then calculates the total forced electric charge.

	name
Program	none
Setup file	TDDB.MES

Application Overview



Figure 6-3. Device Connection

The measurement flow is as follows:

1. Forces a constant voltage to the gate.
2. Measures gate current by sampling measurement.
3. If gate current exceeds specified threshold, measurement is stopped.
4. Calculates total electric charge that was forced by using a user function with definition `INTEG(Ig, @TIME)`.

Customization

Measurement setup file is stored in "TDDB.MES" file on provided diskette. In the TDDB.MES file, the following is set up. You can modify these settings in the TDDB.MES file or your own file, then use the setup for your own application.

- Constant source setup

Units	Output	Compliance
SMU1 (Gate)	20 V	1.001 μA
SMU4 (Substrate)	0 V	100 μA

- Sampling Parameters

Mode	Initial interval	No. of samples	Total samp. time
Thinned-out	100 ms	1001	999.9 s

- Stop Condition

  This setup is used to judge the oxide breakdown. If gate current exceeds the specified threshold, measurement is stopped.

Enable Delay	Threshold
200 ms	1 μA

Electromigration

This setup forces a constant current to the DUT (metal), measures time-to-failure of DUT, then calculates the total forced electric charge.

	name
Program	none
Setup file	EM.MES

Application Overview



Figure 6-4. Device Connection

The measurement flow is as follows:

1. Forces constant current.
2. Monitors DUT voltage by sampling measurement.
3. If the DUT voltage reaches specified threshold, the forcing stops.
4. Calculates total electric charge that was forced by using a user function with definition `INTEG(Idut1, @TIME)`.

Customization

Measurement setup file is stored in "EM.MES" file on provided diskette. In the EM.MES file, the following is set up. You can modify these settings in the EM.MES file or your own file, then use the setup for your own application.

- Constant source setup

Units	Output	Compliance
SMU1	50 mA	20.002 V

- Sampling Parameters

Mode	Initial interval	No. of samples	Total samp. time
Linear	1 s	10001	AUTO[1]

1 Initial interval * No. of samples

- Stop Condition

  If the DUT voltage exceeds the specified threshold, measurement is stopped.

Enable Delay	Threshold
20 ms	20 V

7

Manual Changes Depending on ROM Version

Manual Changes Depending on ROM Version

HP 4155A/4156A may vary slightly, depending on the version of the ROM based firmware. The information in this manual applies to an HP 4155A/4156A with the following ROM version.

Manual Applies to this ROM Version

ROM	ROM Version
HOSTC	01.02

ROM version

To confirm your ROM version, check the `SOFTWARE REVISION` field on the SYSTEM: CONFIGURATION page.

This chapter contains information for customizing this manual so that it is correct for the HP 4155A/4156A that you are using.

To customize this manual for your HP 4155A/4156A, refer to the following table, and make the manual changes depending on the ROM version of your HP 4155A/56A.

Manual Changes by ROM version

ROM version (HOSTC)	Make Manual Changes
01.00	1
01.01	1

Change 1

Add the following section to Chapter 5.

Timing Considerations

You may need timing considerations for synchronizing measurements with external instruments.

Following example program controls a DMM with the HP 4155A/4156A. Measurement circuit is shown below:



The HP 4155A/4156A has an HP-IB I/O buffer and can receive commands before executing previous command. So the execution order of HP 4155A/4156A and DMM is different from the order in the program.

In the following example, the HP 3458A receives a measurement trigger command right after the HP 4155A/4156A receives a force voltage trigger.

But due to the HP-IB I/O buffer, the voltage measurement is made by HP 3458A before HP 4155A/4156A forces voltage.

```
10   ASSIGN @Hp415x TO 717
20   ASSIGN @Hp3458 TO 722
30   !
40   DIM Hoe_data$[17]
50   DIM Dmm_data$[100]
60   !
70   OUTPUT @Hp3458;"RESET"                        ! Reset DMM
80   OUTPUT @Hp3458;"TARM HOLD"                    ! Suspend reading
90   OUTPUT @Hp3458;"DCV 10"                       ! DC voltage 10 V range
100  OUTPUT @Hp3458;"NPLC 1"                       ! Integ NPLC=1
110  OUTPUT @Hp3458;"AZERO OFF"                    ! Auto zero off
120  !
130  OUTPUT @Hp415x;"US"                           ! User mode
140  OUTPUT @Hp415x;"IT1 CAO BC"                   ! Integ Short, Cal off, buffer clear
150  Force_v=1.5
160  OUTPUT @Hp415x;"DV1, 1, ";Force_v;", 20E-3"   ! SMU1 forces 1.5 V
170  OUTPUT @Hp415x;"DV2, 1, 0, 20E-3"             ! SMU2 forces 0 V
180  !
190  OUTPUT @Hp3458;"TARM SGL"                     ! V meas trigger to DMM
200  ENTER @Hp3458;Dmm_data$
210  !
220  OUTPUT @Hp415x;"TI1"                          ! I meas trigger to 4155/56
230  ENTER @Hp415x;Hoe_data$
240  !
250  PRINT Dmm_data$
260  PRINT Hoe_data$
270  !
280  OUTPUT @Hp415x;"DV1; DV2"                     ! Reset SMU1 & SMU2
290  END
```

Following is an example of test result:

```
-3.870540468E-04
NAI-1.256700E-010
```

DMM should measure about 1.5 V, but does not because DMM measures before SMU1 forces voltage.

You can insert a **WAIT** command before sending trigger command to the DMM (HP 3458A). In this example, 0.1 sec should be sufficient. So, you can insert the following in the above program:

```
181   WAIT .1                          ! <<< Modified
```

Or maybe the following is a better way to modify the program. The measurement order is changed. First, SMU measures current, and controller enters data from HP 4155A/4156A. Then, the measurement trigger is sent to the DMM.

```
10    ASSIGN @Hp415x TO 717
20    ASSIGN @Hp3458 TO 722
30    !
40    DIM Hoe_data$[17]
50    DIM Dmm_data$[100]
60    !
70    OUTPUT @Hp3458;"RESET"
80    OUTPUT @Hp3458;"TARM HOLD"
90    OUTPUT @Hp3458;"DCV 10"
100   OUTPUT @Hp3458;"NPLC 1"
110   OUTPUT @Hp3458;"AZERO OFF"
120   !
130   OUTPUT @Hp415x;"US"
140   OUTPUT @Hp415x;"IT1 CAO BC"
150   Force_v=1.5
160   OUTPUT @Hp415x;"DV1, 1, ";Force_v;", 20E-3"
170   OUTPUT @Hp415x;"DV2, 1, 0, 20E-3"
180   !
190   OUTPUT @Hp415x;"TI1"
200   ENTER @Hp415x;Hoe_data$
210   !
220   OUTPUT @Hp3458;"TARM SGL"     ! <<< Changed order of measurement
230   ENTER @Hp3458;Dmm_data$
240   !
250   PRINT Dmm_data$
260   PRINT Hoe_data$
270   !
280   OUTPUT @Hp415x;"DV1; DV2"
290   END
```

The test result should be similar to following example:

```
 1.499301638E+00
NAI+7.150000E-011
```

Data Length Considerations

When one of the following commands is sent from a controller to the HP 4155A/4156A in the *HP 4145 syntax command mode*, the HP 4155A/4156A outputs data to the controller.

DO Data output request in the system mode

TI Current measurement trigger and data output request in the user mode

TV Voltage measurement trigger and data output request in the user mode

ID Identification output

Different Data Length in System Mode.
The following program is equivalent to sample program 1 listed in Section 3 of HP 4145B manual. This program is an example of data output in the system mode. This program controls instrument to measure the IC-VC characteristics of a bipolar transistor, then returns IC data to the controller.

```
10    ASSIGN @Hp415x TO 717
20    DIM A$[1100]
30    OUTPUT @Hp415x;"IT1 CA1 DR0 BC"
40    OUTPUT @Hp415x;"DE CH1, 'VE', 'IE', 3, 3; CH2, 'VB', 'IB', 2, 2; CH3, 'VC', 'IC', 1, 1; CH4"
50    OUTPUT @Hp415x;"VS1; VS2;VM1; VM2"
60    OUTPUT @Hp415x;"SS VR1, 0, 1, .05, 50E-3; IP 10E-6, 10E-6, 4, 3"
70    OUTPUT @Hp415x;"SM DM1; XN 'VC', 1, 0, 1; YA 'IC', 1, 0, 10E-3"
80    OUTPUT @Hp415x;"MD ME1"
90    A=SPOLL(@Hp415x)
100   IF BIT(A,0)=0 THEN 90
110   OUTPUT @Hp415x;"DO 'IC'"
120   ENTER @Hp415x;A$
130   PRINT A$
140   END
```

You must execute this program in the *HP 4145 syntax command mode* of the HP 4155A/4156A. The following will be displayed on the controller's screen. (Or you can display on HP 4155A/4156A's screen by changing the select code/HP-IB address assigned in line 10 to 800, which will execute the program using built-in IBASIC of the HP 4155A/4156A.) As you can see, the length of data string A$ is to short, so not all the data is displayed.

```
N-9.985900E-006,N-5.923600E-007,N+6.087900E-005,N+3.772100E-004,
N+1.106890E-003,N+1.642500E-003,N+1.824900E-003,N+1.870700E-003,
N+1.880600E-003,N+1.883000E-003,N+1.884600E-003,N+1.885700E-003,
N+1.886900E-003,N+1.887900E-003,N+1.888900E-003,N+1.889800E-003,
N+1.890800E-003,N+1.891700E-003,N+1.892600E-003,N+1.893000E-003,
N+1.893800E-003,N-1.996500E-005,N+1.685000E-006,N+1.426100E-004,
```

```
N+8.321900E-004,N+2.290100E-003,N+3.298500E-003,N+3.635200E-003,
N+3.723200E-003,N+3.744200E-003,N+3.750600E-003,N+3.753600E-003,
N+3.756300E-003,N+3.758900E-003,N+3.761200E-003,N+3.763700E-003,
N+3.765400E-003,N+3.767700E-003,N+3.769600E-003,N+3.771600E-003,
N+3.773400E-003,N+3.775300E-003,N-2.994400E-005,N+5.370400E-006,
N+2.329300E-004,N+1.301200E-003,N+3.434400E-003,N+4.893700E-003,
N+5.395400E-003,N+5.534300E-003,N+5.571000E-003,N+5.582700E-003,
N+5.588900E-003,N+5.593600E-003,N+5.597700E-003,N+5.601500E-003,
N+5.605600E-003,N+5.608800E-003,N+5.612400E-003,N+5.615200E-003,
N+5.618500E-003,N+5.621600E-003,N+5.624300E-003,N-3.992100E-005,
N+1.014500E-005,N+3.293800E-004,N+1.772600E-003,N+4.517500E-003,
N+6.396500E-
```

This is because the length of data output from the HP 4155A/4156A and HP 4145B is different as follows:

HP 4145B data output format in system mode:

```
X+NN.NNNE+NN,X+NN.NNNE+NN, ... X+NN.NNNE+NN[cr][lf]
```

Each data consists of 13 characters (including a comma) except for the last data, which consists of 12 characters.

HP 4155A/4156A data output format in system mode of 4145 syntax command mode:

```
X+N.NNNNNNE+NNN,X+N.NNNNNNE+NNN,X+N.NNNNNNE+NNN, ... X+N.NNNNNNE+NNN
```

Each data consists of 16 characters (including a comma), except for the last data, which consists of 15 characters.

`X`	Data status
`+`	+ or −
`N`	Numeric character

In this example, number of data points is 21×4=84 (21 Var1 steps and 4 Var2 steps). So, length of data string A$ should be at least 1343 (16 char × 84 points - 1). Please modify line 20 as follows:

```
20    DIM A$[1343]            ! <<< Modified 16 x 84 - 1 = 1343
```

The result display is as follows (all data is displayed):

```
N-9.984700E-006,N-5.928300E-007,N+6.088100E-005,N+3.772600E-004,
N+1.107160E-003,N+1.642400E-003,N+1.825100E-003,N+1.870300E-003,
N+1.879800E-003,N+1.882700E-003,N+1.884200E-003,N+1.885200E-003,
N+1.886500E-003,N+1.887500E-003,N+1.888700E-003,N+1.889700E-003,
N+1.890600E-003,N+1.891200E-003,N+1.892300E-003,N+1.893100E-003,
N+1.893700E-003,N-1.996500E-005,N+1.678600E-006,N+1.426100E-004,
N+8.323300E-004,N+2.290400E-003,N+3.298600E-003,N+3.634900E-003,
N+3.722600E-003,N+3.743600E-003,N+3.749900E-003,N+3.753400E-003,
N+3.756100E-003,N+3.758400E-003,N+3.761100E-003,N+3.763100E-003,
N+3.765100E-003,N+3.767500E-003,N+3.769500E-003,N+3.771400E-003,
N+3.773200E-003,N+3.775000E-003,N-2.994400E-005,N+5.361800E-006,
```

Change 1

```
N+2.329700E-004,N+1.301800E-003,N+3.434800E-003,N+4.893700E-003,
N+5.394900E-003,N+5.533900E-003,N+5.570600E-003,N+5.582100E-003,
N+5.588300E-003,N+5.593000E-003,N+5.597400E-003,N+5.601100E-003,
N+5.605200E-003,N+5.608200E-003,N+5.611900E-003,N+5.614500E-003,
N+5.617800E-003,N+5.620900E-003,N+5.624000E-003,N-3.991500E-005,
N+1.014500E-005,N+3.295400E-004,N+1.773000E-003,N+4.518400E-003,
N+6.396700E-003,N+7.076300E-003,N+7.288700E-003,N+7.354300E-003,
N+7.376500E-003,N+7.387800E-003,N+7.395800E-003,N+7.402100E-003,
N+7.408100E-003,N+7.413600E-003,N+7.418800E-003,N+7.423500E-003,
N+7.428000E-003,N+7.432600E-003,N+7.437100E-003,N+7.440800E-003
```

Different Data Length in User Mode.
The following program is equivalent to sample program 2 listed in Section 3 of the HP 4145B manual. This program is an example of data output in the user mode.

```
10    ASSIGN @Hp415x TO 717
20    DIM A$[14]
30    OUTPUT @Hp415x;"US"
40    OUTPUT @Hp415x;"IT1 CA0 BC"
50    Force_v=1.5
60    OUTPUT @Hp415x;"DV1, 1, ";Force_v;", 20E-3"
70    OUTPUT @Hp415x;"DV2, 1, 0, 20E-3"
80    OUTPUT @Hp415x;"TI1"
90    ENTER @Hp415x;A$
100   PRINT A$
110   OUTPUT @Hp415x;"DV1; DV2"
120   END
```

This program displays the following for example. As you can see, the length of data string A$ is too short, so not all the data is displayed.

```
NAI+4.300000E-
```

This is because the length of data output from the HP 4155A/4156A and HP 4145B is different as follows:

HP 4145B data output format in user mode (14 characters):

```
XXX+NN.NNNE+NN[cr][lf]
```

HP 4155A/4156A data output format in user mode of 4145 syntax command mode (17 characters):

```
XXX+N.NNNNNNE+NNN[cr][lf]
```

X	Data status
+	+ or −
N	Numeric character

So, you need to change line 20 as follows:

```
20    DIM A$[17]           ! <<< Modified
```

The test result will be as follows:

```
NAI+4.300000E-013
```

TV and ID Commands

For "TV" command, the data length is 14 characters for HP 4145, and 17 characters for HP 4155A/4156A. For "ID" command, the data length is 16 characters for HP 4145 and 41 characters for HP 4155A/4156A. So, if these commands are used, you need to change the data string length accordingly.

Index

Index

FINAL TRIM SIZE : 7.5 in x 9.0 in